
CHAPTER 10
VHDL

The U.S. Department of Defense (DoD) supported the development of VHDL (VHSIC hardware
description language) as part of the VHSIC (very high-speed IC) program in the early 1980s. The
companies in the VHSIC program found they needed something more than schematic entry to describe
large ASICs, and proposed the creation of a hardware description language. VHDL was then handed
over to the Institute of Electrical and Electronics Engineers (IEEE) in order to develop and approve the
IEEE Standard 1076-1987. 1 As part of its standardization process the DoD has specified the use of
VHDL as the documentation, simulation, and verification medium for ASICs (MIL-STD-454). Partly
for this reason VHDL has gained rapid acceptance, initially for description and documentation, and then
for design entry, simulation, and synthesis as well.

The first revision of the 1076 standard was approved in 1993. References to the VHDL Language
Reference Manual (LRM) in this chapter--[VHDL 87LRM2.1, 93LRM2.2] for example--point to the
1987 and 1993 versions of the LRM [IEEE, 1076-1987 and 1076-1993]. The prefixes 87 and 93 are
omitted if the references are the same in both editions. Technically 1076-1987 (known as VHDL-87) is
now obsolete and replaced by 1076-1993 (known as VHDL-93). Except for code that is marked
’VHDL-93 only’ the examples in this chapter can be analyzed (the VHDL word for "compiled") and
simulated using both VHDL-87 and VHDL-93 systems.

10.1  A Counter

The following VHDL model describes an electrical "black box" that contains a 50 MHz clock generator
and a counter. The counter increments on the negative edge of the clock, counting from zero to seven,
and then begins at zero again. The model contains separate processes that execute at the same time as
each other. Modeling concurrent execution is the major difference between HDLs and computer
programming languages such as C.

entity Counter_1 is end; -- declare a "black box" called Counter_1
library STD; use STD.TEXTIO.all; -- we need this library to print
architecture Behave_1 of Counter_1 is -- describe the "black box" 
-- declare a signal for the clock, type BIT, initial value ’0’
        signal Clock : BIT := ’0’;
-- declare a signal for the count, type INTEGER, initial value 0
        signal Count : INTEGER := 0;
begin 
        process begin -- process to generate the clock
                wait for 10 ns; -- a delay of 10 ns is half the clock cycle
                Clock <= not Clock;
                if (now > 340 ns) then wait; end if; -- stop after 340 ns
        end process;
-- process to do the counting, runs concurrently with other processes
        process begin
-- wait here until the clock goes from 1 to 0
                wait until (Clock = ’0’);
-- now handle the counting
                if (Count = 7) then Count <= 0;
                else Count <= Count + 1;
                end if;
        end process;
        process (Count) variable L: LINE; begin -- process to print
                write(L, now); write(L, STRING’(" Count="));
                write(L, Count); writeline(output, L);
        end process;
end;



Throughout this book VHDL keywords (reserved words that are part of the language) are shown in bold
type in code examples (but not in the text). The code examples use the bold keywords to improve
readability. VHDL code is often lengthy and the code in this book is always complete wherever
possible. In order to save space many of the code examples do not use the conventional spacing and
formatting that is normally considered good practice. So "Do as I say and not as I do."

The steps to simulate the model and the printed results for Counter_1 using the Model Technology
V-System/Plus common-kernel simulator are as follows:

> vlib work
> vcom Counter_1.vhd
Model Technology VCOM V-System VHDL/Verilog 4.5b
-- Loading package standard
-- Compiling entity counter_1
-- Loading package textio
-- Compiling architecture behave_1 of counter_1
> vsim -c counter_1
# Loading /../std.standard
# Loading /../std.textio(body)
# Loading work.counter_1(behave_1)
VSIM 1> run 500
# 0 ns Count=0
# 20 ns Count=1
(...15 lines omitted...)
# 340 ns Count=1
VSIM 2> quit
>

10.2  A 4-bit Multiplier

This section presents a more complex VHDL example to motivate the study of the syntax and semantics
of VHDL in the rest of this chapter.

10.2.1  An 8-bit Adder

Table 10.1 shows a VHDL model for the full adder that we described in Section 2.6, "Datapath Logic
Cells." Table 10.2 shows a VHDL model for an 8-bit ripple-carry adder that uses eight instances of the
full adder.



TABLE 10.1    A full adder.

entity Full_Adder is
        generic (TS : TIME := 0.11 ns; TC : TIME := 0.1 ns);
        port (X, Y, Cin: in BIT; Cout, Sum: out  BIT);
end Full_Adder;
architecture Behave of Full_Adder is
begin 
Sum  <= X xor Y xor Cin after TS;
Cout <= (X and Y) or (X and Cin) or (Y and Cin) after TC;
end;

 

Timing:

TS (Input to Sum) = 0.1 1
ns

TC (Input to Cout) = 0.1
ns

 

TABLE 10.2    An 8-bit ripple-carry adder.
entity Adder8 is
        port (A, B: in BIT_VECTOR(7 downto 0);
        Cin: in BIT; Cout: out BIT; 
        Sum: out BIT_VECTOR(7 downto 0));
end Adder8;
architecture Structure of Adder8 is
component Full_Adder
port (X, Y, Cin: in BIT; Cout, Sum: out BIT);
end component;
signal C: BIT_VECTOR(7 downto 0);
begin 
Stages: for i in 7 downto 0 generate
        LowBit: if i = 0 generate
        FA:Full_Adder port map (A(0),B(0),Cin,C(0),Sum(0)); 
        end generate;
        OtherBits: if i /= 0 generate
        FA:Full_Adder port map 
                (A(i),B(i),C(i-1),C(i),Sum(i));
        end generate;
end generate;
Cout <= C(7);
end;

 

10.2.2  A Register Accumulator

Table 10.3 shows a VHDL model for a positive-edge-triggered D flip-flop with an active-high
asynchronous clear. Table 10.4 shows an 8-bit register that uses this D flip-flop model (this model only
provides the Q output from the register and leaves the QN flip-flop outputs unconnected).



TABLE 10.3    Positive-edge-triggered D flip-flop with asynchronous clear.
entity DFFClr is 
        generic(TRQ : TIME := 2 ns; TCQ : TIME := 2 ns);
        port (CLR, CLK, D : in BIT; Q, QB : out BIT); 
end;
architecture Behave of DFFClr is
signal Qi : BIT;
begin QB <= not Qi; Q <= Qi;
process (CLR, CLK) begin
        if CLR = ’1’ then Qi <= ’0’ after TRQ;
        elsif CLK’EVENT and CLK = ’1’ 
                then Qi <= D after TCQ;
        end if;
end process;
end;

 

Timing:

TRQ (CLR to Q/QN) = 2 ns

TCQ (CLK to Q/QN) = 2 ns

TABLE 10.4    An 8-bit register.

entity Register8 is 
        port (D : in BIT_VECTOR(7 downto 0); 
        Clk, Clr: in BIT ; Q : out BIT_VECTOR(7 downto 0));
end;
architecture Structure of Register8 is
        component DFFClr 
                port (Clr, Clk, D : in BIT; Q, QB : out BIT); 
        end component;
        begin 
                STAGES: for i in 7 downto 0 generate
                FF: DFFClr port map (Clr, Clk, D(i), Q(i), open);
                end generate;
end;

 

8-bit register. Uses

DFFClr positive
edge-triggered
flip-flop model.

Table 10.5 shows a model for a datapath multiplexer that consists of eight 2:1 multiplexers with a
common select input (this select signal would normally be a control signal in a datapath). The multiplier
will use the register and multiplexer components to implement a register accumulator.

TABLE 10.5    An 8-bit multiplexer.

entity Mux8 is 
        generic (TPD : TIME := 1 ns);
        port (A, B : in BIT_VECTOR (7 downto 0); 
        Sel : in BIT := ’0’; Y : out BIT_VECTOR (7 downto 0));
end;
architecture Behave of Mux8 is
begin 
         Y <= A after TPD when Sel = ’1’ else B after TPD;
end;

 

Eight 2:1 MUXs with

single select input.

Timing:

TPD (input to Y) = 1 ns



10.2.3  Zero Detector

Table 10.6 shows a model for a variable-width zero detector that accepts a bus of any width and will
produce a single-bit output of ’1’ if all input bits are zero.

TABLE 10.6    A zero detector.

entity AllZero is 
        generic (TPD : TIME := 1 ns);
        port (X : BIT_VECTOR; F : out BIT );
end;
architecture Behave of AllZero is
begin process (X) begin F <= ’1’ after TPD;
         for j in X’RANGE loop
                if X(j) = ’1’ then F <= ’0’ after TPD; end if;
         end loop;
end process;
end;

 

Variable-width zero
detector.

Timing:

TPD (X to F) = 1 ns

10.2.4  A Shift Register

Table 10.7 shows a variable-width shift register that shifts (left or right under input control, DIR ) on the
positive edge of the clock, CLK , gated by a shift enable, SH . The parallel load, LD , is synchronous and
aligns the input LSB to the LSB of the output, filling unused MSBs with zero. Bits vacated during shifts
are zero filled. The clear, CLR , is asynchronous.

TABLE 10.7    A variable-width shift register.

entity ShiftN is
        generic (TCQ : TIME := 0.3 ns; TLQ : TIME := 0.5 ns;
                TSQ : TIME := 0.7 ns);
        port(CLK, CLR, LD, SH, DIR: in BIT; 
                D: in BIT_VECTOR; Q: out BIT_VECTOR);

 

CLK Clock

CLR Clear,
active high

LD Load, active
high

SH Shift, active
high

DIR Direction,
1 = left



        begin assert (D’LENGTH <= Q’LENGTH) 
                report "D wider than output Q" severity Failure;
end ShiftN;
architecture Behave of ShiftN is
        begin Shift: process (CLR, CLK)
        subtype InB  is NATURAL range D’LENGTH-1 downto 0;
        subtype OutB is NATURAL range Q’LENGTH-1 downto 0;
        variable St: BIT_VECTOR(OutB);
        begin
                if CLR = ’1’ then 
                        St := (others => ’0’); Q <= St after TCQ;
                elsif CLK’EVENT and CLK=’1’ then
                        if LD = ’1’ then 
                                St := (others => ’0’); 
                                St(InB) := D; 
                                Q <= St after TLQ;
                        elsif SH = ’1’ then
                                case DIR is 
                                when ’0’ => St := ’0’ & St(St’LEFT downto 1);
                                when ’1’ => St := St(St’LEFT-1 downto 0) & ’0’;
                                end case;
                                Q <= St after TSQ;
                        end if;
                end if;
        end process;
end;

1 = left

D Data in

Q Data out

 

Variable-width
shift register.
Input width
must be less
than output
width. Output is
left-shifted or
right-shifted
under control of
DIR. Unused
MSBs are
zero-padded
during load.
Clear is
asynchronous.
Load is
synchronous.

 

Timing:

TCQ (CLR to Q)
= 0.3 ns

TLQ (LD to Q) =
0.5 ns

TSQ (SH to Q) =
0. 7 ns

10.2.5  A State Machine

To multiply two binary numbers A and B , we can use the following algorithm:

If the LSB of A is ’1’, then add B into an accumulator.

Shift A one bit to the right and B one bit to the left.

Stop when all bits of A are zero.



Table 10.8 shows the VHDL model for a Moore (outputs depend only on the state) finite-state machine
for the multiplier, together with its state diagram.

TABLE 10.8    A Moore state machine for the multiplier.

entity SM_1 is 
        generic (TPD : TIME := 1 ns);
        port(Start, Clk, LSB, Stop, Reset: in BIT; 
        Init, Shift, Add, Done : out BIT);
end;
architecture Moore of SM_1 is
type STATETYPE is (I, C, A, S, E);
signal State: STATETYPE;
begin 
Init <= ’1’ after TPD when State = I
        else ’0’ after TPD;
Add  <= ’1’ after TPD when State = A
        else ’0’ after TPD;
Shift <= ’1’ after TPD when State = S
        else ’0’ after TPD;
Done <= ’1’ after TPD when State = E
        else ’0’ after TPD;
process (CLK, Reset) begin
        if Reset = ’1’ then State <= E;
        elsif CLK’EVENT and CLK = ’1’ then
                case State is
                when I => State <= C;
                when C => 
                        if LSB = ’1’ then State <= A;
                        elsif Stop = ’0’ then State <= S;
                        else State <= E;
                        end if;
                when A => State <= S;
                when S => State <= C;
                when E => 
                        if Start = ’1’ then State <= I; end if; 
                end case;
        end if;
end process;
end;

 

State Function

 

E End of multiply cycle.

I Initialize: clear output

register and load input

registers.

C Check if LSB of register
A

is zero.

A Add shift register B to

accumulator.

S Shift input register A right

and input register B left.

10.2.6  A Multiplier



Table 10.9 shows a schematic and the VHDL code that describes the interconnection of all the
components for the multiplier. Notice that the schematic comprises two halves: an 8-bit-wide datapath
section (consisting of the registers, adder, multiplexer, and zero detector) and a control section (the
finite-state machine). The arrows in the schematic denote the inputs and outputs of each component. As
we shall see in Section 10.7, VHDL has strict rules about the direction of connections.

TABLE 10.9    A 4-bit by 4-bit multiplier.

 

entity Mult8 is
port (A, B: in BIT_VECTOR(3 downto 0); Start, CLK, Reset: in BIT;
Result: out BIT_VECTOR(7 downto 0); Done: out BIT); end Mult8;
architecture Structure of Mult8 is use work.Mult_Components.all;
signal SRA, SRB, ADDout, MUXout, REGout: BIT_VECTOR(7 downto 0);
signal Zero, Init, Shift, Add, Low: BIT := ’0’; signal High: BIT := ’1’;
signal F, OFL, REGclr: BIT; 
begin 
REGclr <= Init or Reset; Result  <= REGout;
SR1 : ShiftN port map(CLK=>CLK,CLR=>Reset,LD=>Init,SH=>Shift,DIR=>Low ,D=>A,Q=>SRA);
SR2 : ShiftN port map(CLK=>CLK,CLR=>Reset,LD=>Init,SH=>Shift,DIR=>High,D=>B,Q=>SRB);
Z1 : AllZero port map(X=>SRA,F=>Zero);
A1 : Adder8  port map(A=>SRB,B=>REGout,Cin=>Low,Cout=>OFL,Sum=>ADDout);
M1 : Mux8    port map(A=>ADDout,B=>REGout,Sel=>Add,Y=>MUXout);
R1 : Register8 port map(D=>MUXout,Q=>REGout,Clk=>CLK,Clr=>REGclr);
F1 : SM_1    port map(Start,CLK,SRA(0),Zero,Reset,Init,Shift,Add,Done);
end;

10.2.7  Packages and Testbench

To complete and test the multiplier design we need a few more items. First we need the following
"components list" for the items in Table 10.9:

package Mult_Components is
component Mux8 port (A,B:BIT_VECTOR(7 downto 0);
        Sel:BIT;Y:out BIT_VECTOR(7 downto 0));end component;
component AllZero port (X : BIT_VECTOR;



        F:out BIT );end component;
component Adder8 port (A,B:BIT_VECTOR(7 downto 0);Cin:BIT;
        Cout:out BIT;Sum:out BIT_VECTOR(7 downto 0));end component;
component Register8 port (D:BIT_VECTOR(7 downto 0);
        Clk,Clr:BIT; Q:out BIT_VECTOR(7 downto 0));end component;
component ShiftN port (CLK,CLR,LD,SH,DIR:BIT;D:BIT_VECTOR;
        Q:out BIT_VECTOR);end component;
component SM_1 port (Start,CLK,LSB,Stop,Reset:BIT;
        Init,Shift,Add,Done:out BIT);end component;
end;

Next we need some utility code to help test the multiplier. The following VHDL generates a clock with
programmable "high" time ( HT ) and "low" time ( LT ):

package Clock_Utils is 
procedure Clock (signal C: out Bit; HT, LT:TIME);
end Clock_Utils;
package body Clock_Utils is
procedure Clock (signal C: out Bit; HT, LT:TIME) is
begin 
        loop C<=’1’ after LT, ’0’ after LT + HT; wait for LT + HT;
        end loop;
end;
end Clock_Utils;

Finally, the following code defines two functions that we shall also use for testing--the functions convert
an array of bits to a number and vice versa:

package Utils is 
        function Convert (N,L: NATURAL) return BIT_VECTOR;
        function Convert (B: BIT_VECTOR) return NATURAL;
end Utils;
package body Utils is
        function Convert (N,L: NATURAL) return BIT_VECTOR is
                variable T:BIT_VECTOR(L-1 downto 0);
                variable V:NATURAL:= N;
                begin for i in T’RIGHT to T’LEFT loop
                        T(i) := BIT’VAL(V mod 2); V:= V/2;
                end loop; return T;
        end;
        function Convert (B: BIT_VECTOR) return NATURAL is
                variable T:BIT_VECTOR(B’LENGTH-1 downto 0) := B;
                variable V:NATURAL:= 0;
                begin for i in T’RIGHT to T’LEFT loop
                        if T(i) = ’1’ then V:= V + (2**i); end if;
                        end loop; return V;
                end;
end Utils;

The following code tests the multiplier model. This is a testbench (this simple example is not a
comprehensive test). First we reset the logic (line 17) and then apply a series of values to the inputs, A
and B . The clock generator (line 14) supplies a clock with a 20 ns period. The inputs are changed 1 ns
after a positive clock edge, and remain stable for 20 ns through the next positive clock edge.

entity Test_Mult8_1 is end; -- runs forever, use break!!
architecture Structure of Test_Mult8_1 is 
use Work.Utils.all; use Work.Clock_Utils.all;
        component Mult8 port



                (A, B : BIT_VECTOR(3 downto 0); Start, CLK, Reset : BIT; 
                Result : out BIT_VECTOR(7 downto 0); Done : out BIT);
        end component;
signal A, B : BIT_VECTOR(3 downto 0);
signal Start, Done : BIT := ’0’;
signal CLK, Reset : BIT;
signal Result : BIT_VECTOR(7 downto 0);
signal DA, DB, DR : INTEGER range 0 to 255;
begin 
C: Clock(CLK, 10 ns, 10 ns);
UUT: Mult8 port map (A, B, Start, CLK, Reset, Result, Done);
DR <= Convert(Result);
Reset  <= ’1’, ’0’ after 1 ns; 
process begin 
        for i in 1 to 3 loop for j in 4 to 7 loop
                DA <= i; DB <= j;
                A<=Convert(i,A’Length);B<=Convert(j,B’Length);
                wait until CLK’EVENT and CLK=’1’; wait for 1 ns; 
                Start <= ’1’, ’0’ after 20 ns; wait until Done = ’1’;
                wait until CLK’EVENT and CLK=’1’;
        end loop; end loop; 
        for i in 0 to 1 loop for j in 0 to 15 loop
                DA <= i; DB <= j;
                A<=Convert(i,A’Length);B<=Convert(j,B’Length);
                wait until CLK’EVENT and CLK=’1’; wait for 1 ns;
                Start <= ’1’, ’0’ after 20 ns; wait until Done = ’1’; 
                wait until CLK’EVENT and CLK=’1’;
        end loop; end loop;
        wait;
end process;
end;

Here is the signal trace output from the Compass Scout simulator:

      Time(fs) + Cycle            da           db           dr
----------------------  ------------ ------------ ------------
                  0+ 0:            0            0            0
                  0+ 1: *          1 *          4 *          0
...
           92000000+ 3:            1            4 *          4
...
          150000000+ 1: *          1 *          5            4
...
          193000000+ 3:            1            5 *          0
...
          252000000+ 3:            1            5 *          5
...
          310000000+ 1: *          1 *          6            5
...
          353000000+ 3:            1            6 *          0
...
          412000000+ 3:            1            6 *          6

Positive clock edges occur at 10, 30, 50, 70, 90, ... ns. You can see that the output (dr) changes from ’0’
to ’4’ at 92 ns, after five clock edges (with a 2 ns delay due to the output register, R1).

10.3  Syntax and Semantics of VHDL

We might define the syntax of a very small subset of the English language in Backus-Naur form (BNF)
using constructs as follows:

sentence ::= subject verb object.
subject  ::= The|A noun 
object   ::= [article] noun {, and article noun}
article  ::= the|a
noun     ::= man|shark|house|food
verb     ::= eats|paints

::= means "can be replaced by" 
|   means "or" 
[]  means "contents optional"
{}  means "contents can be left out, used once, or repeated"

The following two English sentences are correct according to these syntax rules:

A shark eats food.
The house paints the shark, and the house, and a man.

We need semantic rules to tell us that the second sentence does not make much sense. Most of the
VHDL LRM is dedicated to the definition of the language semantics. Appendix A of the LRM (which is
not officially part of the standard) explains the complete VHDL syntax using BNF.

The rules that determine the characters you can use (the "alphabet" of VHDL), where you can put
spaces, and so on are lexical rules [VHDL LRM13]. Any VHDL description may be written using a
subset of the VHDL character set:

basic_character ::= upper_case_letter|digit|special_character
        |space_character|format_effector

The two space characters are: space ( SP ) and the nonbreaking space ( NBSP ). The five format
effectors are: horizontal tabulation ( HT ), vertical tabulation ( VT ), carriage return ( CR ), line feed ( LF
), and form feed ( FF ). The characters that are legal in VHDL constructs are defined as the following
subsets of the complete character set:

graphic_character ::=
         upper_case_letter|digit|special_character|space_character
        |lower_case_letter|other_special_character
special_character ::= " # & ’ () * + , - . / : ; < = > [ ] _ |

The 11 other special characters are: ! $ % @ ? \ ^ ‘ { } ~ , and (in VHDL-93 only) 34 other
characters from the ISO Latin-1 set [ISO, 1987]. If you edit code using a word processor, you either
need to turn smart quotes off or override this feature (use Tools... Preferences... General in MS Word;
and use CTRL-’ and CTRL-" in Frame).

When you learn a language it is difficult to understand how to use a noun without using it in a sentence.
Strictly this means that we ought to define a sentence before we define a noun and so on. In this chapter
I shall often break the "Define it before you use it" rule and use code examples and BNF definitions that
contain VHDL constructs that we have not yet defined. This is often frustrating. You can use the book



index and the table of important VHDL constructs at the end of this chapter (Table 10.28) to help find
definitions if you need them.

We shall occasionally refer to the VHDL BNF syntax definitions in this chapter using references--BNF
[10.1], for example. Only the most important BNF constructs for VHDL are included here in this
chapter, but a complete description of the VHDL language syntax is contained in Appendix A.

10.4  Identifiers and Literals

Names (the "nouns" of VHDL) are known as identifiers [VHDL LRM13.3]. The correct "spelling" of an
identifier is defined in BNF as follows:

identifier ::= 
   letter {[underline] letter_or_digit}
  |\graphic_character{graphic_character}\

In this book an underline in VHDL BNF marks items that are new or that have changed in VHDL-93
from VHDL-87. The following are examples of identifiers:

s -- A simple name.
S -- A simple name, the same as s. VHDL is not case sensitive.
a_name -- Imbedded underscores are OK.
-- Successive underscores are illegal in names: Ill__egal
-- Names can’t start with underscore: _Illegal
-- Names can’t end with underscore: Illegal_
Too_Good -- Names must start with a letter.
-- Names can’t start with a number: 2_Bad 
\74LS00\ -- Extended identifier to break rules (VHDL-93 only).
VHDL \vhdl\ \VHDL\ -- Three different names (VHDL-93 only).
s_array(0) -- A static indexed name (known at analysis time).
s_array(i) -- A non-static indexed name, if i is a variable.

You may not use a reserved word as a declared identifier, and it is wise not to use units, special
characters, and function names: ns , ms , FF , read , write, and so on. You may attach qualifiers to
names as follows [VHDL LRM6]:

CMOS.all -- A selected or expanded name, all units in library CMOS.
Data’LEFT(1) -- An attribute name, LEFT is the attribute designator.
Data(24 downto 1) -- A slice name, part of an array: Data(31 downto 0)
Data(1) -- An indexed name, one element of an array.

Comments follow two hyphens ’--’ and instruct the analyzer to ignore the rest of the line. There are no
multiline comments in VHDL. Tabs improve readability, but it is best not to rely on a tab as a space in
case the tabs are lost or deleted in conversion. You should thus write code that is still legal if all tabs are
deleted.

There are various forms of literals (fixed-value items) in VHDL [VHDL LRM13.4-13.7]. The following
code shows some examples:



entity Literals_1 is end;
architecture Behave of Literals_1 is
begin process
        variable I1 : integer; variable Rl : real;
        variable C1 : CHARACTER; variable S16 : STRING(1 to 16);
        variable BV4: BIT_VECTOR(0 to 3); 
        variable BV12 : BIT_VECTOR(0 to 11);
        variable BV16 : BIT_VECTOR(0 to 15);
        begin
-- Abstract literals are decimal or based literals.
-- Decimal literals are integer or real literals.
-- Integer literal examples (each of these is the same):
                I1 := 120000; Int := 12e4; Int := 120_000; 
-- Based literal examples (each of these is the same):
                I1 := 2#1111_1111#; I1 := 16#FFFF#; 
-- Base must be an integer from 2 to 16:
                I1 := 16:FFFF:; -- you may use a : if you don’t have #
-- Real literal examples (each of these is the same):
                Rl := 120000.0; Rl := 1.2e5; Rl := 12.0E4; 
-- Character literal must be one of the 191 graphic characters.
-- 65 of the 256 ISO Latin-1 set are non-printing control characters
                C1 := ’A’; C1 := ’a’; -- different from each other
-- String literal examples:
                S16 := "  string" & " literal";         -- concatenate long strings
                S16 := """Hello,"" I said!";            -- doubled quotes
                S16 := %  string literal%;              -- can use % instead of "
                S16 := %Sale: 50%% off!!!%;             -- doubled %
-- Bit-string literal examples:
                BV4  := B"1100";                        -- binary bit-string literal
                BV12 := O"7777";                        -- octal  bit-string literal
                BV16 := X"FFFF";                        -- hex    bit-string literal
wait; end process; -- the wait prevents an endless loop
end;

10.5  Entities and Architectures

The highest-level VHDL construct is the design file [VHDL LRM11.1]. A design file contains design
units that contain one or more library units. Library units in turn contain: entity, configuration, and
package declarations (primary units); and architecture and package bodies (secondary units).

design_file ::= 
        {library_clause|use_clause} library_unit
        {{library_clause|use_clause} library_unit}
library_unit ::= primary_unit|secondary_unit
primary_unit ::=
        entity_declaration|configuration_declaration|package_declaration
secondary_unit ::= architecture_body|package_body

Using the written language analogy: a VHDL library unit is a "book," a VHDL design file is a
"bookshelf," and a VHDL library is a collection of bookshelves. A VHDL primary unit is a little like the
chapter title and contents that appear on the first page of each chapter in this book and a VHDL
secondary unit is like the chapter contents (though this is stretching our analogy a little far).

I shall describe the very important concepts of entities and architectures in this section and then cover



libraries, packages, and package bodies. You define an entity, a black box, using an entity declaration
[VHDL LRM1.1]. This is the BNF definition:

entity_declaration ::= 
entity identifier is
                [generic (formal_generic_interface_list);]
                [port (formal_port_interface_list);]
                {entity_declarative_item}
        [begin
                {[label:] [postponed] assertion ;
                |[label:] [postponed] passive_procedure_call ;
                |passive_process_statement}]
end [entity] [entity_identifier] ;

The following is an example of an entity declaration for a black box with two inputs and an output:

entity Half_Adder is 
        port (X, Y : in BIT := ’0’; Sum, Cout : out BIT); -- formals
end;

Matching the parts of this code with the constructs in BNF [10.7] you can see that the identifier is
Half_Adder and that (X, Y: in BIT := ’0’; Sum, Cout: out BIT) corresponds to
(port_interface_list) in the BNF. The ports X, Y, Sum, and Cout are formal ports or formals. This
particular entity Half_Adder does not use any of the other optional constructs that are legal in an entity
declaration.

The architecture body [VHDL LRM1.2] describes what an entity does, or the contents of the black box
(it is architecture body and not architecture declaration).

architecture_body ::=
        architecture identifier of entity_name is
                {block_declarative_item}
                        begin
                        {concurrent_statement}
        end [architecture] [architecture_identifier] ;

For example, the following architecture body (I shall just call it an architecture from now on) describes
the contents of the entity Half_Adder :

architecture Behave of Half_Adder is
        begin Sum <= X xor Y; Cout <= X and Y;
end Behave;

We use the same signal names (the formals: Sum , X , Y , and Cout ) in the architecture as we use in the
entity (we say the signals of the "parent" entity are visible inside the architecture "child"). An
architecture can refer to other entity-architecture pairs--so we can nest black boxes. We shall often refer
to an entity-architecture pair as entity(architecture). For example, the architecture Behave of the
entity Half_Adder is Half_Adder(Behave).

Why would we want to describe the outside of a black box (an entity) separately from the description of
its contents (its architecture)? Separating the two makes it easier to move between different architectures
for an entity (there must be at least one). For example, one architecture may model an entity at a
behavioral level, while another architecture may be a structural model.



A structural model that uses an entity in an architecture must declare that entity and its interface using a
component declaration as follows [VHDL LRM4.5]:

component_declaration ::=
        component identifier [is]
                [generic (local_generic_interface_list);]
                [port (local_port_interface_list);]
        end component [component_identifier];

For example, the following architecture, Netlist , is a structural version of the behavioral architecture,
Behave :

architecture Netlist of Half_Adder is
component MyXor port (A_Xor,B_Xor : in BIT; Z_Xor : out BIT);
end component; -- component with locals
component MyAnd port (A_And,B_And : in BIT; Z_And : out BIT);
end component; -- component with locals
begin
        Xor1: MyXor port map (X, Y, Sum);                                                                                                                       -- instance with actuals
        And1 : MyAnd port map (X, Y, Cout);                                                                                                                     -- instance with actuals
end;

Notice that:

We declare the components: MyAnd, MyXor and their local ports (or locals): A_Xor, B_Xor,
Z_Xor, A_And, B_And, Z_And. 
We instantiate the components with instance names: And1 and Xor1. 
We connect instances using actual ports (or actuals): X, Y , Sum , Cout. 

Next we define the entities and architectures that we shall use for the components MyAnd and MyXor .
You can think of an entity-architecture pair (and its formal ports) as a data-book specification for a logic
cell; the component (and its local ports) corresponds to a software model for the logic cell; and an
instance (and its actual ports) is the logic cell.

We do not need to write VHDL code for MyAnd and MyXor ; the code is provided as a technology library
(also called an ASIC vendor library because it is often sold or distributed by the ASIC company that will
manufacture the chip--the ASIC vendor--and not the software company):

-- These definitions are part of a technology library:
entity AndGate is 
        port (And_in_1, And_in_2 : in BIT; And_out : out BIT); -- formals
end;
architecture Simple of AndGate is
        begin And_out <= And_in_1 and And_in_2;
end;
entity XorGate is
        port (Xor_in_1, Xor_in_2 : in BIT; Xor_out : out BIT); -- formals
end;
architecture Simple of XorGate is
        begin Xor_out <= Xor_in_1 xor Xor_in_2;
end;

If we keep the description of a circuit’s interface (the entity ) separate from its contents (the
architecture ), we need a way to link or bind them together. A configuration declaration [VHDL
LRM1.3] binds entities and architectures.



configuration_declaration ::= 
        configuration identifier of entity_name is
                {use_clause|attribute_specification|group_declaration}
                block_configuration
        end [configuration] [configuration_identifier] ;

An entity-architecture pair is a design entity. The following configuration declaration defines which
design entities we wish to use and associates the formal ports (from the entity declaration) with the local
ports (from the component declaration):

configuration Simplest of Half_Adder is
use work.all;
        for Netlist
                for And1 : MyAnd use entity AndGate(Simple) 
                        port map -- association: formals => locals
                                (And_in_1 => A_And, And_in_2 => B_And, And_out => Z_And);
                end for;
                for Xor1 : MyXor use entity XorGate(Simple)
                        port map 
                        (Xor_in_1 => A_Xor, Xor_in_2 => B_Xor, Xor_out => Z_Xor);
                end for;
        end for;
end;

Figure 10.1 diagrams the use of entities, architectures, components, and configurations. This figure
seems very complicated, but there are two reasons that VHDL works this way:

Separating the entity, architecture, component, and configuration makes it easier to reuse code and
change libraries. All we have to do is change names in the port maps and configuration
declaration. 
We only have to alter and reanalyze the configuration declaration to change which architectures
we use in a model--giving us a fast debug cycle. 



  

FIGURE 10.1   Entities, architectures, components, ports, port maps, and configurations.

You can think of design units, the analyzed entity-architecture pairs, as compiled object-code modules.
The configuration then determines which object-code modules are linked together to form executable
binary code.

You may also think of an entity as a block diagram, an architecture for an entity a more detailed circuit
schematic for the block diagram, and a configuration as a parts list of the circuit components with their
part numbers and manufacturers (also known as a BOM for bill of materials, rather like a shopping list).
Most manufacturers (including the U.S. DoD) use schematics and BOMs as control documents for
electronic systems. This is part of the rationale behind the structure of VHDL.

page  Next  page

10.6  Packages and Libraries

After the VHDL tool has analyzed entities, architectures, and configurations, it stores the resulting
design units in a library. Much of the power of VHDL comes from the use of predefined libraries and
packages. A VHDL design library [VHDL LRM11.2] is either the current working library (things we are
currently analyzing) or a predefined resource library (something we did yesterday, or we bought, or that
came with the tool). The working library is named work and is the place where the code currently being
analyzed is stored. Architectures must be in the same library (but they do not have to be in the same
physical file on disk) as their parent entities.



You can use a VHDL package [VHDL LRM2.5-2.6] to define subprograms (procedures and functions),
declare special types, modify the behavior of operators, or to hide complex code. Here is the BNF for a
package declaration:

package_declaration ::= 
package identifier is
{subprogram_declaration         | type_declaration      | subtype_declaration
        | constant_declaration  | signal_declaration    | file_declaration
        | alias_declaration     | component_declaration
        | attribute_declaration | attribute_specification
        | disconnection_specification | use_clause
        | shared_variable_declaration | group_declaration
        | group_template_declaration}
        end [package] [package_identifier] ;

You need a package body if you declare any subprograms in the package declaration (a package
declaration and its body do not have to be in the same file):

package_body ::=
        package body package_identifier is
{subprogram_declaration         | subprogram_body
        | type_declaration      | subtype_declaration
        | constant_declaration  | file_declaration      | alias_declaration
        | use_clause     
        | shared_variable_declaration | group_declaration
        | group_template_declaration}
        end [package body] [package_identifier] ;

To make a package visible [VHDL LRM10.3] (or accessible, so you can see and use the package and its
contents), you must include a library clause before a design unit and a use clause either before a design
unit or inside a unit, like this:

library MyLib; -- library clause
use MyLib.MyPackage.all; -- use clause
-- design unit (entity + architecture, etc.) follows:

The STD and WORK libraries and the STANDARD package are always visible. Things that are visible to an
entity are visible to its architecture bodies.

10.6.1  Standard Package

The VHDL STANDARD package [VHDL LRM14.2] is defined in the LRM and implicitly declares the
following implementation dependent types: TIME , INTEGER , REAL . We shall use uppercase for types
defined in an IEEE standard package. Here is part of the STANDARD package showing the explicit type
and subtype declarations:

package Part_STANDARD is
type BOOLEAN
 is (FALSE, TRUE); type BIT
 is (’0’, ’1’);
type SEVERITY_LEVEL
 is (NOTE, WARNING, ERROR, FAILURE);
subtype NATURAL
 is INTEGER range 0 to INTEGER’HIGH;
subtype POSITIVE



 is INTEGER range 1 to INTEGER’HIGH;
type BIT_VECTOR
 is array (NATURAL range <>) of BIT;
type STRING
 is array (POSITIVE range <>) of CHARACTER;
-- the following declarations are VHDL-93 only:
attribute FOREIGN: STRING; -- for links to other languages
subtype DELAY_LENGTH is TIME range 0 fs to TIME’HIGH;
type FILE_OPEN_KIND is (READ_MODE,WRITE_MODE,APPEND_MODE);
type FILE_OPEN_STATUS is
(OPEN_OK,STATUS_ERROR,NAME_ERROR,MODE_ERROR);
end Part_STANDARD;

Notice that a STRING array must have a positive index. The type TIME is declared in the STANDARD
package as follows:

type TIME is range implementation_defined -- and varies with software
        units fs; ps = 1000 fs; ns = 1000 ps; us = 1000 ns; ms = 1000 us; 
        sec = 1000 ms; min = 60 sec; hr = 60 min; end units;

The STANDARD package also declares the function now that returns the current simulation time (with type
TIME in VHDL-87 and subtype DELAY_LENGTH in VHDL-93).

In VHDL-93 the CHARACTER type declaration extends the VHDL-87 declaration (the 128 ASCII
characters):

type Part_CHARACTER is ( -- 128 ASCII characters in VHDL-87 
NUL, SOH, STX, ETX, EOT, ENQ, ACK, BEL, -- 33 control characters 
 BS,  HT,  LF,  VT,  FF,  CR,  SO,  SI, -- including:
DLE, DC1, DC2, DC3, DC4, NAK, SYN, ETB, -- format effectors:
CAN,  EM, SUB, ESC, FSP, GSP, RSP, USP, -- horizontal tab = HT
’ ’, ’!’, ’"’, ’#’, ’$’, ’%’, ’&’, ’’’, -- line feed = LF
’(’, ’)’, ’*’, ’+’, ’,’, ’-’, ’.’, ’/’, -- vertical tab = VT
’0’, ’1’, ’2’, ’3’, ’4’, ’5’, ’6’, ’7’, -- form feed = FF
’8’, ’9’, ’:’, ’;’, ’<’, ’=’, ’>’, ’?’, -- carriage return = CR
’@’, ’A’, ’B’, ’C’, ’D’, ’E’, ’F’, ’G’, -- and others:
’H’, ’I’, ’J’, ’K’, ’L’, ’M’, ’N’, ’O’, -- FSP, GSP, RSP, USP use P
’P’, ’Q’, ’R’, ’S’, ’T’, ’U’, ’V’, ’W’, -- suffix to avoid conflict
’X’, ’Y’, ’Z’, ’[’, ’\’, ’]’, ’^’, ’_’, -- with TIME units
’‘’, ’a’, ’b’, ’c’, ’d’, ’e’, ’f’, ’g’, 
’h’, ’i’, ’j’, ’k’, ’l’, ’m’, ’n’, ’o’, 
’p’, ’q’, ’r’, ’s’, ’t’, ’u’, ’v’, ’w’, 
’x’, ’y’, ’z’, ’{’, ’|’, ’}’, ’~’, DEL  -- delete = DEL
-- VHDL-93 includes 96 more Latin-1 characters, like ¥ (Yen) and 
-- 32 more control characters, better not to use any of them.
);

The VHDL-87 character set is the 7-bit coded ISO 646-1983 standard known as the ASCII character set.
Each of the printable ASCII graphic character codes (there are 33 nonprintable control codes, like DEL
for delete) is represented by a graphic symbol (the shapes of letters on the keyboard, on the display, and
that actually print). VHDL-93 uses the 8-bit coded character set ISO 8859-1:1987(E), known as ISO
Latin-1. The first 128 characters of the 256 characters in ISO Latin-1 correspond to the 128-character
ASCII code. The graphic symbols for the printable ASCII characters are well defined, but not part of the
standard (for example, the shape of the graphic symbol that represents ’lowercase a’ is recognizable on
every keyboard, display, and font). However, the graphic symbols that represent the printable characters
from other 128-character codes of the ISO 8-bit character set are different in various fonts, languages,



and computer systems. For example, a pound sterling sign in a U.K. character set looks like this-’£’, but
in some fonts the same character code prints as ’#’ (known as number sign, hash, or pound). If you use
such characters and want to share your models with people in different countries, this can cause
problems (you can see all 256 characters in a character set by using Insert... Symbol in MS Word).

10.6.2  Std_logic_1164 Package

VHDL does not have a built-in logic-value system. The STANDARD package predefines the type BIT with
two logic values, ’0’ and ’1’ , but we normally need at least two more values: ’X’ (unknown) and ’Z’
(high-impedance). Unknown is a metalogical value because it does not exist in real hardware but is
needed for simulation purposes. We could define our own logic-value system with four logic values:

type MVL4 is (’X’, ’0’, ’1’, ’Z’); -- a four-value logic system

The proliferation of VHDL logic-value systems prompted the creation of the Std_logic_1164 package
(defined in IEEE Std 1164-1993) that includes functions to perform logical, shift, resolution, and
conversion functions for types defined in the Std_logic_1164 system. To use this package in a design
unit, you must include the following library clause (before each design unit) and a use clause (either
before or inside the unit):

library IEEE; use IEEE.std_logic_1164.all;

This Std_Logic_1164 package contains definitions for a nine-value logic system. The following code
and comments show the definitions and use of the most important parts of the package 1:

package Part_STD_LOGIC_1164 is
type STD_ULOGIC is
(       ’U’, -- Uninitialized
        ’X’, -- Forcing Unknown
        ’0’, -- Forcing 0
        ’1’, -- Forcing 1
        ’Z’, -- High Impedance
        ’W’, -- Weak Unknown
        ’L’, -- Weak 0
        ’H’, -- Weak 1
        ’-’  -- Don’t Care);
type STD_ULOGIC_VECTOR is array (NATURAL range <>) of STD_ULOGIC;
function resolved (s : STD_ULOGIC_VECTOR) return STD_ULOGIC;
subtype STD_LOGIC is resolved STD_ULOGIC;
type STD_LOGIC_VECTOR is array (NATURAL range <>) of STD_LOGIC;
subtype X01   is resolved STD_ULOGIC range ’X’ to ’1’;
subtype X01Z  is resolved STD_ULOGIC range ’X’ to ’Z’;
subtype UX01  is resolved STD_ULOGIC range ’U’ to ’1’;
subtype UX01Z is resolved STD_ULOGIC range ’U’ to ’Z’; 
-- Vectorized overloaded logical operators:
function "and"  (L : STD_ULOGIC; R : STD_ULOGIC) return UX01;
-- Logical operators not, and, nand, or, nor, xor, xnor (VHDL-93),
-- overloaded for STD_ULOGIC STD_ULOGIC_VECTOR STD_LOGIC_VECTOR.
-- Strength strippers and type conversion functions:
-- function To_T (X : F) return T; 
-- defined for types, T and F, where 
-- F=BIT BIT_VECTOR STD_ULOGIC STD_ULOGIC_VECTOR STD_LOGIC_VECTOR
-- T=types F plus types X01 X01Z UX01 (but not type UX01Z)
-- Exclude _’s in T in name: TO_STDULOGIC not TO_STD_ULOGIC
-- To_XO1 : L->0, H->1 others->X



-- To_XO1Z: Z->Z, others as To_X01
-- To_UX01: U->U, others as To_X01
-- Edge detection functions:
function rising_edge  (signal s: STD_ULOGIC) return BOOLEAN;
function falling_edge (signal s: STD_ULOGIC) return BOOLEAN;
-- Unknown detection (returns true if s = U, X, Z, W):
-- function Is_X (s : T) return BOOLEAN;
-- defined for T = STD_ULOGIC STD_ULOGIC_VECTOR STD_LOGIC_VECTOR.
end Part_STD_LOGIC_1164;

Notice:

The type STD_ULOGIC has nine logic values. For this reason IEEE Std 1164 is sometimes referred
to as MVL9--multivalued logic nine. There are simpler, but nonstandard, MVL4 and MVL7
packages, as well as packages with more than nine logic values, available. Values ’U’ , ’X’ , and
’W’ are all metalogical values. 
There are weak and forcing logic-value strengths. If more than one logic gate drives a node (there
is more than one driver) as in wired-OR logic or a three-state bus, for example, the simulator
checks the driver strengths to resolve the actual logic value of the node using the resolution
function, resolved , defined in the package. 
The subtype STD_LOGIC is the resolved version of the unresolved type STD_ULOGIC. Since
subtypes are compatible with types (you can assign one to the other) you can use either STD_LOGIC
or STD_ULOGIC for a signal with a single driver, but it is generally safer to use STD_LOGIC. 
The type STD_LOGIC_VECTOR is the resolved version of unresolved type STD_ULOGIC_VECTOR.
Since these are two different types and are not compatible, you should use STD_LOGIC_VECTOR.
That way you will not run into a problem when you try to connect a STD_LOGIC_VECTOR to a
STD_ULOGIC_VECTOR. 
The don’t care logic value ’-’ (hyphen), is principally for use by synthesis tools. The value ’-’ is
almost always treated the same as ’X’. 
The 1164 standard defines (or overloads) the logical operators for the STD_LOGIC types but not the
arithmetic operators (see Section 10.12). 

10.6.3  Textio Package

You can use the textio package, which is part of the library STD , for text input and output [VHDL
LRM14.3]. The following code is a part of the TEXTIO package header and, together with the comments,
shows the declarations of types, subtypes, and the use of the procedures in the package:

package Part_TEXTIO is -- VHDL-93 version.
type LINE is access STRING; -- LINE is a pointer to a STRING value.
type TEXT is file of STRING; -- File of ASCII records. 
type SIDE is (RIGHT, LEFT); -- for justifying output data. 
subtype WIDTH is NATURAL; -- for specifying widths of output fields. 
file INPUT : TEXT open READ_MODE is "STD_INPUT"; -- Default input file.
file OUTPUT : TEXT open WRITE_MODE is "STD_OUTPUT"; -- Default output.
-- The following procedures are defined for types, T, where 
-- T = BIT BIT_VECTOR BOOLEAN CHARACTER INTEGER REAL TIME STRING
--              procedure READLINE(file F : TEXT; L : out LINE);
--              procedure READ(L : inout LINE; VALUE : out T);
--              procedure READ(L : inout LINE; VALUE : out T; GOOD: out BOOLEAN);
--              procedure WRITELINE(F : out TEXT; L : inout LINE);
--              procedure WRITE(
--                      L : inout LINE; 



--                      VALUE : in T; 
--                      JUSTIFIED : in SIDE:= RIGHT; 
--                      FIELD:in WIDTH := 0; 
--                      DIGITS:in NATURAL := 0; -- for T = REAL only
--                      UNIT:in TIME:= ns); -- for T = TIME only
-- function ENDFILE(F : in TEXT) return BOOLEAN;
end Part_TEXTIO;

Here is an example that illustrates how to write to the screen (STD_OUTPUT ):

library std; use std.textio.all; entity Text is end;
architecture Behave of Text is signal count : INTEGER := 0;
begin count <= 1 after 10 ns, 2 after 20 ns, 3 after 30 ns;
process (count) variable L: LINE; begin 
if (count > 0) then 
        write(L, now); -- Write time.
        write(L, STRING’(" count=")); -- STRING’ is a type qualification.
        write(L, count); writeline(output, L);
end if; end process; end;
10 ns count=1
20 ns count=2
30 ns count=3

10.6.4  Other Packages

VHDL does not predefine arithmetic operators on types that hold bits. Many VHDL simulators provide
one or more arithmetic packages that allow you to perform arithmetic operations on std_logic_1164
types. Some companies also provide one or more math packages that contain functions for floating-point
algebra, trigonometry, complex algebra, queueing, and statistics (see also [IEEE 1076.2, 1996]).

Synthesis tool companies often provide a special version of an arithmetic package, a synthesis package,
that allows you to synthesize VHDL that includes arithmetic operators. This type of package may
contain special instructions (normally comments that are recognized by the synthesis software) that map
common functions (adders, subtracters, multipliers, shift registers, counters, and so on) to ASIC library
cells. I shall introduce the IEEE synthesis package in Section 10.12.

Synthesis companies may also provide component packages for such cells as power and ground pads,
I/O buffers, clock drivers, three-state pads, and bus keepers. These components may be
technology-independent (generic) and are mapped to primitives from technology-dependent libraries
after synthesis.

10.6.5  Creating Packages

It is often useful to define constants in one central place rather than using literals wherever you need a
specific value in your code. One way to do this is by using VHDL packaged constants [VHDL
LRM4.3.1.1] that you define in a package. Packages that you define are initially part of the working
library, work . Here are two example packages [VHDL LRM2.5-2.7]:

package Adder_Pkg is -- a package declaration
        constant BUSWIDTH : INTEGER := 16; 
end Adder_Pkg;
use work.Adder_Pkg.all; -- a use clause
entity Adder is end Adder;
architecture Flexible of Adder is -- work.Adder_Pkg is visible here



        begin process begin 
                MyLoop : for j in 0 to BUSWIDTH loop -- adder code goes here
                end loop; wait; -- the wait prevents an endless cycle
        end process;
end Flexible;
package GLOBALS is 
        constant HI : BIT := ’1’; constant LO: BIT := ’0’;
end GLOBALS;

Here is a package that declares a function and thus requires a package body:

package Add_Pkg_Fn is
function add(a, b, c : BIT_VECTOR(3 downto 0)) return BIT_VECTOR;
end Add_Pkg_Fn;
package body Add_Pkg_Fn is
function add(a, b, c : BIT_VECTOR(3 downto 0)) return BIT_VECTOR is
        begin return a xor b xor c; end; 
end Add_Pkg_Fn;

The following example is similar to the VITAL (VHDL Initiative Toward ASIC Libraries) package that
provides two alternative methods (procedures or functions) to model primitive gates (I shall describe
functions and procedures in more detail in Section 10.9.2):

package And_Pkg is 
        procedure V_And(a, b : BIT; signal c : out BIT); 
        function V_And(a, b : BIT) return BIT;
end;
package body And_Pkg is 
        procedure V_And(a, b : BIT; signal c : out BIT) is 
                begin c <= a and b; end;
        function V_And(a, b : BIT) return BIT is 
                begin return a and b; end;
end And_Pkg;

The software determines where it stores the design units that we analyze. Suppose the package
Add_Pkg_Fn is in library MyLib . Then we need a library clause (before each design unit) and use clause
with a selected name to use the package:

library MyLib; -- use MyLib.Add_Pkg.all; -- use all the package
use MyLib.Add_Pkg_Fn.add; -- just function ’add’ from the package
entity Lib_1 is port (s : out BIT_VECTOR(3 downto 0) := "0000"); end;
architecture Behave of Lib_1 is begin process
begin s <= add ("0001", "0010", "1000"); wait; end process; end;

The VHDL software dictates how you create the library MyLib from the library work and the actual
name and directory location for the physical file or directory on the disk that holds the library. The
mechanism to create the links between the file and directory names in the computer world and the
library names in the VHDL world depends on the software. There are three common methods:

Use a UNIX environment variable (SETENV MyLib ~/MyDirectory/
MyLibFile , for example). 
Create a separate file that establishes the links between the filename known to the operating
system and the library name known to the VHDL software. 
Include the links in an initialization file (often with an ’.ini’ suffix). 



1. 
rights reserved.

10.7  Interface Declarations

An interface declaration declares interface objects that may be interface constants, signals, variables, or
files [VHDL 87LRM4.3.3, 93LRM4.3.2]. Interface constants are generics of a design entity, a
component, or a block, or parameters of subprograms. Interface signals are ports of a design entity,
component, or block, and parameters of subprograms. Interface variables and interface files are
parameters of subprograms.

Each interface object has a mode that indicates the direction of information flow. The most common
modes are in (the default), out , inout , and buffer (a fifth mode, linkage , is used to communicate
with other languages and is infrequently used in ASIC design). The restrictions on the use of objects
with these modes are listed in Table 10.10. An interface object is read when you use it on the RHS of an
assignment statement, for example, or when the object is associated with another interface object of
modes in , inout (or linkage ). An interface object is updated when you use it on the LHS side of an
assignment statement or when the object is associated with another interface object of mode out ,
buffer , inout (or linkage ). The restrictions on reading and updating objects generate the diagram at
the bottom of Table 10.10 that shows the 10 allowed types of interconnections (these rules for modes
buffer and inout are the same). The interface objects ( Inside and Outside ) in the example in this
table are ports (and thus interface signals), but remember that interface objects may also be interface
constants, variables, and files.



TABLE 10.10    Modes of interface objects and their properties.
entity E1 is port (Inside : in BIT); end; architecture Behave of E1 is begin end;
entity E2 is port (Outside : inout BIT := ’1’); end; architecture Behave of E2 is 
component E1 port (Inside: in BIT); end component; signal UpdateMe : BIT; begin 
I1 : E1 port map (Inside => Outside); -- formal/local (mode in) => actual (mode inout)
UpdateMe <= Outside; -- OK to read Outside (mode inout)
Outside  <= ’0’ after 10 ns; -- and OK to update Outside (mode inout)
end;

Possible modes of interface
object, Outside

in (default) out inout buffer

Can you read Outside (RHS of
assignment)?

Yes No Yes Yes

Can you update Outside (LHS
of assignment)?

No Yes Yes Yes

Modes of Inside that Outside
may connect to (see below) 1

in out any any

 

There are other special-case rules for reading and updating interface signals, constants, variables, and
files that I shall cover in the following sections. The situation is like the spelling rule, "i before e except
after c." Table 10.10 corresponds to the rule "i before e."

10.7.1  Port Declaration

Interface objects that are signals are called ports [VHDL 93LRM1.1.1.2]. You may think of ports as
"connectors" and you must declare them as follows:

port (port_interface_list)
interface_list ::=
        port_interface_declaration {; port_interface_declaration}

A port interface declaration is a list of ports that are the inputs and outputs of an entity, a block, or a
component declaration:

interface_declaration ::=
        [signal] 
                identifier {, identifier}:[in|out|inout|buffer|linkage]
                subtype_indication [bus] [:= static_expression]



Each port forms an implicit signal declaration and has a port mode. I shall discuss bus , which is a signal
kind, in Section 10.13.1. Here is an example of an entity declaration that has five ports:

entity Association_1 is 
        port (signal X, Y : in BIT := ’0’; Z1, Z2, Z3 : out BIT);
end;

In the preceding declaration the keyword signal is redundant (because all ports are signals) and may be
omitted. You may also omit the port mode in because it is the default mode. In this example, the input
ports X and Y are driven by a default value (in general a default expression) of ’0’ if (and only if ) the
ports are left unconnected or open. If you do leave an input port open, the port must have a default
expression.

You use a port map and either positional association or named association to connect the formals of an
entity with the locals of a component. Port maps also associate (connect) the locals of a component with
the actuals of an instance. For an example of formal, local, and actual ports, and explanation of their
function, see Section 10.5, where we declared an entity AndGate. The following example shows how to
bind a component to the entity AndGate (in this case we use the default binding) and associate the ports.
Notice that if we mix positional and named association then all positional associations must come first.

use work.all; -- makes analyzed design entity AndGate(Simple) visible.
architecture Netlist of Association_1 is
-- The formal port clause for entity AndGate looks like this:
-- port (And_in_1, And_in_2: in BIT; And_out : out BIT); -- Formals.
component AndGate port 
        (And_in_1, And_in_2 : in BIT; And_out : out BIT); -- Locals.
end component;
begin
-- The component and entity have the same names: AndGate.
-- The port names are also the same: And_in_1, And_in_2, And_out,
-- so we can use default binding without a configuration.
-- The last (and only) architecture for AndGate will be used: Simple.
A1:AndGate port map (X, Y, Z1); -- positional association
A2:AndGate port map (And_in_2=>Y, And_out=>Z2, And_in_1=>X);                                                                                                                                                                                                    -- named
A3:AndGate port map (X, And_out => Z3, And_in_2 => Y);                                                                                                                                                                                                  -- both
end;

The interface object rules of Table 10.10 apply to ports. The rule that forbids updating an interface
object of mode in prevents modifying an input port (by placing the input signal on the left-hand side of
an assignment statement, for example). Less obviously, you cannot read a port of mode out (that is you
cannot place an output signal on the right-hand side of an assignment statement). This stops you from
accidentally reading an output signal that may be connected to a net with multiple drivers. In this case
the value you would read (the unresolved output signal) might not be the same as the resolved signal
value. For example, in the following code, since Clock is a port of mode out , you cannot read Clock
directly. Instead you can transfer Clock to an intermediate variable and read the intermediate variable
instead:

entity ClockGen_1 is port (Clock : out BIT); end;
architecture Behave of ClockGen_1 is
begin process variable Temp : BIT := ’1’;
        begin
--      Clock <= not Clock; -- Illegal, you cannot read Clock (mode out),
        Temp := not Temp; -- use a temporary variable instead.
        Clock <= Temp after 10 ns; wait for 10 ns;



        if (now > 100 ns) then wait; end if; end process;
end;

TABLE 10.11    Properties of ports.

Example entity declaration:

entity E is port (F_1:BIT; F_2:out BIT; F_3:inout BIT; F_4:buffer BIT); end;  -- formals

Example component declaration:

component C port (L_1:BIT; L_2:out BIT; L_3:inout BIT; L_4:buffer BIT); -- locals
end component;

Example component instantiation:

I1 : C port map 
(L_1 => A_1, L_2 => A_2, L_3 => A_3, L_4 => A_4); -- locals => actuals

Example configuration:

for I1 : C use entity E(Behave) port map 
(F_1 => L_1, F_2 => L_2, F_3 => L_3, F_4 => L_4); -- formals => locals

Interface object,
port F

 F_1  F_2  F_3  F_4

Mode of F in (default) out inout buffer

Can you read
attributes of F?

[VHDL
LRM4.3.2]

Yes, but not the
attributes:

’STABLE

’QUIET

’DELAYED

’TRANSACTION

Yes, but not the
attributes:

’STABLE ’QUIET

’DELAYED

’TRANSACTION

’EVENT ’ACTIVE

’LAST_EVENT

’LAST_ACTIVE

’LAST_VALUE

Yes, but not the
attributes:

’STABLE

’QUIET

’DELAYED

’TRANSACTION

Yes

Table 10.10 lists the restrictions on reading and updating interface objects including interface signals
that form ports. Table 10.11 lists additional special rules for reading and updating the attributes of
interface signals.

There is one more set of rules that apply to port connections [VHDL LRM 1.1.1.2]. If design entity E2
contains an instance, I1 , of design entity E1 , then the formals (of design entity E1 ) are associated with



actuals (of instance I1 ). The actuals (of instance I1 ) are themselves formal ports (of design entity E2 ).
The restrictions illustrated in Table 10.12 apply to the modes of the port connections from E1 to E2
(looking from the inside to the outside).

Notice that the allowed connections diagrammed in Table 10.12 (looking from inside to the outside) are
a superset of those of Table 10.10 (looking from the outside to the inside). Only the seven types of
connections shown in Table 10.12 are allowed between the ports of nested design entities. The
additional rule that ports of mode buffer may only have one source, together with the restrictions on
port mode interconnections, limits the use of ports of mode buffer .

TABLE 10.12    Connection rules for port modes.
 entity E1 is port (Inside : in BIT); end; architecture Behave of E1 is begin end;
entity E2 is port (Outside : inout BIT := ’1’); end; architecture Behave of E2 is 
component E1 port (Inside : in BIT); end component; begin 
I1 : E1 port map (Inside => Outside);                                                                                                                           -- formal/local (mode in) => actual (mode inout)
end;

Possible modes of interface object, Inside in (default) out

Modes of Outside that Inside may
connect to (see below)

in  inout  buffer out  inout

 

10.7.2  Generics

Ports are signals that carry changing information between entities. A generic is similar to a port, except
generics carry constant, static information [VHDL LRM1.1.1.1]. A generic is an interface constant that,
unlike normal VHDL constants, may be given a value in a component instantiation statement or in a
configuration specification. You declare generics in an entity declaration and you use generics in a
similar fashion to ports. The following example uses a generic parameter to alter the size of a gate:

entity AndGateNWide is 
        generic (N : NATURAL := 2);
        port (Inputs : BIT_VECTOR(1 to N); Result : out BIT); 
end;

Notice that the generic interface list precedes the port interface list. Generics are useful to carry timing
(delay) information, as in the next example:

entity AndT is 
        generic (TPD : TIME := 1 ns);



        port (a, b : BIT := ’0’; q: out BIT);
end;
architecture Behave of AndT is 
        begin q <= a and b after TPD;
end;
entity AndT_Test_1 is end;
architecture Netlist_1 of AndT_Test_1 is 
        component MyAnd 
                port (a, b : BIT; q : out BIT);
        end component;
        signal a1, b1, q1 : BIT := ’1’;
        begin 
                And1 : MyAnd port map (a1, b1, q1);
end Netlist_1;
configuration Simplest_1 of AndT_Test_1 is use work.all;
        for Netlist_1 for And1 : MyAnd 
                use entity AndT(Behave) generic map (2 ns); 
        end for; end for;
end Simplest_1;

The configuration declaration, Simplest_1, changes the default delay (equal to 1 ns, declared as a
default expression in the entity) to 2 ns. Techniques based on this method are useful in ASIC design.
Prelayout simulation uses the default timing values. Back-annotation alters the delay in the configuration
for postlayout simulation. When we change the delay we only need to reanalyze the configuration, not
the rest of the ASIC model.

There was initially no standard in VHDL for how timing generics should be used, and the lack of a
standard was a major problem for ASIC designers. The IEEE 1076.4 VITAL standard addresses this
problem (see Section 13.5.5).

1. There are additional rules for interface objects that are signals (ports)--see Tables 10.11 and 10.12.

2. A signal of mode inout can be updated by any number of sources [VHDL 87LRM4.3.3,
93LRM4.3.2].

3. A signal of mode buffer can be updated by at most one source [VHDL LRM1.1.1.2].

10.8  Type Declarations

In some programming languages you must declare objects to be integer, real, Boolean, and so on. VHDL
(and ADA, the DoD programming language to which VHDL is related) goes further: You must declare
the type of an object, and there are strict rules on mixing objects of different types. We say VHDL is
strongly typed. For example, you can use one type for temperatures in Centigrade and a different type
for Fahrenheit, even though both types are real numbers. If you try to add a temperature in Centigrade to
a temperature in Fahrenheit, VHDL catches your error and tells you that you have a type mismatch.



This is the formal (expanded) BNF definition of a type declaration:

type_declaration ::= 
        type identifier ;
| type identifier is 
(identifier|’graphic_character’ {, identifier|’graphic_character’}) ;
| range_constraint ;            | physical_type_definition ;
| record_type_definition ;      | access subtype_indication ;
| file of type_name ;           | file of subtype_name ;
| array index_constraint of element_subtype_indication ;
| array 
        (type_name|subtype_name range <>
                {, type_name|subtype_name range <>}) of
                element_subtype_indication ;

There are four type classes in VHDL [VHDL LRM3]: scalar types, composite types, access types, and
file types. The scalar types are: integer type, floating-point type, physical type, and enumeration type.
Integer and enumeration types are discrete types. Integer, floating-point, and physical types are numeric
types. The range of an integer is implementation dependent but is guaranteed to include -2147483647 to
+2147483647. Notice the integer range is symmetric and equal to -(231- 1) to (231- 1). Floating-point
size is implementation dependent, but the range includes the bounds -1.0E38 and +1.0E38, and must
include a minimum of six decimal digits of precision. Physical types correspond to time, voltage,
current, and so on and have dimensions--a unit of measure (seconds, for example). Access types are
pointers, useful in abstract data structures, but less so in ASIC design. File types are used for file I/O.

You may also declare a subset of an existing type, known as a subtype, in a subtype declaration. We
shall discuss the different treatment of types and subtypes in expressions in Section 10.12.

Here are some examples of scalar type [VHDL LRM4.1] and subtype declarations [VHDL LRM4.2]:

entity Declaration_1 is end; architecture Behave of Declaration_1 is
type F is range 32 to 212; -- Integer type, ascending range.
type C is range 0 to 100; -- Range 0 to 100 is the range constraint.
subtype G is INTEGER range 9 to 0; -- Base type INTEGER, descending.
-- This is illegal: type Bad100 is INTEGER range 0 to 100; 
-- don’t use INTEGER in declaration of type (but OK in subtype).
type Rainbow is (R, O, Y, G, B, I, V); -- An enumeration type.
-- Enumeration types always have an ascending range.
type MVL4 is (’X’, ’0’, ’1’, ’Z’); 
-- Note that ’X’ and ’x’ are different character literals.
-- The default initial value is MVL4’LEFT = ’X’.
-- We say ’0’ and ’1’ (already enumeration literals
-- for predefined type BIT) are overloaded.
-- Illegal enumeration type: type Bad4 is ("X", "0", "1", "Z"); 
-- Enumeration literals must be character literals or identifiers.
begin end;

The most common composite type is the array type [VHDL LRM3.2.1]. The following examples
illustrate the semantics of array declarations:

entity Arrays_1 is end; architecture Behave of Arrays_1 is
type Word is array (0 to 31) of BIT; -- a 32-bit array, ascending
type Byte is array (NATURAL range 7 downto 0) of BIT; -- descending
type BigBit is array (NATURAL range <>) of BIT;
-- We call <> a box, it means the range is undefined for now.
-- We call BigBit an unconstrained array.



-- This is OK, we constrain the range of an object that uses
-- type BigBit when we declare the object, like this:
subtype Nibble is BigBit(3 downto 0);
type T1 is array (POSITIVE range 1 to 32) of BIT;
-- T1, a constrained array declaration, is equivalent to a type T2 
-- with the following three declarations:
subtype index_subtype is POSITIVE range 1 to 32;
type array_type is array (index_subtype range <>) of BIT;
subtype T2 is array_type (index_subtype);
-- We refer to index_subtype and array_type as being
-- anonymous subtypes of T1 (since they don’t really exist).
begin end;

You can assign values to an array using aggregate notation [VHDL LRM7.3.2]:

entity Aggregate_1 is end; architecture Behave of Aggregate_1 is
type D is array (0 to 3) of BIT; type Mask is array (1 to 2) of BIT;
signal MyData : D := (’0’, others => ’1’); -- positional aggregate 
signal MyMask : Mask := (2 => ’0’, 1 => ’1’); -- named aggregate
begin end;

The other composite type is the record type that groups elements together:

entity Record_2 is end; architecture Behave of Record_2 is 
type Complex is record real : INTEGER; imag : INTEGER; end record;
signal s1 : Complex := (0, others => 1); signal s2: Complex;
begin s2 <= (imag => 2, real => 1); end;

10.9  Other Declarations

A declaration is one of the following [VHDL LRM4]:

declaration ::=
  type_declaration      | subtype_declaration   | object_declaration
| interface_declaration         | alias_declaration     | attribute_declaration
| component_declaration         | entity_declaration
| configuration_declaration  | subprogram_declaration
| package_declaration
| group_template_declaration | group_declaration 

I discussed entity, configuration, component, package, interface, type, and subtype declarations in
Sections 10.5-10.8. Next I shall discuss the other types of declarations (except for groups or group
templates [VHDL 93LRM4.6-4.7], new to VHDL-93, that are not often used in ASIC design).

10.9.1  Object Declarations

There are four object classes in VHDL: constant, variable, signal, and file [VHDL LRM 4.3.1.1-4.3.1.3].
You use a constant declaration, signal declaration, variable declaration, or file declaration together with
a type. Signals can only be declared in the declarative region (before the first begin ) of an architecture
or block, or in a package (not in a package body). Variables can only be declared in the declarative



region of a process or subprogram (before the first begin ). You can think of signals as representing real
wires in hardware. You can think of variables as memory locations in the computer. Variables are more
efficient than signals because they require less overhead.

You may assign an (explicit) initial value when you declare a type. If you do not provide initial values,
the (implicit) default initial value of a type or subtype T is T’LEFT (the leftmost item in the range of the
type). For example:

entity Initial_1 is end; architecture Behave of Initial_1 is
type Fahrenheit is range 32 to 212;                                                                                                                             -- Default initial value is 32.
type Rainbow is (R, O, Y, G, B, I, V);                                                                                                                          -- Default initial value is R.
type MVL4 is (’X’, ’0’, ’1’, ’Z’);                                                                                                                              -- MVL4’LEFT = ’X’.
begin end;

The details of initialization and assignment of initial values are important--it is difficult to implement the
assignment of initial values in hardware--instead it is better to mimic the hardware and use explicit reset
signals.

Here are the formal definitions of constant and signal declarations:

constant_declaration ::= constant
identifier {, identifier}:subtype_indication [:= expression] ;
signal_declaration ::= signal
identifier {, identifier}:subtype_indication [register|bus] [:=expression];

I shall explain the use of signals of kind register or bus in Section 10.13.1. Signal declarations are
explicit signal declarations (ports declared in an interface declaration are implicit signal declarations).
Here is an example that uses a constant and several signal declarations:

entity Constant_2 is end; 
library IEEE; use IEEE.STD_LOGIC_1164.all;
architecture Behave of Constant_2 is
constant Pi : REAL := 3.14159;                                                                                                                                          -- A constant declaration.
signal B : BOOLEAN; signal s1, s2: BIT; 
signal sum : INTEGER range 0 to 15;                                                                                                                                             -- Not a new type.
signal SmallBus : BIT_VECTOR (15 downto 0);                                                                                                                                             -- 16-bit bus.
signal GBus : STD_LOGIC_VECTOR (31 downto 0); bus; -- A guarded signal.
begin end;

Here is the formal definition of a variable declaration:

variable_declaration ::= [shared] variable
identifier {, identifier}:subtype_indication [:= expression] ;

A shared variable can be used to model a varying quantity that is common across several parts of a
model, temperature, for example, but shared variables are rarely used in ASIC design. The following
examples show that variable declarations belong inside a process statement, after the keyword process
and before the first appearance of the keyword begin inside a process:

library IEEE; use IEEE.STD_LOGIC_1164.all; entity Variables_1 is end;
architecture Behave of Variables_1 is begin process
        variable i : INTEGER range 1 to 10 := 10; -- Initial value = 10.
        variable v : STD_LOGIC_VECTOR (0 to 31) := (others => ’0’); 
        begin wait; end process; -- The wait stops an endless cycle.
end;



10.9.2  Subprogram Declarations

VHDL code that you use several times can be declared and specified as subprograms (functions or
procedures) [VHDL LRM2.1]. A function is a form of expression, may only use parameters of mode in
, and may not contain delays or sequence events during simulation (no wait statements, for example).
Functions are useful to model combinational logic. A procedure is a form of statement and allows you to
control the scheduling of simulation events without incurring the overhead of defining several separate
design entities. There are thus two forms of subprogram declaration: a function declaration or a
procedure declaration.

subprogram_declaration ::= subprogram_specification ; ::=
 procedure 
        identifier|string_literal [(parameter_interface_list)]
| [pure|impure] function
        identifier|string_literal [(parameter_interface_list)]
return type_name|subtype_name;

Here are a function and a procedure declaration that illustrate the difference:

function add(a, b, c : BIT_VECTOR(3 downto 0)) return BIT_VECTOR is
-- A function declaration, a function can’t modify a, b, or c.
procedure Is_A_Eq_B (signal A, B : BIT; signal Y : out BIT);
-- A procedure declaration, a procedure can change Y.

Parameter names in subprogram declarations are called formal parameters (or formals). During a call to
a subprogram, known as subprogram invocation, the passed values are actual parameters (or actuals). An
impure function, such as the function now or a function that writes to or reads from a file, may return
different values each time it is called (even with the same actuals). A pure function (the default) returns
the same value if it is given the same actuals. You may call subprograms recursively. Table 10.13 shows
the properties of subprogram parameters.



TABLE 10.13    Properties of subprogram parameters.

Example subprogram declarations:

 function my_function(Ff) return BIT is -- Formal function parameter, Ff.
procedure my_procedure(Fp);           -- Formal procedure parameter, Fp.

Example subprogram calls:

 my_result := my_function(Af); -- Calling a function with an actual parameter, Af.
MY_LABEL:my_procedure(Ap);   -- Using a procedure with an actual parameter, Ap.

Mode of Ff or Fp (formals) in out inout No mode

Permissible classes for Af

(function actual parameter)

constant (default)

signal
Not allowed Not allowed file

Permissible classes for Ap

(procedure actual parameter)

 

constant (default)

variable

signal

constant

variable (default)

signal

constant 

variable (default)

signal

file

Can you read attributes of

Ff or Fp (formals)?

Yes, except:

 ’STABLE

 ’QUIET

 ’DELAYED

 ’TRANSACTION

of a signal

Yes, except:

 ’STABLE ’QUIET

 ’DELAYED

 ’TRANSACTION

 ’EVENT ’ACTIVE

 ’LAST_EVENT

 ’LAST_ACTIVE

 ’LAST_VALUE

of a signal

Yes, except:

 ’STABLE

 ’QUIET

 ’DELAYED

 ’TRANSACTION

of a signal

 

A subprogram declaration is optional, but a subprogram specification must be included in the
subprogram body (and must be identical in syntax to the subprogram declaration--see BNF [10.19]):

subprogram_body ::=
        subprogram_specification is
        {subprogram_declaration|subprogram_body
        |type_declaration|subtype_declaration
        |constant_declaration|variable_declaration|file_declaration
        |alias_declaration|attribute_declaration|attribute_specification
        |use_clause|group_template_declaration|group_declaration}
        begin



                {sequential_statement}
        end [procedure|function] [identifier|string_literal] ;

You can include a subprogram declaration or subprogram body in a package or package body (see
Section 10.6) or in the declarative region of an entity or process statement. The following is an
example of a function declaration and its body:

function subset0(sout0 : in BIT) return BIT_VECTOR -- declaration
-- Declaration can be separate from the body.
function subset0(sout0 : in BIT) return BIT_VECTOR is -- body
variable y : BIT_VECTOR(2 downto 0);
begin 
if (sout0 = ’0’) then y := "000"; else y := "100"; end if;
return result;
end;
procedure clockGen (clk : out BIT)                                                                                                                                                                                                                              -- Declaration
procedure clockGen (clk : out BIT) is                                                                                                                                                                   -- Specification
begin -- Careful this process runs forever: 
        process begin wait for 10 ns; clk <= not clk; end process;
end;

One reason for having the optional (and seemingly redundant) subprogram declaration is to allow
companies to show the subprogram declarations (to document the interface) in a package declaration,
but to hide the subprogram bodies (the actual code) in the package body. If a separate subprogram
declaration is present, it must conform to the specification in the subprogram body [VHDL 93LRM2.7].
This means the specification and declaration must be almost identical; the safest method is to copy and
paste. If you define common procedures and functions in packages (instead of in each entity or
architecture, for example), it will be easier to reuse subprograms. In order to make a subprogram
included in a package body visible outside the package, you must declare the subprogram in the package
declaration (otherwise the subprogram is private).

You may call a function from any expression, as follows:

entity F_1 is port (s : out BIT_VECTOR(3 downto 0) := "0000"); end;
architecture Behave of F_1 is begin process
function add(a, b, c : BIT_VECTOR(3 downto 0)) return BIT_VECTOR is
begin return a xor b xor c; end;
begin s <= add("0001", "0010", "1000"); wait; end process; end;
package And_Pkg is 
        procedure V_And(a, b : BIT; signal c : out BIT); 
        function V_And(a, b : BIT) return BIT;
end;
package body And_Pkg is 
        procedure V_And(a,b : BIT; signal c : out BIT) is 
                begin c <= a and b; end;
        function V_And(a,b : BIT) return BIT is 
                begin return a and b; end;
end And_Pkg;
entity F_2 is port (s: out BIT := ’0’); end;
use work.And_Pkg.all; -- use package already analyzed
architecture Behave of F_2 is begin process begin 
s <= V_And(’1’, ’1’); wait; end process; end;

I shall discuss the two different ways to call a procedure in Sections 10.10.4 and 10.13.3.

10.9.3  Alias and Attribute Declarations



An alias declaration [VHDL 87LRM4.3.4, 93LRM4.3.3] names parts of a type:

alias_declaration ::= 
alias 
        identifier|character_literal|operator_symbol  [ :subtype_indication]
        is name [signature] ;

(the subtype indication is required in VHDL-87, but not in VHDL-93).

Here is an example of alias declarations for parts of a floating-point number:

entity Alias_1 is end; architecture Behave of Alias_1 is
begin process variable Nmbr: BIT_VECTOR (31 downto 0);
-- alias declarations to split Nmbr into 3 pieces :
alias Sign : BIT is Nmbr(31);
alias Mantissa : BIT_VECTOR (23 downto 0) is Nmbr (30 downto 7);
alias Exponent : BIT_VECTOR ( 6 downto 0) is Nmbr ( 6 downto 0);
begin wait; end process; end; -- the wait prevents an endless cycle

An attribute declaration [VHDL LRM4.4] defines attribute properties:

attribute_declaration ::=
 attribute identifier:type_name ; | attribute identifier:subtype_name ;

Here is an example:

entity Attribute_1 is end; architecture Behave of Attribute_1 is
begin process type COORD is record X, Y : INTEGER; end record; 
attribute LOCATION : COORD; -- the attribute declaration
begin wait ; -- the wait prevents an endless cycle
end process; end;

You define the attribute properties in an attribute specification (the following example specifies an
attribute of a component label). You probably will not need to use your own attributes very much in
ASIC design.

attribute LOCATION of adder1 : label is (10,15);

You can then refer to your attribute as follows:

positionOfComponent := adder1’LOCATION;

10.9.4  Predefined Attributes

The predefined attributes for scalar and array types in VHDL-93 are shown in Table 10.14 [VHDL
93LRM14.1]. There are two attributes, ’STRUCTURE and ’BEHAVIOR , that are present in VHDL-87, but
removed in VHDL-93. Both of these attributes apply to architecture bodies. The attribute name
A’BEHAVIOR is TRUE if the architecture A does not contain component instantiations. The attribute name
A’STRUCTURE is TRUE if the architecture A contains only passive processes (those with no assignments to
signals) and component instantiations. These two attributes were not widely used. The attributes shown
in Table 10.14, however, are used extensively to create packages and functions for type conversion and
overloading operators, but should not be needed by an ASIC designer. Many of the attributes do not
correspond to "real" hardware and cannot be implemented by a synthesis tool.



TABLE 10.14    Predefined attributes for scalar and array types.

Attribute Kind
1

Prefix 

T, A, E 2

Parameter X
or N 3

Result type
3 Result

T’BASE T any  base(T) base(T), use only with
other attribute

T’LEFT V scalar  T Left bound of T

T’RIGHT V scalar  T Right bound of T

T’HIGH V scalar  T Upper bound of T

T’LOW V scalar  T Lower bound of T

T’ASCENDING V scalar  BOOLEAN
True if range of T is
ascending 4

T’IMAGE(X) F scalar base(T) STRING
String representation of X
in T 4

T’VALUE(X) F scalar STRING base(T) Value in T with
representation X 4

T’POS(X) F discrete base(T) UI Position number of X in T
(starts at 0)

T’VAL(X) F discrete UI base(T) Value of position X in T

T’SUCC(X) F discrete base(T) base(T) Value of position X in T
plus one

T’PRED(X) F discrete base(T) base(T) Value of position X in T
minus one

T’LEFTOF(X) F discrete base(T) base(T) Value to the left of X in T

T’RIGHTOF(X) F discrete base(T) base(T) Value to the right of X in
T

A’LEFT[(N)] F array UI T(Result) Left bound of index N of
array A

A’RIGHT[(N)] F array UI T(Result) Right bound of index N
of array A

A’HIGH[(N)] F array UI T(Result) Upper bound of index N
of array A

A’LOW[(N)] F array UI T(Result) Lower bound of index N
of array A

A’RANGE[(N)] R array UI T(Result) Range A’LEFT(N) to
A’RIGHT(N) 5

A’REVERSE_RANGE[(N)] R array UI T(Result) Opposite range to
A’RANGE[(N)]



A’LENGTH[(N)] V array UI UI Number of values in
index N of array A

A’ASCENDING[(N)] V array UI BOOLEAN
True if index N of A is
ascending 4

E’SIMPLE_NAME V name  STRING Simple name of E 4

E’INSTANCE_NAME V name  STRING
Path includes instantiated
entities 4

E’PATH_NAME V name  STRING
Path excludes instantiated
entities 4

The attribute ’LEFT is important because it determines the default initial value of a type. For example,
the default initial value for type BIT is BIT’LEFT , which is ’0’ . The predefined attributes of signals
are listed in Table 10.15. The most important signal attribute is ’EVENT , which is frequently used to
detect a clock edge. Notice that Clock’EVENT , for example, is a function that returns a value of type
BOOLEAN , whereas the otherwise equivalent not(Clock’STABLE) , is a signal. The difference is subtle
but important when these attributes are used in the wait statement that treats signals and values
differently. 

TABLE 10.15    Predefined attributes for signals.

Attribute Kind 6 Parameter T 7 Result type 8 Result/restrictions

S’DELAYED [(T)] S TIME base(S) S delayed by time T

S’STABLE [(T)] S TIME BOOLEAN TRUE if no event on S for time T

S’QUIET [(T)] S TIME BOOLEAN TRUE if S is quiet for time T

S’TRANSACTION S  BIT Toggles each cycle if S becomes active

S’EVENT F  BOOLEAN TRUE when event occurs on S

S’ACTIVE F  BOOLEAN TRUE if S is active

S’LAST_EVENT F  TIME Elapsed time since the last event on S

S’LAST_ACTIVE F  TIME Elapsed time since S was active

S’LAST_VALUE F  base(S) Previous value of S, before last event 9

S’DRIVING F  BOOLEAN TRUE if every element of S is driven 10

S’DRIVING_VALUE F  base(S) Value of the driver for S in the current
process 10

1. T = Type, F = Function, V = Value, R = Range.

2. any = any type or subtype, scalar = scalar type or subtype, discrete = discrete or physical type or
subtype, name = entity name = identifier, character literal, or operator symbol.

3. base(T) = base type of T, T = type of T, UI = universal_integer, T(Result) = type of object described



in result column.

4. Only available in VHDL-93. For ’ASCENDING all enumeration types are ascending.

5. Or reverse for descending ranges.

6. F = function, S = signal.

7. Time T >= 0 ns. The default, if T is not present, is T = 0 ns.

8. base(S) = base type of S.

9. VHDL-93 returns last value of each signal in array separately as an aggregate, VHDL-87 returns the
last value of the composite signal.

10. VHDL-93 only.

10.10 Sequential Statements

A sequential statement [VHDL LRM8] is defined as follows:

sequential_statement ::=
  wait_statement        | assertion_statement
| signal_assignment_statement
| variable_assignment_statement                                                                                                                         | procedure_call_statement
| if_statement  | case_statement | loop_statement
| next_statement        | exit_statement
| return_statement      | null_statement | report_statement

Sequential statements may only appear in processes and subprograms. In the following sections I shall
describe each of these different types of sequential statements in turn.

10.10.1  Wait Statement

The wait statement is central to VHDL, here are the BNF definitions [VHDL 93LRM8.1]:

wait_statement ::= [label:] wait [sensitivity_clause] 
        [condition_clause] [timeout_clause] ;
sensitivity_clause ::= on sensitivity_list
sensitivity_list ::= signal_name { , signal_name }
condition_clause ::= until condition
condition ::= boolean_expression
timeout_clause ::= for time_expression

A wait statement suspends (stops) a process or procedure (you cannot use a wait statement in a
function). The wait statement may be made sensitive to events (changes) on static signals (the value of
the signal must be known at analysis time) that appear in the sensitivity list after the keyword on . These



signals form the sensitivity set of a wait statement. The process will resume (restart) when an event
occurs on any signal (and only signals) in the sensitivity set. 

A wait statement may also contain a condition to be met before the process resumes. If there is no
sensitivity clause (there is no keyword on ) the sensitivity set is made from signals (and only signals)
from the condition clause that appears after the keyword until (the rules are quite complicated [VHDL
93LRM8.1]). 

Finally a wait statement may also contain a timeout (following the keyword for ) after which the
process will resume. Here is the expanded BNF definition, which makes the structure of the wait
statement easier to see (but we lose the definitions of the clauses and the sensitivity list):

wait_statement ::= [label:]

 wait 
        [on signal_name {, signal_name}] 
        [until boolean_expression]
        [for time_expression] ;

For example, the statement, wait on light , makes you wait until a traffic light changes (any change).
The statement, wait until light = green , makes you wait (even at a green light) until the traffic
signal changes to green. The statement,

if light = (red or yellow) then wait 
until light = green; 
end 
if;

accurately describes the basic rules at a traffic intersection. 

The most common use of the wait statement is to describe synchronous logic, as in the following model
of a D flip-flop:

entity DFF is port (CLK, D : BIT; Q : out BIT); end;
architecture Behave of DFF is
process begin wait 
until C
lk = ’1’; Q <= D ; end process;
end;

Notice that the statement in line 3 above, wait until C lk = ’1’, is equivalent to wait on Clk until
C lk = ’1’, and detects a clock edge and not the clock level. Here are some more complex examples of
the use of the wait statement:

entity Wait_1 is port (Clk, s1, s2 :in BIT); end; 
architecture Behave of Wait_1 is
signal x : BIT_VECTOR (0 to 15);
        begin process variable v : BIT; begin 
        wait;    -- Wait forever, stops simulation.
        wait on s1 until s2 = ’1’; -- Legal, but s1, s2 are signals so
        -- s1 is in sensitivity list, and s2 is not in the sensitivity set.
        -- Sensitivity set is s1 and process will not resume at event on s2.
        wait on s1, s2; -- resumes at event on signal s1 or s2.
        wait on s1 for 10 ns; -- resumes at event on s1 or after 10 ns.
        wait on x; -- resumes when any element of array x has an event.



-- wait on x(1 to v); -- Illegal, nonstatic name, since v is a variable.
end process;
end;
entity Wait_2 is port (Clk, s1, s2:in BIT); end;
architecture Behave of Wait_2 is
        begin process variable v : BIT; begin 
        wait on Clk; -- resumes when Clk has an event: rising or falling.
        wait until Clk = ’1’; -- resumes on rising edge.
        wait on Clk until Clk = ’1’; -- equivalent to the last statement.
        wait on Clk until v = ’1’; 
        -- The above is legal, but v is a variable so
        -- Clk is in sensitivity list, v is not in the sensitivity set.
        -- Sensitivity set is Clk and process will not resume at event on v.
        wait on Clk until s1 = ’1’; 
        -- The above is legal, but s1 is a signal so
        -- Clk is in sensitivity list, s1 is not in the sensitivity set.
        -- Sensitivity set is Clk, process will not resume at event on s1.
        end process;
end;

You may only use interface signals that may be read (port modes in , inout , and buffer --see
Section 10.7) in the sensitivity list of a wait statement.

10.10.2  Assertion and Report Statements

You can use an assertion statement to conditionally issue warnings. The report statement (VHDL-93
only) prints an expression and is useful for debugging.

assertion_statement ::= [label:] assert
boolean_expression [report expression] [severity expression] ;
report_statement 
::= [label:] report expression [severity expression] ;

Here is an example of an assertion statement:

entity Assert_1 is port (I:INTEGER:=0); end;
architecture Behave of Assert_1 is
        begin process begin 
        assert (I > 0) report "I is negative or zero"; wait;
        end process;
end;

The expression after the keyword report must be of type STRING (the default is "Assertion
violation" for the assertion statement), and the expression after the keyword severity must be of
type SEVERITY_LEVEL (default ERROR for the assertion statement, and NOTE for the report statement)
defined in the STANDARD package. The assertion statement prints if the assertion condition (after the
keyword assert ) is FALSE . Simulation normally halts for severity of ERROR or FAILURE (you can
normally control this threshold in the simulator).

10.10.3  Assignment Statements

There are two sorts of VHDL assignment statements: one for signals and one for variables [VHDL
93LRM8.4-8.5]. The difference is in the timing of the update of the LHS. A variable assignment
statement is the closest equivalent to the assignment statement in a computer programming language.
Variable assignment statements are always sequential statements and the LHS of a variable assignment



statement is always updated immediately. Here is the definition and an example:

variable_assignment_statement ::= 
         [label:] name|aggregate := expression ;
entity Var_Assignment is end;
architecture Behave of Var_Assignment is
        signal s1 : INTEGER := 0; 
        begin process variable v1,v2 : INTEGER := 0; begin 
        assert (v1/=0) report "v1 is 0" severity note ; -- this prints
        v1 := v1 + 1; -- after this statement v1 is 1
        assert (v1=0) report "v1 isn’t 0" severity note ; -- this prints
        v2 := v2 + s1; -- signal and variable types must match
        wait;
        end process;
end;

This is the output from Cadence Leapfrog for the preceding example:

ASSERT/NOTE (time 0 FS) from :$PROCESS_000 (design unit WORK.VAR_ASSIGNMENT:BEHAVE) v1 is 0
ASSERT/NOTE (time 0 FS) from :$PROCESS_000 (design unit WORK.VAR_ASSIGNMENT:BEHAVE) v1 isn’t 0

A signal assignment statement schedules a future assignment to a signal:

signal_assignment_statement::= 
        [label:] target <=
        [transport | [ reject time_expression ] inertial ] waveform ;

The following example shows that, even with no delay, a signal is updated at the end of a simulation
cycle after all the other assignments have been scheduled, just before simulation time is advanced:

entity Sig_Assignment_1 is end; 
architecture Behave of Sig_Assignment_1 is
        signal s1,s2,s3 : INTEGER := 0; 
        begin process variable v1 : INTEGER := 1; begin 
        assert (s1 /= 0) report "s1 is 0" severity note ; -- this prints.
        s1 <= s1 + 1; -- after this statement s1 is still 0.
        assert (s1 /= 0) report "s1 still 0" severity note ; -- this prints.
        wait;
        end process;
end;
ASSERT/NOTE (time 0 FS) from :$PROCESS_000 (design unit WORK.SIG_ASSIGNMENT_1:BEHAVE) s1 is 0
ASSERT/NOTE (time 0 FS) from :$PROCESS_000 (design unit WORK.SIG_ASSIGNMENT_1:BEHAVE) s1 still 0

Here is an another example to illustrate how time is handled:

entity Sig_Assignment_2 is end; 
architecture Behave of Sig_Assignment_2 is
        signal s1, s2, s3 : INTEGER := 0; 
        begin process variable v1 : INTEGER := 1; begin 
        -- s1, s2, s3 are initially 0; now consider the following:
        s1 <= 1 ; -- schedules updates to s1 at end of 0 ns cycle.
        s2 <= s1; -- s2 is 0, not 1.
        wait for 1 ns;
        s3 <= s1; -- now s3 will be 1 at 1 ns.
        wait;
        end process;
end;



The Compass simulator produces the following trace file for this example:

      Time(fs) + Cycle            s1           s2           s3
----------------------  ------------ ------------ ------------
                  0+ 0:            0            0            0
                  0+ 1: *          1 *          0            0
...
            1000000+ 1:            1            0 *          1

Time is indicated in femtoseconds for each simulation cycle plus the number of delta cycles (we call this
delta time, measured in units of delta) needed to calculate all transactions on signals. A transaction
consists of a new value for a signal (which may be the same as the old value) and the time delay for the
value to take effect. An asterisk ’*’ before a value in the preceding trace indicates that a transaction has
occurred and the corresponding signal updated at that time. A transaction that does result in a change in
value is an event. In the preceding simulation trace for Sig_Assignment_2:Behave

At 0 ns + 0 delta: all signals are 0 . 

At 0 ns + 1 delta: s1 is updated to 1 , s2 is updated to 0 (not to 1 ). 

At 1 ns + 1 delta: s3 is updated to a 1 . 

The following example shows the behavior of the different delay models: transport and inertial (the
default):

entity Transport_1 is end; 
architecture Behave of Transport_1 is
signal s1, SLOW, FAST, WIRE : BIT := ’0’; 
        begin process begin 
        s1 <= ’1’ after 1 ns, ’0’ after 2 ns, ’1’ after 3 ns ;
        -- schedules s1 to be ’1’ at t+1 ns, ’0’ at t+2 ns,’1’ at t+3 ns
        wait; end process;
-- inertial delay: SLOW rejects pulsewidths less than 5ns:
process (s1) begin SLOW <= s1 after 5 ns ; end process;
-- inertial delay: FAST rejects pulsewidths less than 0.5ns:
process (s1) begin FAST <= s1 after 0.5 ns ; end process;
-- transport delay: WIRE passes all pulsewidths...
process (s1) begin WIRE <= transport s1 after 5 ns ; end process;
end;

Here is the trace file from the Compass simulator:

  
      Time(fs) + Cycle    s1 slow fast wire
----------------------  ---- ---- ---- ----
                  0+ 0:  ’0’  ’0’  ’0’  ’0’
             500000+ 0:  ’0’  ’0’ *’0’  ’0’
            1000000+ 0: *’1’  ’0’  ’0’  ’0’
            1500000+ 0:  ’1’  ’0’ *’1’  ’0’
            2000000+ 0: *’0’  ’0’  ’1’  ’0’
            2500000+ 0:  ’0’  ’0’ *’0’  ’0’
            3000000+ 0: *’1’  ’0’  ’0’  ’0’
            3500000+ 0:  ’1’  ’0’ *’1’  ’0’
            5000000+ 0:  ’1’  ’0’  ’1’ *’0’
            6000000+ 0:  ’1’  ’0’  ’1’ *’1’
            7000000+ 0:  ’1’  ’0’  ’1’ *’0’



            8000000+ 0:  ’1’ *’1’  ’1’ *’1’

Inertial delay mimics the behavior of real logic gates, whereas transport delay more closely models the
behavior of wires. In VHDL-93 you can also add a separate pulse rejection limit for the inertial delay
model as in the following example:

process (s1) begin RJCT <= reject 2 ns s1 after 5 ns ; end process;

10.10.4  Procedure Call

A procedure call in VHDL corresponds to calling a subroutine in a conventional programming language
[VHDL LRM8.6]. The parameters in a procedure call statement are the actual procedure parameters (or
actuals); the parameters in the procedure definition are the formal procedure parameters (or formals).
The two are linked using an association list, which may use either positional or named association
(association works just as it does for ports--see Section 10.7.1):

procedure_call_statement ::=
         [label:] procedure_name [(parameter_association_list)];

Here is an example:

package And_Pkg is 
        procedure V_And(a, b : BIT; signal c : out BIT); 
        function V_And(a, b : BIT) return BIT;
end;
package body And_Pkg is 
        procedure V_And(a, b : BIT; signal c: out BIT) is 
                begin c <= a and b; end;
        function V_And(a, b: BIT) return BIT is 
                begin return a and b; end;
end And_Pkg;
use work.And_Pkg.all; entity Proc_Call_1 is end; 
architecture Behave of Proc_Call_1 is signal A, B, Y: BIT := ’0’;
        begin process begin V_And (A, B, Y); wait; end process;
end;

Table 10.13 on page 416 explains the rules for formal procedure parameters. There is one other way to
call procedures, which we shall cover in Section 10.13.3.

10.10.5  If Statement

An if statement evaluates one or more Boolean expressions and conditionally executes a corresponding
sequence of statements [VHDL LRM8.7].

if_statement ::=
         [if_label:] if boolean_expression then {sequential_statement}
                {elsif boolean_expression then {sequential_statement}}
                [else {sequential_statement}]
        end if [if_label];

The simplest form of an if statement is thus:

if boolean_expression then {sequential_statement} end if;



Here are some examples of the if statement:

entity If_Then_Else_1 is end; 
architecture Behave of If_Then_Else_1 is signal a, b, c: BIT :=’1’; 
        begin process begin
                if c = ’1’ then c <= a ; else c <= b; end if; wait;
        end process;
end;
entity If_Then_1 is end; 
architecture Behave of If_Then_1 is signal A, B, Y : BIT :=’1’;
        begin process begin
                if A = B then Y <= A; end if; wait;
        end process;
end;

10.10.6  Case Statement

A case statement [VHDL LRM8.8] is a multiway decision statement that selects a sequence of
statements by matching an expression with a list of (locally static [VHDL LRM7.4.1]) choices.

case_statement ::=
[case_label:] case expression is
         when choice {| choice} => {sequential_statement}
        {when choice {| choice} => {sequential_statement}}
end case [case_label];

Case statements are useful to model state machines. Here is an example of a Mealy state machine with
an asynchronous reset:

library IEEE; use IEEE.STD_LOGIC_1164.all; 
entity sm_mealy is
        port (reset, clock, i1, i2 : STD_LOGIC; o1, o2 : out STD_LOGIC);
end sm_mealy;
architecture Behave of sm_mealy is 
type STATES is (s0, s1, s2, s3); signal current, new : STATES;
begin
synchronous : process (clock, reset) begin
        if To_X01(reset) = ’0’ then current <= s0;
        elsif rising_edge(clock) then current <= new; end if;
end process;
combinational : process (current, i1, i2) begin 
case current is
        when s0 =>
                if To_X01(i1) = ’1’ then o2 <=’0’; o1 <=’0’; new <= s2;
                else o2 <= ’1’; o1 <= ’1’; new <= s1; end if;
        when s1 => 
                if To_X01(i2) = ’1’ then o2 <=’1’; o1 <=’0’; new <= s1;
                else o2 <=’0’; o1 <=’1’; new <= s3; end if;
        when s2 => 
                if To_X01(i2) = ’1’ then o2 <=’0’; o1 <=’1’; new <= s2; 
                else o2 <= ’1’; o1 <= ’0’; new <= s0; end if; 
        when s3 => o2 <= ’0’; o1 <= ’0’; new <= s0; 
        when others => o2 <= ’0’; o1 <= ’0’; new <= s0;
end case;
end process;
end Behave;

Each possible value of the case expression must be present once, and once only, in the list of choices (or



arms) of the case statement (the list must be exhaustive). You can use ’|’ (that means ’or’) or ’to’ to
denote a range in the expression for choice . You may also use the keyword others as the last, default
choice (even if the list is already exhaustive, as in the preceding example).

10.10.7  Other Sequential Control Statements

A loop statement repeats execution of a series of sequential statements [VHDL LRM8.9]:

loop_statement ::=
[loop_label:] 
[while boolean_expression|for identifier in discrete_range]
loop
        {sequential_statement}
end loop [loop_label];

If the loop variable (after the keyword for ) is used, it is only visible inside the loop. A while loop
evaluates the Boolean expression before each execution of the sequence of statements; if the expression
is TRUE , the statements are executed. In a for loop the sequence of statements is executed once for each
value of the discrete range.

package And_Pkg is function V_And(a, b : BIT) return BIT; end;
package body And_Pkg is function V_And(a, b : BIT) return BIT is 
         begin return a and b; end; end And_Pkg;
entity Loop_1 is port (x, y : in BIT := ’1’; s : out BIT := ’0’); end;
use work.And_Pkg.all; 
architecture Behave of Loop_1 is 
        begin loop
                s <= V_And(x, y); wait on x, y; 
        end loop; 
end;

The next statement [VHDL LRM8.10] forces completion of the current iteration of a loop (the
containing loop unless another loop label is specified). Completion is forced if the condition following
the keyword then is TRUE (or if there is no condition).

next_statement ::=
[label:] next [loop_label] [when boolean_expression];

An exit statement [VHDL LRM8.11] forces an exit from a loop.

exit_statement ::= 
        [label:] exit [loop_label] [when condition] ;

As an example:

loop wait on Clk; exit when Clk = ’0’; end loop;
-- equivalent to: wait until Clk = ’0’;

The return statement [VHDL LRM8.12] completes execution of a procedure or function.

return_statement ::= [label:] return [expression];

A null statement [VHDL LRM8.13] does nothing (but is useful in a case statement where all choices
must be covered, but for some of the choices you do not want to do anything).



null_statement ::= [label:] null;

10.11  Operators

Table 10.16 shows the predefined VHDL operators, listed by their (increasing) order of precedence
[VHDL 93LRM7.2]. The shift operators and the xnor operator were added in VHDL-93.   

TABLE 10.16    VHDL predefined operators (listed by increasing order of precedence). 1

logical_operator 2  ::= and | or | nand | nor | xor | xnor

relational_operator ::= = | /= | < | <= | > | >=

shift_operator 2  ::= sll | srl | sla | sra | rol | ror

adding_operator ::= + | - | &

sign ::= + | -

multiplying_operator ::= * | / | mod | rem

miscellaneous_operator ::= ** | abs | not

  The binary logical operators (and , or , nand , nor , xor , xnor) and the unary not logical operator are
predefined for types BIT or BOOLEAN and one-dimensional arrays whose element type is BIT or BOOLEAN
. The operands must be of the same base type for the binary logical operators and the same length if they
are arrays. Both operands of relational operators must be of the same type and the result type is BOOLEAN
. The equality operator and inequality operator (’=’ and ’/=’) are defined for all types (other than file
types). The remaining relational operators, ordering operators, are predefined for any scalar type, and for
any one-dimensional array whose elements are of a discrete type (enumeration or integer type). 

The left operand of the shift operators (VHDL-93 only) is a one-dimensional array with element type of
BIT or BOOLEAN ; the right operand must be INTEGER . 

The adding operators (’+’ and ’-’) are predefined for any numeric type. You cannot use the adding
operators on BIT or BIT_VECTOR without overloading. The concatenation operator ’&’ is predefined for
any one-dimensional array type. The signs (’+’ and ’-’) are defined for any numeric type. 

The multiplying operators are: ’*’ , ’/’ , mod , and rem . The operators ’*’ and ’/’ are predefined for
any integer or floating-point type, and the operands and the result are of the same type. The operators
mod and rem are predefined for any integer type, and the operands and the result are of the same type. In
addition, you can multiply an INTEGER or REAL by any physical type and the result is the physical type.
You can also divide a physical type by REAL or INTEGER and the result is the physical type. If you divide
a physical type by the same physical type, the result is an INTEGER (actually type UNIVERSAL_INTEGER ,
which is a predefined anonymous type [VHDL LRM7.5]). Once again--you cannot use the multiplying
operators on BIT or BIT_VECTOR types without overloading the operators. 

The exponentiating operator, ’**’ , is predefined for integer and floating-point types. The right operand,
the exponent, is type INTEGER . You can only use a negative exponent with a left operand that is a



floating-point type, and the result is the same type as the left operand. The unary operator abs (absolute
value) is predefined for any numeric type and the result is the same type. The operators abs , ’**’ , and
not are grouped as miscellaneous operators. 

Here are some examples of the use of VHDL operators:

entity Operator_1 is end; architecture Behave of Operator_1 is
begin process
variable b : BOOLEAN; variable bt : BIT := ’1’; variable i : INTEGER;
variable pi : REAL := 3.14; variable epsilon : REAL := 0.01;
variable bv4 : BIT_VECTOR (3 downto 0) := "0001";
variable bv8 : BIT_VECTOR (0 to 7);
begin 
b   := "0000" < bv4;     -- b is TRUE, "0000" treated as BIT_VECTOR.
b   := ’f’ > ’g’;        -- b is FALSE, ’dictionary’ comparison.
bt  := ’0’ and bt;      -- bt is ’0’, analyzer knows ’0’ is BIT.
bv4 := not bv4;         -- bv4 is now "1110".
i   := 1 + 2;            -- Addition, must be compatible types.
i   := 2 ** 3;           -- Exponentiation, exponent must be integer.
i   := 7/3;              -- Division, L/R rounded towards zero, i=2.
i   := 12 rem 7;        -- Remainder, i=5. In general:
                         -- L rem R = L-((L/R)*R).
i   := 12 mod 7;        -- modulus, i=5. In general:
                         -- L mod R = L-(R*N) for an integer N.
-- shift := sll | srl | sla | sra | rol | ror (VHDL-93 only)
bv4 := "1001" srl 2; -- Shift right logical, now bv4="0100".
-- Logical shift fills with T’LEFT.
bv4 := "1001" sra 2; -- Shift right arithmetic, now bv4="0111".
-- Arithmetic shift fills with element at end being vacated.
bv4 := "1001" ror 2; -- Rotate right, now bv4="0110".
-- Rotate wraps around. 
-- Integer argument to any shift operator may be negative or zero.
if (pi*2.718)/2.718 = 3.14 then wait; end if; -- This is unreliable.
if (abs(((pi*2.718)/2.718)-3.14)<epsilon) then wait; end if; -- Better.
bv8 := bv8(1 to 7) & bv8(0); -- Concatenation, a left rotation.
wait; end process;
end;

1. The not operator is a logical operator but has the precedence of a miscellaneous operator. 2. Underline
means "new to VHDL-93."   

10.12  Arithmetic

The following example illustrates type checking and type conversion in VHDL arithmetic operations
[VHDL 93LRM7.3.4-7.3.5]:

entity Arithmetic_1 is end; architecture Behave of Arithmetic_1 is
        begin process
        variable i : INTEGER := 1; variable r : REAL := 3.33;
        variable b : BIT := ’1’;
        variable bv4 : BIT_VECTOR (3 downto 0) := "0001";
        variable bv8 : BIT_VECTOR (7 downto 0) := B"1000_0000";



        begin 
--              i := r;            -- you can’t assign REAL to INTEGER. 
--              bv4 := bv4 + 2;    -- you can’t add BIT_VECTOR and INTEGER.
--              bv4 := ’1’;        -- you can’t assign BIT to BIT_VECTOR.
--              bv8 := bv4;        -- an error, the arrays are different sizes.
r               := REAL(i);        -- OK, uses a type conversion.
i               := INTEGER(r);     -- OK (0.5 rounds up or down).
bv4             := "001" & ’1’;    -- OK, you can mix an array and a scalar.
bv8             := "0001" & bv4;   -- OK, if arguments are the correct lengths.
wait; end process; end;

The next example shows arithmetic operations between types and subtypes, and also illustrates range
checking during analysis and simulation:

entity Arithmetic_2 is end; architecture Behave of Arithmetic_2 is
type TC is range 0 to 100;                                                                                                                              -- Type INTEGER.
type TF is range 32 to 212;                                                                                                                             -- Type INTEGER.
subtype STC is INTEGER range 0 to 100;                                                                                                                          -- Subtype of type INTEGER.
subtype STF is INTEGER range 32 to 212;                                                                                                                                 -- Base type is INTEGER.
begin process 
variable t1 : TC := 25;   variable t2 : TF := 32;
variable st1 : STC := 25; variable st2 : STF := 32;
begin 
--              t1              := t2;            -- Illegal, different types.
--              t1              := st1;           -- Illegal, different types and subtypes.
                st2             := st1;           -- OK to use same base types. 
                st2             := st1 + 1;       -- OK to use subtype and base type.
--              st2             := 213;           -- Error, outside range at analysis time.
--              st2             := 212 + 1;       -- Error, outside range at analysis time.
                st1             := st1 + 100;     -- Error, outside range at initialization.
wait; end process; end;

The MTI simulator, for example, gives the following informative error message during simulation of the
preceding model:

# ** Fatal: Value 25 is out of range 32 to 212
#    Time: 0 ns  Iteration: 0  Instance:/
# Stopped at Arithmetic_2.vhd line 12
# Fatal error at Arithmetic_2.vhd line 12

The assignment st2 := st1 causes this error (since st1 is initialized to 25). 

Operations between array types and subtypes are a little more complicated as the following example
illustrates:

entity Arithmetic_3 is end; architecture Behave of Arithmetic_3 is
type TYPE_1 is array (INTEGER range 3 downto 0) of BIT;
type TYPE_2 is array (INTEGER range 3 downto 0) of BIT;
subtype SUBTYPE_1 is BIT_VECTOR (3 downto 0);
subtype SUBTYPE_2 is BIT_VECTOR (3 downto 0);
begin process
variable bv4 : BIT_VECTOR (3 downto 0) := "0001";
variable st1 : SUBTYPE_1 := "0001"; variable t1 : TYPE_1 := "0001";
variable st2 : SUBTYPE_2 := "0001"; variable t2 : TYPE_2 := "0001";
begin 
                bv4 := st1;                -- OK, compatible type and subtype.
--              bv4 := t1;                 -- Illegal, different types.
                bv4 := BIT_VECTOR(t1);     -- OK, type conversion.



                st1 := bv4;                -- OK, compatible subtype and base type.
--              st1 := t1;                 -- Illegal, different types.
                st1 := SUBTYPE_1(t1);      -- OK, type conversion.
--              t1  := st1;                -- Illegal, different types.
--              t1  := bv4;                -- Illegal, different types.
                t1  := TYPE_1(bv4);        -- OK, type conversion.
--              t1  := t2;                 -- Illegal, different types.
                t1  := TYPE_1(t2);         -- OK, type conversion.
                st1 := st2;                -- OK, compatible subtypes.
wait; end process; end;

The preceding example uses BIT and BIT_VECTOR types, but exactly the same considerations apply to
STD_LOGIC and STD_LOGIC_VECTOR types or other arrays. Notice the use of type conversion, written as
type_mark’(expression), to convert between closely related types. Two types are closely related if they
are abstract numeric types (integer or floating-point) or arrays with the same dimension, each index type
is the same (or are themselves closely related), and each element has the same type [VHDL
93LRM7.3.5].

10.12.1  IEEE Synthesis Packages

The IEEE 1076.3 standard synthesis packages allow you to perform arithmetic on arrays of the type BIT
and STD_LOGIC . 1 The NUMERIC_BIT package defines all of the operators in Table 10.16 (except for the
exponentiating operator ’**’ ) for arrays of type BIT . Here is part of the package header, showing the
declaration of the two types UNSIGNED and SIGNED , and an example of one of the function declarations
that overloads the addition operator ’+’ for UNSIGNED arguments:

package Part_NUMERIC_BIT is
type UNSIGNED is array (NATURAL range <> ) of BIT;
type SIGNED is array (NATURAL range <> ) of BIT;
function "+" (L, R : UNSIGNED) return UNSIGNED;
-- other function definitions that overload +, -, = , >, and so on.
end Part_NUMERIC_BIT;

The package bodies included in the 1076.3 standard define the functionality of the packages. Companies
may implement the functions in any way they wish--as long as the results are the same as those defined
by the standard. Here is an example of the parts of the NUMERIC_BIT package body that overload the
addition operator ’+’ for two arguments of type UNSIGNED (even with my added comments the code is
rather dense and terse, but remember this is code that we normally never see or need to understand):

package body Part_NUMERIC_BIT is
constant NAU : UNSIGNED(0 downto 1) := (others =>’0’); -- Null array.
constant NAS : SIGNED(0 downto 1):=(others => ’0’); -- Null array.
constant NO_WARNING : BOOLEAN := FALSE; -- Default to emit warnings.
function MAX (LEFT, RIGHT : INTEGER) return INTEGER is
begin -- Internal function used to find longest of two inputs.
if LEFT > RIGHT then return LEFT; else return RIGHT; end if; end MAX;
function ADD_UNSIGNED (L, R : UNSIGNED; C: BIT) return UNSIGNED is
constant L_LEFT : INTEGER := L’LENGTH-1; -- L, R must be same length.
alias XL : UNSIGNED(L_LEFT downto 0) is L; -- Descending alias,
alias XR : UNSIGNED(L_LEFT downto 0) is R; -- aligns left ends.
variable RESULT : UNSIGNED(L_LEFT downto 0); variable CBIT : BIT := C;
begin for I in 0 to L_LEFT loop -- Descending alias allows loop.
RESULT(I) := CBIT xor XL(I) xor XR(I); -- CBIT = carry, initially = C.
CBIT := (CBIT and XL(I)) or (CBIT and XR(I)) or (XL(I) and XR(I));
end loop; return RESULT; end ADD_UNSIGNED;



function RESIZE (ARG : UNSIGNED; NEW_SIZE : NATURAL) return UNSIGNED is 
constant ARG_LEFT : INTEGER := ARG’LENGTH-1;
alias XARG : UNSIGNED(ARG_LEFT downto 0) is ARG; -- Descending range.
variable RESULT : UNSIGNED(NEW_SIZE-1 downto 0) := (others => ’0’);
begin -- resize the input ARG to length NEW_SIZE
        if (NEW_SIZE < 1) then return NAU; end if; -- Return null array.
        if XARG’LENGTH = 0 then return RESULT; end if; -- Null to empty.
        if (RESULT’LENGTH < ARG’LENGTH) then -- Check lengths.
                RESULT(RESULT’LEFT downto 0) := XARG(RESULT’LEFT downto 0);
        else -- Need to pad the result with some ’0’s.
                RESULT(RESULT’LEFT downto XARG’LEFT + 1) := (others => ’0’); 
                RESULT(XARG’LEFT downto 0) := XARG;
        end if; return RESULT;
end RESIZE;
function "+" (L, R : UNSIGNED) return UNSIGNED is -- Overloaded ’+’.
constant SIZE : NATURAL := MAX(L’LENGTH, R’LENGTH);
begin -- If length of L or R < 1 return a null array.
if ((L’LENGTH < 1) or (R’LENGTH < 1)) then return NAU; end if;
return ADD_UNSIGNED(RESIZE(L, SIZE), RESIZE(R, SIZE), ’0’); end "+";
end Part_NUMERIC_BIT;

The following conversion functions are also part of the NUMERIC_BIT package:

function TO_INTEGER (ARG : UNSIGNED) return NATURAL;
function TO_INTEGER (ARG : SIGNED) return INTEGER;
function TO_UNSIGNED (ARG, SIZE : NATURAL) return UNSIGNED;
function TO_SIGNED (ARG : INTEGER; SIZE : NATURAL) return SIGNED;
function RESIZE (ARG : SIGNED; NEW_SIZE : NATURAL) return SIGNED;
function RESIZE (ARG : UNSIGNED; NEW_SIZE : NATURAL) return UNSIGNED;
-- set XMAP to convert unknown values, default is ’X’->’0’
function TO_01(S : UNSIGNED; XMAP : STD_LOGIC := ’0’) return UNSIGNED;
function TO_01(S : SIGNED; XMAP : STD_LOGIC := ’0’) return SIGNED;

The NUMERIC_STD package is almost identical to the NUMERIC_BIT package except that the
UNSIGNED and SIGNED types are declared in terms of the STD_LOGIC type from the
Std_Logic_1164 package as follows:

library IEEE; use IEEE.STD_LOGIC_1164.all;
package Part_NUMERIC_STD is
type UNSIGNED is array (NATURAL range <>) of STD_LOGIC;
type SIGNED is array (NATURAL range <>) of STD_LOGIC;
end Part_NUMERIC_STD;

The NUMERIC_STD package body is similar to NUMERIC_BIT with the addition of a comparison
function called STD_MATCH , illustrated by the following:

-- function STD_MATCH (L, R: T) return BOOLEAN;
-- T = STD_ULOGIC UNSIGNED SIGNED STD_LOGIC_VECTOR STD_ULOGIC_VECTOR

The STD_MATCH function uses the following table to compare logic values:

type BOOLEAN_TABLE is array(STD_ULOGIC, STD_ULOGIC) of BOOLEAN;
constant MATCH_TABLE : BOOLEAN_TABLE := (
---------------------------------------------------------------------
-- U     X     0     1     Z     W     L     H     -
---------------------------------------------------------------------
(FALSE,FALSE,FALSE,FALSE,FALSE,FALSE,FALSE,FALSE, TRUE), -- | U | 
(FALSE,FALSE,FALSE,FALSE,FALSE,FALSE,FALSE,FALSE, TRUE), -- | X | 



(FALSE,FALSE, TRUE,FALSE,FALSE,FALSE, TRUE,FALSE, TRUE), -- | 0 | 
(FALSE,FALSE,FALSE, TRUE,FALSE,FALSE,FALSE, TRUE, TRUE), -- | 1 | 
(FALSE,FALSE,FALSE,FALSE,FALSE,FALSE,FALSE,FALSE, TRUE), -- | Z | 
(FALSE,FALSE,FALSE,FALSE,FALSE,FALSE,FALSE,FALSE, TRUE), -- | W | 
(FALSE,FALSE, TRUE,FALSE,FALSE,FALSE, TRUE,FALSE, TRUE), -- | L | 
(FALSE,FALSE,FALSE, TRUE,FALSE,FALSE,FALSE, TRUE, TRUE), -- | H | 
( TRUE, TRUE, TRUE, TRUE, TRUE, TRUE, TRUE, TRUE, TRUE));-- | - |

Thus, for example (notice we need type conversions):

IM_TRUE = STD_MATCH(STD_LOGIC_VECTOR    ("10HLXWZ-"), 
                    STD_LOGIC_VECTOR    ("HL10----"))    -- is TRUE

The following code is similar to the first simple example of Section 10.1, but illustrates the use of the
Std_Logic_1164 and NUMERIC_STD packages:

entity Counter_1 is end;
        library STD; use STD.TEXTIO.all; 
        library IEEE; use IEEE.STD_LOGIC_1164.all;
use work.NUMERIC_STD.all; 
architecture Behave_2 of Counter_1 is 
        signal Clock : STD_LOGIC := ’0’; 
        signal Count : UNSIGNED (2 downto 0) := "000";
        begin 
        process begin
                wait for 10 ns; Clock <= not Clock; 
                if (now > 340 ns) then wait;
                end if;
        end process;
        process begin 
                wait until (Clock = ’0’);
                if (Count = 7) 
                        then Count <= "000"; 
                        else Count <= Count + 1; 
                end if;
        end process;
        process (Count) variable L: LINE; begin write(L, now); 
        write(L, STRING’(" Count=")); write(L, TO_INTEGER(Count)); 
        writeline(output, L); 
        end process;
end;

The preceding code looks similar to the code in Section 10.1 (and the output is identical), but there is
more going on here:

Line 3 is a library clause and a use clause for the std_logic_1164 package, so you can use the
STD_LOGIC type and the NUMERIC_BIT package. 
Line 4 is a use clause for NUMERIC_BIT package that was previously analyzed into the library work
. If the package is instead analyzed into the library IEEE , you would use the name
IEEE.NUMERIC_BIT.all here. The NUMERIC_BIT package allows you to use the type UNSIGNED . 
Line 6 declares Clock to be type STD_LOGIC and initializes it to ’0’ , instead of the default initial
value STD_LOGIC’LEFT (which is ’U’ ). 
Line 7 declares Count to be a 3-bit array of type UNSIGNED from NUMERIC_BIT and initializes it
using a bit-string literal. 
Line 10 uses the overloaded ’not’ operator from std_logic_1164 . 
Line 15 uses the overloaded ’=’ operator from std_logic_1164 . 



Line 16 uses the overloaded ’=’ operator from NUMERIC_BIT . 
Line 17 requires a bit-string literal, you cannot use Count <= 0 here. 
Line 18 uses the overloaded ’+’ operator from NUMERIC_BIT . 
Line 22 converts Count , type UNSIGNED , to type INTEGER . 

1. IEEE Std 1076.3-1997 was approved by the IEEE Standards Board on 20 March 1997. The synthesis
package code on the following pages is reprinted with permission from IEEE Std 1076.3-1997,

page  Next  page

10.13  Concurrent Statements

A concurrent statement [VHDL LRM9] is one of the following statements:

concurrent_statement ::=
        block_statement
        | process_statement
        | [ label : ] [ postponed ] procedure_call ;
        | [ label : ] [ postponed ] assertion ;
        | [ label : ] [ postponed ] conditional_signal_assignment
        | [ label : ] [ postponed ] selected_signal_assignment
        | component_instantiation_statement
        | generate_statement

The following sections describe each of these statements in turn.

10.13.1  Block Statement

A block statement has the following format [VHDL LRM9.1]:

block_statement ::=
        block_label: block [(guard_expression)] [is]
                [generic (generic_interface_list);
                [generic map (generic_association_list);]]
                [port (port_interface_list);
                [port map (port_association_list);]]
                        {block_declarative_item}
                        begin
                        {concurrent_statement}
        end block [block_label] ;

Blocks may have their own ports and generics and may be used to split an architecture into several
hierarchical parts (blocks can also be nested). As a very general rule, for the same reason that it is better
to split a computer program into separate small modules, it is usually better to split a large architecture
into smaller separate entity-architecture pairs rather than several nested blocks.

A block does have a unique feature: It is possible to specify a guard expression for a block. This creates
a special signal, GUARD , that you can use within the block to control execution [VHDL LRM9.5]. It also
allows you to model three-state buses by declaring guarded signals (signal kinds register and bus).



When you make an assignment statement to a signal, you define a driver for that signal. If you make
assignments to guarded signals in a block, the driver for that signal is turned off, or disconnected, when
the GUARD signal is FALSE . The use of guarded signals and guarded blocks can become quite
complicated, and not all synthesis tools support these VHDL features.

The following example shows two drivers, A and B , on a three-state bus TSTATE , enabled by signals OEA
and OEB . The drivers are enabled by declaring a guard expression after the block declaration and using
the keyword guarded in the assignment statements. A disconnect statement [VHDL LRM5.3] models
the driver delay from driving the bus to the high-impedance state (time to "float").

library ieee; use ieee.std_logic_1164.all;
entity bus_drivers is end;
architecture Structure_1 of bus_drivers is
signal TSTATE: STD_LOGIC bus; signal A, B, OEA, OEB : STD_LOGIC:= ’0’;
begin 
process begin OEA <= ’1’ after 100 ns, ’0’ after 200 ns; 
OEB <= ’1’ after 300 ns; wait; end process;
B1 : block (OEA = ’1’)
disconnect all : STD_LOGIC after 5 ns; -- Only needed for float time.
begin TSTATE <= guarded not A after 3 ns; end block;
B2 : block (OEB = ’1’)
disconnect all : STD_LOGIC after 5 ns; -- Float time = 5 ns.
begin TSTATE <= guarded not B after 3 ns; end block;
end;
 
                             1    2    3    4    5        6        7
      Time(fs) + Cycle  tstate    a    b  oea  oeb b1.GUARD b2.GUARD
----------------------  ------ ---- ---- ---- ---- -------- --------
                  0+ 0:    ’U’  ’0’  ’0’  ’0’  ’0’    FALSE    FALSE
                  0+ 1: *  ’Z’  ’0’  ’0’  ’0’  ’0’    FALSE    FALSE
          100000000+ 0:    ’Z’  ’0’  ’0’ *’1’  ’0’ *   TRUE    FALSE
          103000000+ 0: *  ’1’  ’0’  ’0’  ’1’  ’0’     TRUE    FALSE
          200000000+ 0:    ’1’  ’0’  ’0’ *’0’  ’0’ *  FALSE    FALSE
          200000000+ 1: *  ’Z’  ’0’  ’0’  ’0’  ’0’    FALSE    FALSE
          300000000+ 0:    ’Z’  ’0’  ’0’  ’0’ *’1’    FALSE *   TRUE
          303000000+ 0: *  ’1’  ’0’  ’0’  ’0’  ’1’    FALSE     TRUE

Notice the creation of implicit guard signals b1.GUARD and b2.GUARD for each guarded block. There
is another, equivalent, method that uses the high-impedance value explicitly as in the following
example:

architecture Structure_2 of bus_drivers is
signal TSTATE : STD_LOGIC; signal A, B, OEA, OEB : STD_LOGIC := ’0’;
begin 
process begin
OEA <= ’1’ after 100 ns, ’0’ after 200 ns; OEB <= ’1’ after 300 ns; wait; end process
process(OEA, OEB, A, B) begin
        if    (OEA = ’1’) then TSTATE <= not A after 3 ns; 
        elsif (OEB = ’1’) then TSTATE <= not B after 3 ns; 
        else TSTATE <= ’Z’ after 5 ns;
        end if;
end process;
end;

This last method is more widely used than the first, and what is more important, more widely accepted
by synthesis tools. Most synthesis tools are capable of recognizing the value ’Z’ on the RHS of an



assignment statement as a cue to synthesize a three-state driver. It is up to you to make sure that multiple
drivers are never enabled simultaneously to cause contention.

10.13.2  Process Statement

A process statement has the following format [VHDL LRM9.2]:

process_statement ::=
[process_label:]
[postponed] process  [(signal_name {, signal_name})]
[is] {subprogram_declaration    | subprogram_body
    | type_declaration  | subtype_declaration
    | constant_declaration      | variable_declaration
    | file_declaration  | alias_declaration
    | attribute_declaration     | attribute_specification
    | use_clause
    | group_declaration         | group_template_declaration}
begin 
        {sequential_statement}
end [postponed] process [process_label];

The following process models a 2:1 MUX (combinational logic):

entity Mux_1 is port (i0, i1, sel : in BIT := ’0’; y : out BIT); end; 
architecture Behave of Mux_1 is
        begin process (i0, i1, sel) begin -- i0, i1, sel = sensitivity set
        case sel is when ’0’ => y <= i0; when ’1’ => y <= i1; end case;
end process; end;

This process executes whenever an event occurs on any of the signals in the process sensitivity set (i0,
i1, sel). The execution of a process occurs during a simulation cycle--a delta cycle. Assignment
statements to signals may trigger further delta cycles. Time advances when all transactions for the
current time step are complete and all signals updated.

The following code models a two-input AND gate (combinational logic):

entity And_1 is port (a, b : in BIT := ’0’; y : out BIT); end; 
architecture Behave of And_1 is
begin process (a, b) begin y <= a and b; end process; end;

The next example models a D flip-flop (sequential logic). The process statement is executed
whenever there is an event on clk . The if statement updates the output q with the input d on the rising
edge of the signal clk . If the if statement condition is false (as it is on the falling edge of clk ), then
the assignment statement q <= d will not be executed, and q will keep its previous value. The process
thus requires the value of q to be stored between successive process executions, and this implies
sequential logic.

entity FF_1 is port (clk, d: in BIT := ’0’; q : out BIT); end; 
architecture Behave of FF_1 is
begin process (clk) begin
        if clk’EVENT and clk = ’1’ then q <= d; end if;
end process; end;

The behavior of the next example is identical to the previous model. Notice that the wait statement is at



the end of the equivalent process with the signals in the sensitivity set (in this case just one signal, clk )
included in the sensitivity list (that follows the keyword on ).

entity FF_2 is port (clk, d: in BIT := ’0’; q : out BIT); end; 
architecture Behave of FF_2 is
begin process begin -- The equivalent process has a wait at the end:
                if clk’event and clk = ’1’ then q <= d; end if; wait on clk;
end process; end;

If we use a wait statement in a process statement, then we may not use a process sensitivity set (the
reverse is true: If we do not have a sensitivity set for a process, we must include a wait statement or the
process will execute endlessly):

entity FF_3 is port (clk, d: in BIT := ’0’; q : out BIT); end; 
architecture Behave of FF_3 is
begin process begin -- No sensitivity set with a wait statement.
        wait until clk = ’1’; q <= d; 
end process; end;

If you include ports (interface signals) in the sensitivity set of a process statement, they must be ports
that can be read (they must be of mode in , inout , or buffer , see Section 10.7).

10.13.3  Concurrent Procedure Call

A concurrent procedure call appears outside a process statement [VHDL LRM9.3]. The concurrent
procedure call is a shorthand way of writing an equivalent process statement that contains a procedure
call (Section 10.10.4):

package And_Pkg is procedure V_And(a,b:BIT; signal c:out BIT); end;
package body And_Pkg is procedure V_And(a,b:BIT; signal c:out BIT) is 
        begin c <= a and b; end; end And_Pkg;
use work.And_Pkg.all; entity Proc_Call_2 is end; 
architecture Behave of Proc_Call_2 is signal A, B, Y : BIT := ’0’;
        begin V_And (A, B, Y); -- Concurrent procedure call.
process begin wait; end process; -- Extra process to stop.
end;

10.13.4  Concurrent Signal Assignment

There are two forms of concurrent signal assignment statement. A selected signal assignment statement
is equivalent to a case statement inside a process statement [VHDL LRM9.5.2]:

selected_signal_assignment ::=
        with expression select
                name|aggregate <= [guarded]
                        [transport|[reject time_expression] inertial] 
                                waveform when choice {| choice}
                                        {, waveform when choice {| choice} } ;

The following design unit, Selected_1, uses a selected signal assignment. The equivalent unit,
Selected_2, uses a case statement inside a process statement.

entity Selected_1 is end; architecture Behave of Selected_1 is
signal y,i1,i2 : INTEGER; signal sel : INTEGER range 0 to 1;



begin with sel select y <= i1 when 0, i2 when 1; end;
entity Selected_2 is end; architecture Behave of Selected_2 is
signal i1,i2,y : INTEGER; signal sel : INTEGER range 0 to 1;
begin process begin
        case sel is when 0 => y <= i1; when 1 => y <= i2; end case;
        wait on i1, i2;
end process; end;

The other form of concurrent signal assignment is a conditional signal assignment statement that, in its
most general form, is equivalent to an if statement inside a process statement [VHDL LRM9.5.1]:

conditional_signal_assignment ::=
                name|aggregate <= [guarded]
        [transport|[reject time_expression] inertial]
                        {waveform when boolean_expression else}
                                waveform [when boolean_expression];

Notice that in VHDL-93 the else clause is optional. Here is an example of a conditional signal
assignment, followed by a model using the equivalent process with an if statement:

entity Conditional_1 is end; architecture Behave of Conditional_1 is
signal y,i,j : INTEGER; signal clk : BIT;
begin y <= i when clk = ’1’ else j; -- conditional signal assignment
end;
entity Conditional_2 is end; architecture Behave of Conditional_2 is
signal y,i : INTEGER; signal clk : BIT;
begin process begin
        if clk = ’1’ then y <= i; else y <= y ; end if; wait on clk;
end process; end;

A concurrent signal assignment statement can look just like a sequential signal assignment statement, as
in the following example:

entity Assign_1 is end; architecture Behave of Assign_1 is
signal Target, Source : INTEGER;
        begin Target <= Source after 1 ns; -- looks like signal assignment
end;

However, outside a process statement, this statement is a concurrent signal assignment and has its own
equivalent process statement. Here is the equivalent process for the example:

entity Assign_2 is end; architecture Behave of Assign_2 is
signal Target, Source : INTEGER; 
begin process begin 
        Target <= Source after 1 ns; wait on Source;
end process; end;

Every process is executed once during initialization. In the previous example, an initial value will be
scheduled to be assigned to Target even though there is no event on Source . If, for some reason, you
do not want this to happen, you need to rewrite the concurrent assignment statement as a process
statement with a wait statement before the assignment statement:

entity Assign_3 is end; architecture Behave of Assign_3 is
signal Target, Source : INTEGER; begin process begin 
        wait on Source; Target <= Source after 1 ns;
end process; end;



10.13.5  Concurrent Assertion Statement

A concurrent assertion statement is equivalent to a passive process statement (without a sensitivity list)
that contains an assertion statement followed by a wait statement [VHDL LRM9.4].

concurrent_assertion_statement
::= [ label : ] [ postponed ] assertion ;

If the assertion condition contains a signal, then the equivalent process statement will include a final
wait statement with a sensitivity clause. A concurrent assertion statement with a condition that is static
expression is equivalent to a process statement that ends in a wait statement that has no sensitivity
clause. The equivalent process will execute once, at the beginning of simulation, and then wait
indefinitely.

10.13.6  Component Instantiation

A component instantiation statement in VHDL is similar to placement of a component in a
schematic--an instantiated component is somewhere between a copy of the component and a reference to
the component. Here is the definition [VHDL LRM9.6]:

component_instantiation_statement ::=
instantiation_label: 
 [component] component_name 
|entity entity_name [(architecture_identifier)]
|configuration configuration_name
        [generic map (generic_association_list)]
        [port map (port_association_list)] ;

We examined component instantiation using a component_name in Section 10.5. If we instantiate a
component in this way we must declare the component (see BNF [10.9]). To bind a component to an
entity-architecture pair we can use a configuration, as illustrated in Figure 10.1, or we can use the
default binding as described in Section 10.7. In VHDL-93 we have another alternative--we can directly
instantiate an entity or configuration. For example:

entity And_2 is port (i1, i2 : in BIT; y : out BIT); end;
architecture Behave of And_2 is begin y <= i1 and i2; end;
entity Xor_2 is port (i1, i2 : in BIT; y : out BIT); end;
architecture Behave of Xor_2 is begin y <= i1 xor i2; end;
entity Half_Adder_2 is port (a,b : BIT := ’0’; sum, cry : out BIT); end;
architecture Netlist_2 of Half_Adder_2 is
use work.all; -- need this to see the entities Xor_2 and And_2
begin
        X1 : entity Xor_2(Behave) port map (a, b, sum); -- VHDL-93 only
        A1 : entity And_2(Behave) port map (a, b, cry); -- VHDL-93 only
end;

10.13.7  Generate Statement

A generate statement [VHDL LRM9.7] simplifies repetitive code:

generate_statement ::=
generate_label:          for generate_parameter_specification
                 |if boolean_expression



generate [{block_declarative_item} begin]
        {concurrent_statement}
end generate [generate_label] ;

Here is an example (notice the labels are required):

entity Full_Adder is port (X, Y, Cin : BIT; Cout, Sum: out BIT); end;
architecture Behave of Full_Adder is begin Sum <= X xor Y xor Cin; 
Cout <= (X and Y) or (X and Cin) or (Y and Cin); end;
entity Adder_1 is 
        port (A, B : in BIT_VECTOR (7 downto 0) := (others => ’0’); 
        Cin : in BIT := ’0’; Sum : out BIT_VECTOR (7 downto 0); 
        Cout : out BIT); end; 
architecture Structure of Adder_1 is use work.all;
component Full_Adder port (X, Y, Cin: BIT; Cout, Sum: out BIT);
end component; 
signal C : BIT_VECTOR(7 downto 0);
begin AllBits : for i in 7 downto 0 generate
        LowBit : if i = 0 generate
                FA : Full_Adder port map (A(0), B(0), Cin, C(0), Sum(0)); 
        end generate;
        OtherBits : if i /= 0 generate
                FA : Full_Adder port map (A(i), B(i), C(i-1), C(i), Sum(i));
        end generate; 
end generate;
Cout <= C(7); 
end;

The instance names within a generate loop include the generate parameter. For example for i=6 ,
FA’INSTANCE_NAME is

:adder_1(structure):allbits(6):otherbits:fa:

10.14  Execution

Two successive statements may execute in either a concurrent or sequential fashion depending on where
the statements appear.

statement_1; statement_2;

In sequential execution, statement_1 in this sequence is always evaluated before statement 2 . In
concurrent execution, statement_1 and statement_2 are evaluated at the same time (as far as we are
concerned--obviously on most computers exactly parallel execution is not possible). Concurrent
execution is the most important difference between VHDL and a computer programming language.
Suppose we have two signal assignment statements inside a process statement. In this case
statement_1 and statement_2 are sequential assignment statements:

entity Sequential_1 is end; architecture Behave of Sequential_1 is
signal s1, s2 : INTEGER := 0; 
begin 
        process begin
                s1 <= 1;      -- sequential signal assignment 1
                s2 <= s1 + 1; -- sequential signal assignment 2
                wait on s1, s2 ; 



        end process; 
end;
      Time(fs) + Cycle            s1           s2 
----------------------  ------------ ------------ 
                  0+ 0:            0            0 
                  0+ 1: *          1 *          1 
                  0+ 2: *          1 *          2 
                  0+ 3: *          1 *          2

If the two statements are outside a process statement they are concurrent assignment statements, as in
the following example:

entity Concurrent_1 is end; architecture Behave of Concurrent_1 is
signal s1, s2 : INTEGER := 0; begin 
        L1 : s1 <= 1;      -- concurrent signal assignment 1
        L2 : s2 <= s1 + 1; -- concurrent signal assignment 2
end;
      Time(fs) + Cycle            s1           s2 
----------------------  ------------ ------------ 
                  0+ 0:            0            0 
                  0+ 1: *          1 *          1 
                  0+ 2:            1 *          2

The two concurrent signal assignment statements in the previous example are equivalent to the two
processes, labeled as P1 and P2 , in the following model.

entity Concurrent_2 is end; architecture Behave of Concurrent_2 is
signal s1, s2 : INTEGER := 0; begin 
        P1 : process begin s1 <= 1;      wait on s2 ; end process;
        P2 : process begin s2 <= s1 + 1; wait on s1 ; end process;
end; 
      Time(fs) + Cycle            s1           s2 
----------------------  ------------ ------------ 
                  0+ 0:            0            0 
                  0+ 1: *          1 *          1 
                  0+ 2: *          1 *          2 
                  0+ 3: *          1            2

Notice that the results are the same (though the trace files are slightly different) for the architectures
Sequential_1, Concurrent_1, and Concurrent_2. Updates to signals occur at the end of the
simulation cycle, so the values used will always be the old values. So far things seem fairly simple: We
have sequential execution or concurrent execution. However, variables are updated immediately, so the
variable values that are used are always the new values. The examples in Table 10.17 illustrate this very
important difference.



TABLE 10.17    Variables and signals in VHDL.

Variables Signals
entity Execute_1 is end; 
architecture Behave of Execute_1 is
begin 
        process 
        variable v1 : INTEGER := 1; 
        variable v2 : INTEGER := 2;
        begin
                v1 := v2; -- before: v1 = 1, v2 = 2
                v2 := v1; -- after:  v1 = 2, v2 = 2
                wait; 
        end process; 
end;

entity Execute_2 is end; 
architecture Behave of Execute_2 
signal s1 : INTEGER := 1; 
signal s2 : INTEGER := 2;
begin 
        process 
        begin
                s1 <= s2; -- before: s1 = 1, s2 = 2
                s2 <= s1; -- after:  s1 = 2, s2 = 1
                wait; 
        end process; 
end;

The various concurrent and sequential statements in VHDL are summarized in Table 10.18.

TABLE 10.18    Concurrent and sequential statements in VHDL.

Concurrent [VHDL LRM9] Sequential [VHDL LRM8]

block

process

concurrent_procedure_call

concurrent_assertion

concurrent_signal_assignment

component_instantiation

generate

wait

assertion

signal_assignment

variable_assignment

procedure_call

if

case

loop

next

exit

return

null

10.15  Configurations and Specifications

The difference between, the interaction, and the use of component/configuration declarations and
specifications is probably the most confusing aspect of VHDL. Fortunately this aspect of VHDL is not
normally important for ASIC design. The syntax of component/configuration declarations and
specifications is shown in Table 10.19.



TABLE 10.19    VHDL binding.
configuration
declaration 1
[VHDL LRM1.3]

configuration identifier of entity_name is
  {use_clause|attribute_specification|group_declaration}
  block_configuration
end [configuration] [configuration_identifier];

block
configuration
[VHDL LRM1.3.1]

for architecture_name
  |block_statement_label
  |generate_statement_label [(index_specification)]
{use selected_name {, selected_name};}
{block_configuration|component_configuration}
end for ;

configuration
specification 1
[VHDL LRM5.2]

for
  instantiation_label{,instantiation_label}:component_name
  |others:component_name
  |all:component_name
[use
  entity entity_name [(architecture_identifier)]
  |configuration configuration_name
  |open]
[generic map (generic_association_list)]
[port map (port_association_list)];

component
declaration 1
[VHDL LRM4.5]

component identifier [is]
  [generic (local_generic_interface_list);]
  [port (local_port_interface_list);]
end component [component_identifier];

component
configuration 1
[VHDL LRM1.3.2]

for
instantiation_label {, instantiation_label}:component_name
|others:component_name
|all:component_name
[[use
   entity entity_name [(architecture_identifier)]
  |configuration configuration_name
  |open]
    [generic map (generic_association_list)]
    [port map (port_association_list)];]
[block_configuration]
end for;

 

A configuration declaration defines a configuration--it is a library unit and is one of the basic units
of VHDL code. 
A block configuration defines the configuration of a block statement or a design entity. A block
configuration appears inside a configuration declaration, a component configuration, or nested in
another block configuration. 
A configuration specification may appear in the declarative region of a generate statement, block
statement, or architecture body. 
A component declaration may appear in the declarative region of a generate statement, block
statement, architecture body, or package. 
A component configuration defines the configuration of a component and appears in a block



configuration. 

Table 10.20 shows a simple example (identical in structure to the example of Section 10.5) that
illustrates the use of each of the preceding constructs.

TABLE 10.20    VHDL binding examples.

 

entity AD2 is port (A1, A2: in BIT; Y: out BIT); end;
architecture B of AD2 is begin Y <= A1 and A2; end;
entity XR2 is port (X1, X2: in BIT; Y: out BIT); end;
architecture B of XR2 is begin Y <= X1 xor X2; end;

 
 
component
declaration
 configuration
 specification
 

entity Half_Adder is port (X, Y: BIT; Sum, Cout: out BIT); end;
architecture Netlist of Half_Adder is use work.all;
component MX port (A, B: BIT; Z :out BIT);end component; 
component MA port (A, B: BIT; Z :out BIT);end component; 
for G1:MX use entity XR2(B) port map(X1 => A,X2 => B,Y => Z);
begin
        G1:MX port map(X, Y, Sum); G2:MA port map(X, Y, Cout); 
end;

configuration
declaration
 block
 configuration
  component
  configuration
 

configuration C1 of Half_Adder is
use work.all;
        for Netlist
                for G2:MA 
                        use entity AD2(B) port map(A1 => A,A2 => B,Y => Z);
                end for;
        end for;
end;

1. Underline means "new to VHDL-93".

10.16  An Engine Controller

This section describes part of a controller for an automobile engine. Table 10.21 shows a temperature
converter that converts digitized temperature readings from a sensor from degrees Centigrade to degrees
Fahrenheit.



TABLE 10.21    A temperature converter.
library IEEE; 
use IEEE.STD_LOGIC_1164.all; -- type STD_LOGIC, rising_edge
use IEEE.NUMERIC_STD.all ; -- type UNSIGNED, "+", "/" 
entity tconv is generic TPD : TIME:= 1 ns;
        port (T_in : in UNSIGNED(11 downto 0);
        clk, rst : in STD_LOGIC; T_out : out UNSIGNED(11 downto 0)); 
end;
architecture rtl of tconv is
signal T : UNSIGNED(7 downto 0);
constant T2  : UNSIGNED(1 downto 0) := "10" ;
constant T4  : UNSIGNED(2 downto 0) := "100" ;
constant T32 : UNSIGNED(5 downto 0) := "100000" ;
begin 
        process(T) begin T_out <= T + T/T2 + T/T4 + T32 after TPD;
        end process;
end rtl;

T_in =
temperature in
degC

 

T_out =
temperature in
degF

 

The conversion
formula from
Centigrade to
Fahrenheit is:

T(degF) = (9/5)
x T(degC) + 32

 

This converter
uses the
approximation:

9/5 = 1.75 = 1
+ 0.5 + 0.25

To save area the temperature conversion is approximate. Instead of multiplying by 9/5 and adding 32 (so
0 degC becomes 32 degF and 100 degC becomes 212 degF) we multiply by 1.75 and add 32 (so 100
degC becomes 207 degF). Since 1.75 = 1 + 0.5 + 0.25, we can multiply by 1.75 using shifts (for divide
by 2, and divide by 4) together with a very simple constant addition (since 32 = "100000"). Using shift
to multiply and divide by powers of 2 is free in hardware (we just change connections to a bus). For
large temperatures the error approaches 0.05/1.8 or approximately 3 percent. We play these kinds of
tricks often in hardware computation. Notice also that temperatures measured in degC and degF are
defined as unsigned integers of the same width. We could have defined these as separate types to take
advantage of VHDL’s type checking.

Table 10.22 describes a digital filter to compute a "moving average" over four successive samples in
time (i(0), i(1), i(2), and i(3), with i(0) being the first sample).

TABLE 10.22    A digital filter.
library IEEE; 
use IEEE.STD_LOGIC_1164.all; -- STD_LOGIC type, rising_edge
use IEEE.NUMERIC_STD.all; -- UNSIGNED type, "+" and "/"

The filter
computes



use IEEE.NUMERIC_STD.all; -- UNSIGNED type, "+" and "/"
entity filter is 
        generic TPD : TIME := 1 ns;
        port (T_in : in UNSIGNED(11 downto 0); 
        rst, clk : in STD_LOGIC; 
        T_out: out UNSIGNED(11 downto 0));
end;
architecture rtl of filter is
type arr is array (0 to 3) of UNSIGNED(11 downto 0); 
signal i : arr ;
constant T4 : UNSIGNED(2 downto 0) := "100"; 
begin 
        process(rst, clk) begin
                if (rst = ’1’) then
                        for n in 0 to 3 loop i(n) <= (others =>’0’) after TPD; 
                        end loop; 
                else 
                        if(rising_edge(clk)) then
                        i(0) <= T_in after TPD;i(1) <= i(0) after TPD;
                        i(2) <= i(1) after TPD;i(3) <= i(2) after TPD;
                        end if;
                end if;
        end process;
        process(i) begin
                T_out <= ( i(0) + i(1) + i(2) + i(3) )/T4 after TPD;
        end process;
end rtl;

a moving
average
over four
successive
samples in
time.

 

Notice

i(0) i(1)
i(2) i(3)

are each
12 bits
wide.

 

Then the
sum

i(0) + i(1)
+ i(2) +
i(3)

is 14 bits
wide, and
the

average

 

( i(0) +
i(1) + i(2)
+ i(3) )/T4

 

is 12 bits
wide.

 

 



 

All delays
are generic
TPD .

 

The filter uses the following formula:

T_out <= ( i(0) + i(1) + i(2) + i(3) )/T4

Division by T4 = "100" is free in hardware. If instead, we performed the divisions before the additions,
this would reduce the number of bits to be added for two of the additions and saves us worrying about
overflow. The drawback to this approach is round-off errors. We can use the register shown in
Table 10.23 to register the inputs.

TABLE 10.23    The input register.

library IEEE; 
use IEEE.STD_LOGIC_1164.all; -- type STD_LOGIC, rising_edge
use IEEE.NUMERIC_STD.all ; -- type UNSIGNED 
entity register_in is 
generic ( TPD : TIME := 1 ns); 
port (T_in : in UNSIGNED(11 downto 0);
clk, rst : in STD_LOGIC; T_out : out UNSIGNED(11 downto 0)); end;
architecture rtl of register_in is
begin 
        process(clk, rst) begin 
                if (rst = ’1’) then T_out <= (others => ’0’) after TPD; 
                else 
                if (rising_edge(clk)) then T_out <= T_in after TPD; end if;
                end if;
        end process;
end rtl ;

12-bit-wide
register for
the
temperature
input

signals.

 

If the input is
asynchronous
(from an A/D

converter
with a
separate
clock, for
example), we
would need to
worry about
metastability.

 

All delays are
generic TPD

Table 10.24 shows a first-in, first-out stack (FIFO). This allows us to buffer the signals coming from the



sensor until the microprocessor has a chance to read them. The depth of the FIFO will depend on the
maximum amount of time that can pass without the microcontroller being able to read from the bus. We
have to determine this with statistical simulations taking into account other traffic on the bus.

TABLE 10.24    A first-in, first-out stack (FIFO).
library IEEE; use IEEE.NUMERIC_STD.all ; -- UNSIGNED type
use ieee.std_logic_1164.all; -- STD_LOGIC type, rising_edge
entity fifo is 
        generic (width : INTEGER := 12; depth : INTEGER := 16);
        port (clk, rst, push, pop : STD_LOGIC;
        Di : in UNSIGNED (width-1 downto 0);
        Do : out UNSIGNED (width-1 downto 0);
        empty, full : out STD_LOGIC);
end fifo;
architecture rtl of fifo is
subtype ptype is INTEGER range 0 to (depth-1);
signal diff, Ai, Ao : ptype; signal  f, e : STD_LOGIC;
type a is array (ptype) of UNSIGNED(width-1 downto 0);
signal mem : a ; 
function bump(signal ptr : INTEGER range 0 to (depth-1))
return INTEGER is begin 
        if (ptr = (depth-1)) then return 0; 
        else return (ptr + 1);
        end if;
end; 
begin
        process(f,e) begin full <= f ; empty <= e; end process;
        process(diff) begin
        if (diff = depth -1) then f <= ’1’; else f <= ’0’; end if;
        if (diff = 0) then e <= ’1’; else e <= ’0’; end if; 
        end process;
        process(clk, Ai, Ao, Di, mem, push, pop, e, f) begin
        if(rising_edge(clk)) then 
        if(push=’0’)and(pop=’1’)and(e = ’0’) then Do <= mem(Ao); end if;
        if(push=’1’)and(pop=’0’)and(f = ’0’) then mem(Ai) <= Di; end if;
        end if ; 
        end process;
        process(rst, clk)  begin
                if(rst = ’1’) then Ai <= 0; Ao <= 0; diff <= 0;
                else if(rising_edge(clk)) then 
                        if (push = ’1’) and (f = ’0’) and (pop = ’0’) then
                                Ai <= bump(Ai); diff <= diff + 1; 
                        elsif (pop = ’1’) and (e = ’0’) and (push = ’0’) then
                                Ao <= bump(Ao); diff <= diff - 1;
                        end if;
                end if;
                end if; 
        end process;
end;

FIFO
(first-in,
first-out)
register

 

Reads (pop
= 1) and
writes (push
= 1) are
synchronous
to the rising
edge of the
clock.

Read and
write should
not occur at
the same
time. The
width
(number of
bits in each
word) and
depth
(number of
words) are
generics.

 

External
signals:

clk , clock

rst , reset
active-high

push , write
to FIFO



pop , read
from FIFO

Di , data in

Do , data out

empty ,
FIFO flag

full , FIFO
flag

 

Internal
signals:

diff ,
difference
pointer

Ai , input
address

Ao , output
address

f , full flag

e , empty
flag

 

 

No delays in
this model.

The FIFO has flags, empty and full , that signify its state. It uses a function to increment two circular
pointers. One pointer keeps track of the address to write to next, the other pointer tracks the address to
read from. The FIFO memory may be implemented in a number of ways in hardware. We shall assume
for the moment that it will be synthesized as a bank of flip-flops.

Table 10.25 shows a controller for the two FIFOs. The controller handles the reading and writing to the
FIFO. The microcontroller attached to the bus signals which of the FIFOs it wishes to read from. The
controller then places the appropriate data on the bus. The microcontroller can also ask for the FIFO



flags to be placed in the low-order bits of the bus on a read cycle. If none of these actions are requested
by the microcontroller, the FIFO controller three-states its output drivers.

Table 10.25 shows the top level of the controller. To complete our model we shall use a package for the
component declarations:

TABLE 10.25    A FIFO controller.
library IEEE;use IEEE.STD_LOGIC_1164.all;use IEEE.NUMERIC_STD.all;
entity fifo_control is generic TPD : TIME := 1 ns;
        port(D_1, D_2 : in UNSIGNED(11 downto 0); 
        sel : in UNSIGNED(1 downto 0) ;
        read , f1, f2, e1, e2 : in STD_LOGIC;
        r1, r2, w12 : out STD_LOGIC; D : out UNSIGNED(11 downto 0)) ;
end;
architecture rtl of fifo_control is
        begin process
        (read, sel, D_1, D_2, f1, f2, e1, e2)
        begin 
        r1 <= ’0’ after TPD; r2 <= ’0’ after TPD;
        if (read = ’1’) then 
                w12 <= ’0’ after TPD; 
                case sel is
                when "01" => D <= D_1 after TPD; r1 <= ’1’ after TPD;
                when "10" => D <= D_2 after TPD; r2 <= ’1’ after TPD;
                when "00" => D(3) <= f1 after TPD; D(2) <= f2 after TPD;
                     D(1) <= e1 after TPD; D(0) <= e2 after TPD;
                when others => D <= "ZZZZZZZZZZZZ" after TPD; 
                end case;
        elsif (read = ’0’) then 
                D <= "ZZZZZZZZZZZZ" after TPD; w12 <= ’1’ after TPD;
        else D <= "ZZZZZZZZZZZZ" after TPD; 
        end if;
        end process;
end rtl;

This handles
the reading
and writing to
the FIFOs
under control
of the
processor
(mpu). The
mpu can ask
for data from
either FIFO or
for status flags
to be placed
on the bus.

 

Inputs:

 D_1

    data in from
FIFO1

 D_2

    data in from
FIFO2

 sel

    FIFO select
from mpu

 read

    FIFO read
from mpu

 f1,f2,e1,e2

    flags from
FIFOs



 

Outputs:

 r1, r2

    read enables
for FIFOs

 w12

    write enable
for FIFOs

 D

    data out to
mpu bus

TABLE 10.26    Top level of temperature controller.
library IEEE; use IEEE.STD_LOGIC_1164.all; use IEEE.NUMERIC_STD.all; 
entity T_Control is port (T_in1, T_in2 : in UNSIGNED (11 downto 0);
        sensor: in UNSIGNED(1 downto 0); 
        clk, RD, rst : in STD_LOGIC; D : out UNSIGNED(11 downto 0));
end;
architecture structure of T_Control is use work.TC_Components.all;
signal F, E : UNSIGNED (2 downto 1);
signal T_out1, T_out2, R_out1, R_out2, F1, F2, FIFO1, FIFO2 : UNSIGNED(11 downto 0);
signal RD1, RD2, WR: STD_LOGIC ;
begin 
RG1 : register_in generic map (1ns) port map (T_in1,clk,rst,R_out1);
RG2 : register_in generic map (1ns) port map (T_in2,clk,rst,R_out2);
TC1 : tconv generic map (1ns) port map (R_out1, T_out1);
TC2 : tconv generic map (1ns) port map (R_out2, T_out2);
TF1 : filter generic map (1ns) port map (T_out1, rst, clk, F1);
TF2 : filter generic map (1ns) port map (T_out2, rst, clk, F2);
FI1 : fifo generic map (12,16) port map (clk, rst, WR, RD1, F1, FIFO1, E(1), F(1));
FI2 : fifo generic map (12,16) port map (clk, rst, WR, RD2, F2, FIFO2, E(2), F(2));
FC1 : fifo_control port map 
(FIFO1, FIFO2, sensor, RD, F(1), F(2), E(1), E(2), RD1, RD2, WR, D);
end structure;

package TC_Components is
component register_in generic (TPD : TIME := 1 ns); 
port (T_in : in UNSIGNED(11 downto 0);
clk, rst : in STD_LOGIC; T_out : out UNSIGNED(11 downto 0));
end component;
component tconv generic (TPD : TIME := 1 ns); 
port (T_in : in UNSIGNED (7 downto 0);
        clk, rst : in STD_LOGIC; T_out : out UNSIGNED(7 downto 0));
end component;
component filter generic (TPD : TIME := 1 ns);
port (T_in : in UNSIGNED (7 downto 0);
        rst, clk : in STD_LOGIC; T_out : out UNSIGNED(7 downto 0));
end component;



component fifo generic (width:INTEGER := 12; depth : INTEGER := 16);
        port (clk, rst, push, pop  :  STD_LOGIC;
                Di :  UNSIGNED (width-1 downto 0);
                Do : out UNSIGNED (width-1 downto 0);
                empty, full : out STD_LOGIC);
end component; 
component fifo_control generic (TPD:TIME := 1 ns);
        port (D_1, D_2 : in UNSIGNED(7 downto 0);
        select : in UNSIGNED(1 downto 0); read, f1, f2, e1, e2 : in STD_LOGIC;
        r1, r2, w12 : out STD_LOGIC; D : out UNSIGNED(7 downto 0)) ; 
end component;
end;

The following testbench completes a set of reads and writes to the FIFOs:

library IEEE;
use IEEE.std_logic_1164.all; -- type STD_LOGIC
use IEEE.numeric_std.all; -- type UNSIGNED
entity test_TC is end;
architecture testbench of test_TC is
component T_Control port (T_1, T_2 : in UNSIGNED(11 downto 0);
        clk : in STD_LOGIC; sensor: in UNSIGNED( 1 downto 0) ;
        read : in STD_LOGIC; rst : in STD_LOGIC; 
        D : out UNSIGNED(7 downto 0)); end component;
signal T_1, T_2 : UNSIGNED(11 downto 0); 
signal clk, read, rst : STD_LOGIC; 
signal sensor : UNSIGNED(1 downto 0); 
signal D : UNSIGNED(7 downto 0); 
begin TT1 : T_Control port map (T_1, T_2, clk, sensor, read, rst, D);
process begin
rst <= ’0’; clk <= ’0’;
wait for 5 ns; rst <= ’1’; wait for 5 ns; rst <= ’0’; 
T_in1 <= "000000000011"; T_in2 <= "000000000111"; read <= ’0’; 
        for i in 0 to 15 loop -- fill the FIFOs
        clk <= ’0’; wait for 5 ns; clk <= ’1’; wait for 5 ns;
        end loop;
        assert (false) report "FIFOs full" severity NOTE;
        clk <= ’0’; wait for 5 ns; clk <= ’1’; wait for 5 ns;
read <= ’1’; sensor <= "01"; 
        for i in 0 to 15 loop -- empty the FIFOs
        clk <= ’0’; wait for 5ns; clk <= ’1’; wait for 5 ns;
        end loop;
        assert (false) report "FIFOs empty" severity NOTE;
        clk <= ’0’; wait for 5ns; clk <= ’1’; wait;
end process;
end;

10.17  Summary

Table 10.27 shows the essential elements of the VHDL language. Table 10.28 shows the most important
BNF definitions and their locations in this chapter. The key points covered in this chapter are as follows:

The use of an entity and an architecture 



The use of a configuration to bind entities and their architectures 
The compile, elaboration, initialization, and simulation steps 
Types, subtypes, and their use in expressions 
The logic systems based on BIT and Std_Logic_1164 types 
The use of the IEEE synthesis packages for BIT arithmetic 
Ports and port modes 
Initial values and the difference between simulation and hardware 
The difference between a signal and a variable 
The different assignment statements and the timing of updates 
The process and wait statements 

VHDL is a "wordy" language. The examples in this chapter are complete rather than code fragments. To
write VHDL "nicely," with indentation and nesting of constructs, requires a large amount of space.
Some of the VHDL code examples in this chapter are deliberately dense (with reduced indentation and
nesting), but the bold keywords help you to see the code structure. Most of the time, of course, we do
not have the luxury of bold fonts (or color) to highlight code. In this case, you should add additional
space, indentation, nesting, and comments.

TABLE 10.27    VHDL summary.

VHDL feature Example Book 93LRM

Comments -- this is a comment 10.3 13.8

Literals (fixed-value items)

12   1.0E6   ’1’   "110"   ’Z’

2#1111_1111#    "Hello world"

STRING’("110")

10.4 13.4

Identifiers

(case-insensitive, start with
letter)

a_good_name   Same   same

2_Bad   bad_   _bad   very__bad
10.4 13.3

Several basic units of code entity    architecture    configuration 10.5 1.1-1.3

Connections made through ports port ( signal in i : BIT; out o : BIT); 10.7 4.3

Default expression port (i : BIT := ’1’); 
-- i=’1’ if left open

10.7 4.3

No built-in logic-value system.

BIT and BIT_VECTOR (STD).

type BIT is (’0’, ’1’); -- predefined

 signal myArray: BIT_VECTOR (7 downto 0);
10.8 14.2

Arrays myArray(1 downto 0) <= (’0’, ’1’); 10.8 3.2.1

Two basic types of logic signals
a signal corresponds to a real wire

a variable is a memory location in RAM
10.9

4.3.1.2

4.3.1.3



Types and explicit initial/default
value signal ONE : BIT := ’1’ ; 10.9 4.3.2

Implicit initial/default value BIT’LEFT = ’0’ 10.9 4.3.2

Predefined attributes clk’EVENT, clk’STABLE 10.9.4 14.1

Sequential statements inside

processes model things that
happen one after another and
repeat

process begin

wait until alarm = ring;
eat; work; sleep;

end process;

10.10 8

Timing with wait statement
wait for 1 ns; -- not wait 1 ns

wait on light until light = green;
10.10.1 8.1

Update to signals occurs at the
end of a simulation cycle

signal <= 1; -- delta time delay

signal <= variable1 after 2 ns;
10.10.3 8.3

Update to variables is immediate variable := 1; -- immediate update 10.10.3 8.4

Processes and concurrent

statements model things that
happen at the same time

process begin rain ; end process ;

process begin sing ; end process ;

process begin dance; end process ;

10.13 9.2

IEEE Std_Logic_1164

(defines logic operators on 1164
types)

 STD_ULOGIC
, STD_LOGIC
, STD_ULOGIC_VECTOR
, and STD_LOGIC_VECTOR

type STD_ULOGIC is
(’U’,’X’,’0’,’1’,’Z’,’W’,’L’,’H’,’-’);

10.6 --

IEEE Numeric_Bit and
Numeric_Std

(defines arithmetic operators on
BIT and 1164 types)

UNSIGNED and SIGNED

X <= "10" * "01" 
-- OK with numeric pkgs.

10.12 --



TABLE 10.28    VHDL definitions.

Structure Page BNF  Structure Page BNF

alias declaration 418 10.21  next statement 429 10.32

architecture body 394 10.8  null statement 430 10.35

assertion statement 423 10.25  package declaration 398 10.11

attribute declaration 418 10.22  port interface declaration 406 10.13

block statement 438 10.37  port interface list 406 10.12

case statement 428 10.30  primary unit 393 10.5

component declaration 395 10.9  procedure call statement 427 10.28

component instantiation 444 10.42  process statement 440 10.38

concurrent statement 438 10.36  return statement 430 10.34

conditional signal assignment 442 10.40  secondary unit 393 10.6

configuration declaration 396 10.10  selected signal assignment 442 10.39

constant declaration 414 10.16  sequential statement 419 10.23

declaration 413 10.15  signal assignment statement 424 10.27

design file 393 10.4  signal declaration 414 10.17

entity declaration 394 10.7  special character 391 10.2

exit statement 430 10.33  subprogram body 416 10.20

generate statement 444 10.43  subprogram declaration 415 10.19

graphic character 391 10.1  type declaration 411 10.14

identifier 392 10.3  variable assignment statement 424 10.26

if statement 427 10.29  variable declaration 415 10.18

loop statement 429 10.31  wait statement 421 10.24

Appendix A contains more detailed definitions and technical reference material.

page

10.18  Problems

* = Difficult ** = Very difficult *** = Extremely difficult

10.1 (Hello World, 10 min.) Set up a new, empty, directory (use mkdir VHDL , for example) to run your
VHDL simulator (the exact details will depend on your computer and simulator). Copy the code below
to a file called hw_1.vhd in your VHDL directory (leave out comments to save typing). Hint: Use the
vi editor ( i inserts text, x deletes text, dd deletes a line, ESC :w writes the file, ESC :q quits) or use



cat > hw_1.vhd and type in the code (use CTRL-D to end typing) on a UNIX machine. Remember to
save in ’Text Only’ mode (Frame or MS Word) on an IBM PC or Apple Macintosh.

Analyze, elaborate, and simulate your model (include the output in your answer). Comment on how easy
or hard it was to follow the instructions to use the software and suggest improvements.

entity HW_1 is end; architecture Behave of HW_1 is
constant M : STRING := "hello, world"; signal Ch : CHARACTER := ’ ’;
begin process begin
        for i in M’RANGE loop Ch <= M(i); wait for 1 ns; end loop; wait; 
end process; end;

10.2 (Running a VHDL simulation, 20 min.) Copy the example from Section 10.1 into a file called
Counter1.vhd in your VHDL directory (leave out the comments to save typing). Complete the compile
(analyze), elaborate (build), and execute (initialize and simulate) or other equivalent steps for your
simulator. After each step list the contents of your directory VHDL and any subdirectories and files that
are created (use ls -alR on a UNIX system).

10.3 (Simulator commands, 10 min.) Make a "cheat sheet" for your simulator, listing the commands that
can be used to control simulation.

10.4 (BNF addresses, 10 min.) Create a BNF description of a name including: optional title (Prof., Dr.,
Mrs., Mr., Miss, or Ms.), optional first name and middle initials (allow up to two), and last name
(including unusual hyphenated and foreign names, such as Miss A-S. de La Salle, and Prof. John T. P.
McTavish-f Fiennes). The lowest level constructs are letter ::= a-Z , ’.’ (period) and ’-’ (hyphen).
Add BNF productions for a postal address in the form: company name, mail stop, street address, address
lines (1 to 4), and country.

10.5 (BNF e-mail, 10 min.) Create a BNF description of a valid internet e-mail address in terms of
letters, ’@’ , ’.’ , ’gov’ , ’com ’, ’org ’, and ’edu’ . Create a state diagram that "parses" an e-mail
address for validity.

10.6 (BNF equivalence) Are the following BNF productions exactly equivalent? If they are not, produce
a counterexample that shows a difference.

                term ::= factor { multiplying_operator factor }
                term ::= factor | term multiplying_operator factor

10.7 (Environment, 20 min.) Write a simple VHDL model to check and demonstrate that you can get to
the IEEE library and have the environment variables, library statements, and such correctly set up for
your simulator.

10.8 (Work, 20 min.) Write simple VHDL models to demonstrate that you can retrieve and use
previously analyzed design units from the library work and that you can also remove design units from
work . Explain how your models prove that access to work is functioning correctly.

10.9 (Packages, 60 min.) Write a simple package (use filename PackH.vhd ) and package body
(filename PackB.vhd ). Demonstrate that you can store your package (call it MyPackage ) in the library
work . Then store, move, or rename (the details will depend on your software) your package to a library
called MyLibrary in a directory called MyDir , and use its contents with a library clause ( library



MyLibrary ) and a use clause ( use MyLibrary.MyPackage.all ) in a testbench called PackTest
(filename PackT.vhd ) in another directory MyWork . You may or may not be amazed at how
complicated this can be and how poorly most software companies document this process.

10.10 (***IEEE Std 1164, 60 min.) Prior to VHDL-93 the xnor function was not available, and
therefore older versions of the std_logic_1164 library did not provide the xnor function for
STD_LOGIC types either (it was actually included but commented out). Write a simple model that checks
to see if you have the newer version of std_logic_1164 . Can you do this without crashing the
simulator?

You are an engineer on a very large project and find that your design fails to compile because your
design must use the xnor function and the library setup on your company’s system still points to the old
IEEE std_logic_1164 library, even though the new library was installed. You are apparently the first
person to realize the problem. Your company has a policy that any time a library is changed all design
units that use that library must be rebuilt from source. This might require days or weeks of work.
Explain in detail, using code, the alternative solutions. What will you recommend to your manager?

10.11 (**VHDL-93 test, 20 min.) Write a simple test to check if your simulator is a VHDL-87 or
VHDL-93 environment--without crashing the simulator.

10.12 (Declarations, 10 min.) Analyze the following changes to the code in Section 10.8 and include the
simulator output in your answers:

Uncomment the declarations for Bad100 and Bad4 in Declaration _1 .

Add the following to Constant_2:

signal wacky : wackytype (31 downto 0); -- wacky

Remove the library and use clause in Constant_2.

10.13 ( STRING type, 10 min.) Replace the write statement that prints the string " count=" in
Text(Behave) in Section 10.6.3 with the following, compile it, and explain the result:

write(L, " count=" ); -- No type qualification.

10.14 (Sequential statements, 10 min.) Uncomment the following line in Wait_1(Behave) in
Section 10.10, analyze the code, and explain the result:

wait on x(1 to v); -- v is a variable.

10.15 (VHDL logical operators, 10 min.)

Explain the problem with the following VHDL statement:

Z <= A nand B nand C;

Explain why this problem does not occur with this statement:

Z <= A and B and C;



What can you say about the logical operators: and , or , nand , nor , xnor , xor ?

Is the following code legal?

Z <= A and B or C;

10.16 (*Initialization, 45 min.) Consider the following code:

entity DFF_Plain is port (Clk, D : in BIT; Q : out BIT); end;
architecture Bad of DFF_Plain is begin process (Clk) begin
        if Clk = ’0’ and Clk’EVENT then Q <= D after 1 ns; end if;
end process; end;

Analyze and simulate this model using a testbench.

Rewrite architecture Bad using an equivalent process including a wait statement. Simulate this
equivalent model and confirm the behaviors are identical.

What is the behavior of the output Q during initial execution of the process?

Why does this happen?

Why does this not happen with the following code:

architecture Good of DFF_Plain is
begin process begin wait until Clk = ’0’; Q <= D after 1 ns;
end process; end;

10.17 (Initial and default values, 20 min.) Use code examples to explain the difference between: default
expression, default value, implicit default value, initial value, initial value expression, and default initial
value.

10.18 (Enumeration types, 20 min.) Explain the analysis results for the following:

type MVL4 is (’X’, ’0’, ’1’, ’Z’); signal test : MVL4;
process begin
        test <= 1; test <= Z; test <= z; test <= ’1’; test <= ’Z’;
end process;

Alter the type declaration to the following, analyze your code again, and comment:

type Mixed4 is (X , ’0’, ’1’, Z);

10.19 (Type declarations, 10 min.) Correct these declarations:

type BadArray is array (0 to 7) of BIT_VECTOR;
type Byte is array (NATURAL range 7 downto 0) of BIT;
subtype BadNibble is Byte(3 downto 0);
type BadByte is array (range 7 downto 0) of BIT;

10.20 (Procedure parameters, 10 min.) Analyze the following package; explain and correct the error.
Finally, build a testbench to check your solution.



package And_Pkg_Bad is procedure V_And(a, b : BIT; c: out BIT); end;
package body And_Pkg_Bad is
procedure V_And(a,b : BIT;c : out BIT) is begin c <= a and b;end;
end And_Pkg_Bad;

10.21 (Type checking, 20 min.) Test the following code and explain the results:

type T is INTEGER range 0 to 32; variable a: T;
a := (16 + 17) - 12; a := 16 - 12 + 17; a := 16 + (17 - 12);

10.22 (Debugging VHDL code, 30 min.) Find and correct the errors in the following code. Create a
testbench for your code to check that it works correctly.

entity UpDownCount_Bad is 
port(clock, reset, up: STD_LOGIC; D: STD_LOGIC_VECTOR (7 to 0));
end UpDownCount_Bad;
architecture Behave of UpDownCount_Bad is
begin process (clock, reset, up); begin 
if (reset = ’0’) then D <= ’0000000’; 
elseif (rising_edge(clock)) then
if (up = 1) D <= D+1; else D <= D-1; end if;
end if; end process; end Behave;

10.23 (Subprograms, 20 min.) Write and test subprograms for these declarations:

function Is_X_Zero (signal X : in BIT) return BIT;
procedure Is_A_Eq_B (signal A, B : BIT; signal Y : out BIT);

10.24 (Simulator error messages, 10 min.) Analyze and attempt to simulate Arithmetic_2(Behave) from
Section 10.12 and compare the error message you receive with that from the MTI simulator (not all
simulators are as informative). There are no standards for error messages.

10.25 (Exhaustive property of case statement, 30 min.) Write and simulate a testbench for the state
machine of Table 10.8 and include your results. Is every state transition tested by your program and is
every transition covered by an assignment statement in the code? (Hint: Think very carefully.) Repeat
this exercise for the state machine in Section 10.10.6.

10.26 (Default values for inputs, 20 min.) Replace the interface declaration for entity Half_Adder in
Section 10.5 with the following (to remove the default values):

port (X, Y: in BIT ; Sum, Cout: out BIT);

Attempt to compile, elaborate, and simulate configuration Simplest (the other entities needed, AndGate
and XorGate , must already be in work or in the same file). You should get an error at some stage
(different systems find this error at different points--just because an entity compiles, that does not mean
it is error-free).

The LRM says "... A port of mode in may be unconnected ...only if its declaration includes a default
expression..." [VHDL 93LRM1.1.1.2].

We face a dilemma here. If we do not drive inputs with test signals and leave an input port unconnected,
we can compile the model (since it is syntactically correct) but the model is not semantically correct. On
the other hand, if we give the inputs default values, we might accidentally forget to make a connection



and not notice.

10.27 (Adder generation, 10 min.) Draw the schematic for Adder_1(Structure) of Section 10.13.7,
labeling each instance with the VHDL instance name.

10.28 (Generate statement, 20 min.) Draw a schematic corresponding to the following code (label the
cells with their instance names):

B1: block begin L1 : C port map (T, B, A(0), B(0)) ;
L2: for i in 1 to 3 generate L3 : for j in 1 to 3 generate
L4: if i+j > 4 generate L5: C port map (A(i-1), B(j-1), A(i), B(j)) ;
end generate; end generate; end generate;
L6: for i in 1 to 3 generate L7: for j in 1 to 3 generate
L8: if i+j < 4 generate L9: C port map (A(i+1), B(j+1), A(i), B(j)) ;
end generate; end generate; end generate;
end block B1;

Rewrite the code without generate statements. How would you prove that your code really is exactly
equivalent to the original?

10.29 (Case statement, 20 min.) Create a package ( my_equal ) that overloads the equality operator so
that ’X’ = ’0’ and ’X’ = ’1’ are both TRUE . Test your package. Simulate the following design unit
and explain the result.

entity Case_1 is end; architecture Behave of Case_1 is
signal r : BIT; use work.my_equal.all;
begin process variable twobit:STD_LOGIC_VECTOR(1 to 2); begin 
        twobit := "X0";
        case twobit is 
                when "10" => r <= ’1’;
                when "00" => r <= ’1’; 
                when others => r <= ’0’;
        end case; wait;
end process; end;

10.30 (State machine) Create a testbench for the state machine of Section 10.2.5.

10.31 (Mealy state machine, 60 min.) Rewrite the state machine of Section 10.2.5 as a Mealy state
machine (the outputs depend on the inputs and on the current state).

10.32 (Gate-level D flip-flop, 30 min.) Draw the schematic for the following D flip-flop model. Create a
testbench (check for correct operation with combinations of Clear , Preset , Clock , and Data ). Have
you covered all possible modes of operation? Justify your answer of yes or no.

architecture RTL of DFF_To_Test is
signal A, B, C, D, QI, QBarI : BIT; begin 
A <= not (Preset and D and B) after 1 ns;
B <= not (A and Clear and Clock) after 1 ns;
C <= not (B and Clock and D) after 1 ns;
D <= not (C and Clear and Data) after 1 ns;
QI <= not (Preset and B and QBarI) after 1 ns;
QBarI <= not (QI and Clear and C) after 1 ns;
Q <= QI; QBar <= QBarI;
end;



10.33 (Flip-flop model, 20 min.) Add an asynchronous active-low preset to the D flip-flop model of
Table 10.3. Generate a testbench that includes interaction of the preset and clear inputs. What issue do
you face and how did you solve it?

10.34 (Register, 45 min.) Design a testbench for the register of Table 10.4. Adapt the 8-bit register
design to a 4-bit version with the following interface declaration:

entity Reg4 is port (D : in  STD_LOGIC_VECTOR(7 downto 0);
Clk,Pre,Clr : in STD_LOGIC;Q,QB : out STD_LOGIC_VECTOR(7 downto 0));
end Reg8;

Create a testbench for your 4-bit register with the following component declaration:

 component DFF
port(Preset,Clear,Clock,Data:STD_LOGIC;Q,QBar:out STD_LOGIC_VECTOR);
end component;

10.35 (*Conversion functions, 30 min.) Write a conversion function from 
NATURAL to STD_LOGIC_VECTOR using the following declaration:

function Convert (N, L: NATURAL) return STD_LOGIC_VECTOR;
-- N is NATURAL, L is length of STD_LOGIC_VECTOR

Write a similar conversion function from STD_LOGIC_VECTOR to NATURAL:

function Convert (B: STD_LOGIC_VECTOR) return NATURAL;

Create a testbench to test your functions by including them in a package.

10.36 (Clock procedure, 20 min.) Design a clock procedure for a two-phase clock (C1, C2) with variable
high times ( HT1 , HT2 ) and low times ( LT1 , LT2 ) and the following interface. Include your procedure
in a package and write a model to test it.

procedure Clock (C1, C2 : out STD_LOGIC; HT1, HT2, LT1, LT2 : TIME);

10.37 (Random number, 20 min.) Design a testbench for the following procedure:

procedure uniform (seed : inout INTEGER range 0 to 15) is
        variable x : INTEGER;
        begin x := (seed*11) + 7; seed := x mod 16;
end uniform;

10.38 (Full-adder, 30 min.) Design and test a behavioral model of a full adder with the following
interface:

entity FA is port (X, Y, Cin : STD_LOGIC; Cout, Sum : out STD_LOGIC);
end;

Repeat the exercise for inputs and outputs of type UNSIGNED .

10.39 (8-bit adder testbench, 60 min.) Write out the code corresponding to the generate statements of
Adder_1 ( Structure ) in Section 10.13.7. Write a testbench to check your adder. What problems do
you encounter? How thorough do you believe your tests are?



10.40 (Shift-register testbench, 60 min.) Design a testbench for the shift register of Table 10.4. Convert
this model to use STD_LOGIC types with the following interface:

entity ShiftN is 
port (CLK, CLR, LD, SH, DIR : STD_LOGIC; 
        D : STD_LOGIC_VECTOR; Q : out STD_LOGIC_VECTOR);
end;

10.41 (Multiplier, 60 min.) Design and test a multiplier with the following interface:

entity Mult8 is 
port (A, B : STD_LOGIC_VECTOR(3 downto 0); 
Start, CLK, Reset : in STD_LOGIC;
Result : out STD_LOGIC_VECTOR(7 downto 0); Done : out BIT);
end;

Create testbench code to check your model.

Catalog each compile step with the syntax errors as you debug your code.

Include a listing of the first code you write together with the final version.

An interesting class project is to collect statistics from other students working on this problem and create
a table showing the types and frequency of syntax errors made with each compile step, and the number
of compile steps required. Does this information suggest ways that you could improve the compiler, or
suggest a new type of tool to use when writing VHDL?

10.42 (Port maps, 5 min.) What is wrong with this VHDL statement?

U1 : nand2 port map (a <= set, b <= qb, c <= q);

10.43 (DRIVING_VALUE, 15 min.) Use the VHDL-93 attribute Clock’DRIVING_VALUE to rewrite
the following clock generator model without using a temporary variable.

entity ClockGen_2 is port (Clock : out BIT); end;
architecture Behave of ClockGen_2 is
begin process variable Temp : BIT := ’1’; begin
        Temp := not Temp ; Clock <= Temp after 10 ns; wait for 10 ns;
        if (now > 100 ns) then wait; end if; end process;
end;

10.44 (Records, 15 min.) Write an architecture (based on the following skeleton) that uses the record
structure shown:

entity Test_Record_1 is end; architecture Behave of Test_Record_1 is
begin process type Coordinate is record X, Y : INTEGER; end record;
-- a record declaration for an attribute declaration:
attribute Location:Coordinate; -- an attribute declaration
begin wait; end process; end Behave;

10.45 (**Communication between processes, 30 min.) Explain and correct the problem with the
following skeleton code:



variable v1 : INTEGER := 1; process begin v1 := v1+3; wait; end process;
process variable v2 : INTEGER := 2; begin v2 := v1  ; wait; end process;

10.46 (*Resolution, 30 min.) Explain and correct the problems with the following:

entity R_Bad_1 is port (i : in BIT; o out BIT); end;
architecture Behave of R_Bad_1 is
begin o <= not i after 1 ns; o <= i after 2 ns; end;

10.47 (*Inputs, 30 min.) Analyze the following and explain the result:

entity And2 is port (A1, A2: in BIT; ZN: out BIT); end;
architecture Simple of And2 is begin ZN <= A1 and A2; end;
entity Input_Bad_1 is end; architecture Netlist of Input_Bad_1 is
component And2 port (A1, A2 : in BIT; ZN : out BIT); end component;
signal X, Z : BIT begin G1 : And2 port map (X, X, Z); end;

10.48 (Association, 15 min.) Analyze the following and explain the problem:

entity And2 is port (A1, A2 : in BIT; ZN : out BIT); end;
architecture Simple of And2 is begin ZN <= A1 and A2; end;
entity Assoc_Bad_1 is port (signal X, Y : in BIT; Z : out BIT); end;
architecture Netlist of Assoc_Bad_1 is
component And2 port (A1, A2 : in BIT; ZN : out BIT); end component;
begin
G1: And2 port map (X, Y, Z);
G2: And2 port map (A2 => Y, ZN => Z, A1 => X);
G3: And2 port map (X, ZN => Z, A2 => Y);
end;

10.49 (Modes, 30 min.) Analyze and explain the errors in the following:

entity And2 is port (A1, A2 : in BIT; ZN : out BIT); end;
architecture Simple of And2 is begin ZN <= A1 and A2; end;
entity Mode_Bad_1 is port (X : in BIT; Y : out BIT; Z : inout BIT); end;
architecture Netlist of Mode_Bad_1 is
component And2 port (A1, A2 : in BIT; ZN : out BIT); end component;
begin G1 : And2 port map (X, Y, Z); end;
entity Mode_Bad_2 is port (X : in BIT; Y : out BIT; Z : inout BIT); end;
architecture Netlist of Mode_Bad_1 is
component And2 port (A1, A2 : in BIT; ZN : inout BIT); end component;
begin G1 : And2 port map (X, Y, Z); end;

10.50 (*Mode association, 60 min.) Analyze and explain the errors in the following code. The number
of errors, types of error, and the information in the error messages given by different simulators vary
tremendously in this area.

entity Allmode is port 
(I : in BIT; O : out BIT; IO : inout BIT; B : buffer BIT); 
end;
architecture Simple of Allmode is begin O<=I; IO<=I; B<=I; end;
entity Mode_1 is port 
(I : in BIT; O : out BIT; IO : inout BIT; B : buffer BIT); 
end;
architecture Netlist of Mode_1 is
component Allmode port 
(I : in BIT; O : out BIT; IO : inout BIT; B : buffer BIT); end component;
begin 



G1:Allmode port map (I , O , IO, B ); 
G2:Allmode port map (O , IO, B , I ); 
G3:Allmode port map (IO, B , I , O ); 
G4:Allmode port map (B , I , O , IO); 
end;

10.51 (**Declarations, 60 min.) Write a tutorial (approximately two pages of text, five pages with code)
with examples explaining the difference between: a component declaration, a component configuration,
a configuration declaration, a configuration specification, and a block configuration.

10.52 (**Guards and guarded signals, 60 min.) Write some simple models to illustrate the use of guards,
guarded signals, and the disconnect statement. Include an experiment that shows and explains the use of
the implicit signal GUARD in assignment statements.

10.53 (** Std_logic_1164 , 120 min.) Write a short (two pages of text) tutorial, with (tested) code
examples, explaining the std_logic_1164 types, their default values, the difference between the
’ulogic’ and ’logic’ types, and their vector forms. Include an example that shows and explains the
problem of connecting a std_logic_vector to a std_ulogic_vector.

10.54 (Data swap, 20 min.) Consider the following code:

library ieee; use ieee.std_logic_1164.all;
package config is
type type1 is record
f1 : std_logic_vector(31 downto 0); f2 : std_logic_vector(3 downto 0);
end record;
type type2 is record
f1 : std_logic_vector(31 downto 0); f2 : std_logic_vector(3 downto 0);
end record;
end config;
library ieee; use ieee.STD_LOGIC_1164.all; use work.config.all;
entity Swap_1 is
port (Data1 : type1; Data2 : type2; sel : STD_LOGIC;
Data1Swap : out type1; Data2Swap : out type2); end Swap_1;
architecture Behave of Swap_1 is begin
Swap: process (Data1, Data2, sel) begin case sel is
when ’0’ => Data1Swap <= Data1; Data2Swap <= Data2;
when others => Data1Swap <= Data2; Data2Swap <= Data1;
end case; end process Swap; end Behave;

Compile this code. What is the problem? Suggest a fix. Now write a testbench and test your code. Have
you considered all possibilities?

10.55 (***RTL, 30 min.) "RTL stands for register-transfer level. ...when referencing VHDL, the term
means that the description includes only concurrent signal assignment statements and possibly block
statements. In particular, VHDL data flow descriptions explicitly do not contain either process
statements (which describe behavior) or component instantiation statements (which describe structure)"
(Dr. VHDL from VHDL International).

With your knowledge of process statements and components, comment on Dr. VHDL’s explanation.

In less than 100 words offer your own definition of the difference between RTL, data flow, behavioral,
and structural models.



10.56 (*Operators mod and rem , 20 min.) Confirm and explain the following:

        i1 := (-12)                                     rem             7;                                      -- i1 = -5
        i2 := 12                                        rem             (-7);                                   -- i2 =  5
        i3 := (12)                                      rem             (-7);                                   -- i3 = -5
        i4 := 12                                        mod             7;                                      -- i4 =  5
        i5 := (-12)                                     mod             7;                                      -- i5 =  2
        i6 := 12                                        mod             (-7);                                   -- i6 = -2
        i7 := (12)                                      mod             (-7);                                   -- i7 = -5

Evaluate -5 rem 2 and explain the result.

10.57 (***Event and stable, 60 min.) Investigate the differences between clk’EVENT and clk’STABLE .
Write a minitutorial (in the form of a "cheat sheet") with examples showing the differences and potential
dangers of using clk’STABLE .

10.58 (PREP benchmark #2, 60 min.) The following code models a benchmark circuit used by PREP to
measure the capacity of FPGAs. Rewrite the concurrent signal assignment statements (labeled mux and
comparator) as equivalent processes. Draw a datapath schematic corresponding to PREP2(Behave_1).
Write a testbench for the model. Finally (for extra credit) rewrite the model and testbench to use
STD_LOGIC instead of BIT types.

library ieee; use ieee.STD_LOGIC_1164.all;
use ieee.NUMERIC_BIT.all; use ieee.NUMERIC_STD.all;
entity PREP2 is 
port(CLK,Reset,Sel,Ldli,Ldhi : BIT; D1,D2 : STD_LOGIC_VECTOR(7 downto 0);
        DQ:out STD_LOGIC_VECTOR(7 downto 0));
end;
architecture Behave_1 of PREP2 is 
signal EQ : BIT; signal y,lo,hi,Q_i : STD_LOGIC_VECTOR(7 downto 0);
begin 
outputDriver: Q <= Q_i;
mux: with Sel select y <= hi when ’0’, D1 when ’1’;
comparator: EQ <= ’1’ when Q_i = lo else ’0’;
        register: process(Reset, CLK) begin
                if Reset = ’1’ then hi <= "00000000"; lo <= "00000000";
                elsif CLK = ’1’ and CLK’EVENT then
                        if Ldhi=’1’ then hi<=D2;end if;if Ldlo=’1’ then lo<=D2;end if
                end if;
        end process register;
        counter: process(Reset, CLK) begin
                if Reset = ’1’ then Q_i <= "00000000";
                elsif CLK = ’1’ and CLK’EVENT then
                        if EQ = ’1’ then Q_i <= y;
                        elsif EQ = ’0’ then Q_i  <= Q_i + "00000001";
                        end if;
                end if;
        end process counter;
end;

10.59 (PREP #3, state machine) Draw the state diagram for the following PREP benchmark (see
Problem 10.58). Is this a Mealy or Moore machine? Write a testbench and test this code.

library ieee; use ieee.STD_LOGIC_1164.all; 
entity prep3_1 is port(Clk, Reset: STD_LOGIC;
        I : STD_LOGIC_VECTOR(7 downto 0); O : out STD_LOGIC_VECTOR(7 downto 0));
end prep3_1;



architecture Behave of prep3_1 is
        type STATE_TYPE is (sX,s0,sa,sb,sc,sd,se,sf,sg);
        signal state : STATE_TYPE; signal Oi : STD_LOGIC_VECTOR(7 downto 0);
begin 
        O <= Oi;
        process (Reset, Clk) begin 
                if (Reset = ’1’) then state <= s0; Oi <= (others => ’0’);
                elsif rising_edge(Clk) then 
                        case state is 
                                when s0 =>
                                        if (I = X"3c") then state <= sa; Oi <= X"82"; 
                                        else state <= s0; Oi <= (others => ’0’);
                                        end if;
                                when sa =>
                                        if (I = X"2A") then state <= sc; Oi <= X"40"; 
                                        elsif (I = X"1F") then state <= sb; Oi <= X"20"; 
                                        else state <= sa; Oi <= X"04"; 
                                        end if;
                                when sb =>
                                        if (I = X"AA") then state <= se; Oi <= X"11";
                                        else state <= sf; Oi <= X"30"; 
                                        end if;
                                when sc => state <= sd; Oi <= X"08"; 
                                when sd => state <= sg; Oi <= X"80"; 
                                when se => state <= s0; Oi <= X"40"; 
                                when sf => state <= sg; Oi <= X"02"; 
                                when sg => state <= s0; Oi <= X"01"; 
                                when others => state <= sX; Oi <= (others => ’X’); 
                        end case;
                end if;
        end process;
end;

10.60 (Edge detection, 30 min) Explain the construction of the IEEE 1164 function to detect the rising
edge of a signal, rising_edge(s) . List all the changes in signal s that correspond to a rising edge.

 function rising_edge (signal s : STD_ULOGIC) return BOOLEAN is
        begin return 
        (s’EVENT and (To_X01(s) = ’1’) and (To_X01(s’LAST_VALUE) = ’0’)); end;

10.61 (*Real, 10 min.) Determine the smallest real in your VHDL environment.

10.62 (*Stop, 30 min.) How many ways are there to stop a VHDL simulator?

10.63 (*Arithmetic package, 60 min.) Write a function for an arithmetic package to subtract two’s
complement numbers. Create a test bench to check your function. Your declarations in the package
header should look like this:

type TC is array (INTEGER range <>) of STD_LOGIC;
function "-"(L : TC; R : TC) return TC;

10.64 (***Reading documentation, hours) There are a few gray areas in the interpretation of the
VHDL-87 LRM some of which were clarified in the VHDL-93 revision. One VHDL system has a
"compatibility mode" that allows alternative interpretations. For each of the following "issues" taken
from the actual tool documentation try to interpret what was meant, determine the interpretation taken
by your own software, and then rewrite the explanation clearly using examples.



* "Unassociated variable and signal parameters. Compatibility mode allows variable and signal
parameters to subprograms to be unassociated if they have a default value. Otherwise, an error is
generated."

Example answer: Consider the following code:

package Util_2 is
procedure C(signal Clk : out BIT; signal P : TIME := 10 ns); 
end Util_2;
package body Util_2 is
procedure C(signal Clk : out BIT; signal P : TIME := 10 ns) is
begin loop Clk <= ’1’ after P/2, ’0’ after P;
wait for P; end loop; end; end Util_2;
entity Test_Compatibility_1 is end; use work.Util_2.all;
architecture Behave of Test_Compatibility_1 is 
signal v,w,x,y,z : BIT; signal s : TIME := 5 ns;
begin process variable v : TIME := 5 ns; begin
C(v, s);                                                        -- parameter s is OK since P is declared as signal 
-- C(w, v);                                                     -- would be OK if P is declared as variable instead
-- C(x, 5 ns);                                                  -- would be OK if P is declared as constant instead
-- C(y);                                                        -- unassociated, an error if P is signal or variable
-- C(z,open);                                                   -- open, an error if P is signal or variable
end process; end;

The Compass Scout simulator (which does not have a compatibility mode) generates an error during
analysis if a signal or variable subprogram parameter is open or unassociated (a constant subprogram
parameter may be unassociated or open).

* "Allow others in an aggregate within a record aggregate. The LRM [7.3.2.2] defines nine situations
where others may appear in an aggregate. In compatibility mode, a tenth case is added. In this case,
others is allowed in an aggregate that appears as an element association in a record element."

* " BIT’(’1’) parsed as BIT ’ (’1’) . The tick ( ’ ) character is being used twice in this example. In
the first case as an attribute indicator, in the second case, to form a character literal. Without the
compatibility option, the analyzer adopts a strict interpretation of the LRM, and without white space
around the first tick, the fragment is parsed as BIT ’(’1’) , that is, the left parenthesis ( ’(’ ) is the
character literal."

** "Generate statement declarative region. Generate statements form their own declarative region. In
compatibility mode, configuration specifications will apply to items being instantiated within a generate
statement."

** "Allow type conversion functions on open parameters. If a parameter is specified as open, it indicates
a parameter without an explicit association. In such cases, the presence of a type conversion function is
meaningless. Compatibility mode allows the type conversion functions."

*** "Entity class flexibility. Section [3.1.2] of the LRM defines the process of creating a new integer
type. The type name given is actually assigned to a subtype name, related to an anonymous base type.
This implies that the entity class used during an attribute specification [LRM 5.1] should indicate
subtype, rather than type. Because the supplied declaration was type rather than subtype, compatibility
mode allows type."



*** "Allowing declarations beyond an all/others specification. Section [5.1] of the LRM states that the
first occurrence of the reserved word all or others in an attribute specification terminates the
declaration of the related entity class. The LRM declares that the entity/architecture and
package/package body library units form single declaration regions [LRM 10.1] that are the
concatenation of the two individual library declarative regions. For example, if a signal attribute
specification with all or others was specified in the entity, it would be impossible to declare a signal
in the architecture. In compatibility mode, this LRM limitation is removed."

*** "User-defined attributes on overloaded functions. In compatibility mode, user-defined attributes are
allowed to be associated with overloaded functions. Note: Even in compatibility mode, there is no way
to retrieve the different attributes."

10.65 (*1076 interpretations, 30 min.) In a DAC paper, the author writes: ‘It was experienced that
(company R) might have interpreted IEEE 1076 differently than (company S) did, e.g. concatenations
(&) are not allowed in "case selector" expressions for (company S).’ Can you use concatenation in your
VHDL tool for either the expression or choices for a case statement?

10.66 (**Interface declarations, 15 min.) Analyze the following and comment:

entity Interface_1 is 
        generic (I : INTEGER; J : INTEGER := I; K, L : INTEGER);
        port (A : BIT_VECTOR; B : BIT_VECTOR(A’RANGE); C : BIT_VECTOR (K to L)); 
        procedure X(P, Q : INTEGER; R : INTEGER range P to Q);
        procedure Y(S : INTEGER range K to L); 
end Interface_1;

10.67 (**Wait statement, 10 min.) Construct the sensitivity set and thus the sensitivity list for the
following wait statement (that is, rewrite the wait statement in the form wait on sensitivity_list
until condition ).

entity Complex_Wait is end;
architecture Behave of Complex_Wait is
        type A is array (1 to 5) of BOOLEAN; 
        function F (P : BOOLEAN) return BOOLEAN;
        signal S : A; signal i, j : INTEGER range 1 to 5;
        begin process begin
                wait until F(S(3)) and (S(i) or S(j));
        end process;
end;

10.68 (**Shared variables, 20 min.) Investigate the following code and comment:

architecture Behave of Shared_1 is 
subtype S is INTEGER range 0 to 1; shared variable C : S := 0; begin
process begin C := C + 1; wait; end process;
process begin C := C - 1; wait; end process;
end;

10.69 (Undocumented code and ranges, 20 min.) Explain the purpose of the following function (part of a
package from a well-known synthesis company) with a parameter of type SIGNED. Write a testbench to
check your explanation. Investigate what happens when you call this function with a string-literal
argument, for example with the statement X <= IM("11100"). What is the problem and why does it
happen? Rewrite the code, including documentation, to avoid this problem.



type SIGNED is array (NATURAL range <> ) of BIT;
function IM (L : SIGNED) return INTEGER is variable M : INTEGER;
begin M := L’RIGHT-1;
        for i in L’LEFT-1 downto L’RIGHT loop
                if (L(i) = (not L(L’LEFT))) then M := i; exit; end if;
        end loop; return M;
end;

10.70 (Timing parameters, 20 min.) Write a model and a testbench for a two-input AND gate with
separate rising (tpLH) and falling (tpHL) delays using the following interface:

entity And_Process is
generic (tpLH, tpHL : TIME); port (a, b : BIT; z : out BIT) end;

10.71 (Passive code in entities, 30 min.) Write a procedure (CheckTiming, part of a package
Timing_Pkg ) to check that two timing parameters (tPLH and tPHL) are both greater than zero. Include
this procedure in a two-input AND gate model ( And_Process ). Write a testbench to show your
procedure and gate model both work. Rewrite the entity for And_Process to include the timing check as
part of the entity declaration. You are allowed to include passive code (no assignments to signals and so
on) directly in each entity. This avoids having to include the timing checks in each architecture.

10.72 (Buried code, 30 min.) Some companies bury instructions to the software within their packages.
Here is an example of part of the arithmetic package from an imaginary company called SissyN:

function UN_plus(A, B : UN) return UN is
variable CRY : STD_ULOGIC; variable X,SUM : UN (A’LEFT downto 0);
-- pragma map_to_operator ADD_UNS_OP
-- pragma type_function LEFT_UN_ARG
-- pragma return_port_name Z
begin 
-- sissyn synthesis_off
if (A(A’LEFT) = ’X’ or B(B’LEFT) = ’X’) then SUM := (others => ’X’); 
return(SUM);
end if;
-- sissyn synthesis_on
CRY := ’0’; X := B;
for i in 0 to A’LEFT loop
SUM(i) := A(i) xor X(i) xor carry;
CRY := (A(i) and X(i)) or (A(i) and CRY) or (CRY and X(i));
end loop; return SUM;
end;

Explain what this function does. Can you now hazard a guess at what each of the comments means?
What are the repercussions of using comments in this fashion?

10.73 (*Deferred constants, 15 min.) "If the assignment symbol ’:=’ followed by an expression is not
present in a constant declaration, then the declaration declares a deferred constant. Such a constant
declaration may only appear in a package declaration. The corresponding full constant declaration,
which defines the value of the constant, must appear in the body of the package" [VHDL
93LRM4.3.1.1].

package Constant is constant s1, s2 : BIT_VECTOR; end Constant;
package body Constant is
constant s0 : BIT_VECTOR := "00"; constant s1 : BIT_VECTOR := "01";
end Constant;



It is tempting to use deferred constants to hide information. However, there are problems with this
approach. Analyze the following code, explain the results, and correct the problems:

entity Deferred_1 is end; architecture Behave of Deferred_1 is
use work.all; signal y,i1,i2 : INTEGER; signal sel : INTEGER range 0 to 1;
begin with sel select y <= i1 when s0, i2 when s1; end;

10.74 (***Viterbi code, days) Convert the Verilog model of the Viterbi decoder in Chapter 11 to
VHDL. This problem is tedious without the help of some sort of Verilog to VHDL conversion process.
There are two main approaches to this problem. The first uses a synthesis tool to read the behavioral
Verilog and write structural VHDL (the Compass ASIC Synthesizer can do this, for example). The
second approach uses conversion programs (Alternative System Concepts Inc. at
http://www.ascinc.com is one source). Some of these companies allow you to e-mail code to them
and they will automatically return a translated version.

10.75 (*Wait statement, 30 min.) Rewrite the code below using a single wait statement and write a
testbench to prove that both approaches are exactly equivalent:

entity Wait_Exit is port (Clk : in BIT); end;
architecture Behave of Wait_Exit is
        begin process begin
                loop wait on Clk; exit when Clk = ’1’; end loop;
        end process;
end;

10.76 (Expressions, 10 min.) Explain and correct the problems with the following:

variable b : BOOLEAN; b := "00" < "11";
variable bv8 : BIT_VECTOR (7 downto 0) := "1000_0000";

10.77 (Combinational logic using case statement, 10 min.) A Verilog user suggests the following
method to model combinational logic. What are the problems with this approach? Can you get it to
work?

entity AndCase is port (a, b : BIT; y : out BIT); end;
architecture Behave of AndCase is begin process (a , b) begin
        case a & b is 
         when ’1’&’1’ => y <= ’1’; when others => y <= ’0’;
        end case;
end process; end;

10.78 (*Generics and back-annotation, 60 min.)

Construct design entities And_3(Behave), a two-input AND gate, and Xor_3(Behave) , a two-input
XOR gate. Include generic constants to model the propagation delay from each input to the output
separately. Use the following entity declaration for And_3:

entity And_3 is port (I1, I2 : BIT; O : out BIT);
        generic (I1toO, I2toO : DELAY_LENGTH := 0.4 ns); end;

Create and test a package, P_1, that contains And_3 and Xor_3 as components.

Create and test a design entity Half_Adder_3 (Structure_3) that uses P_1, with the following



interface:

entity Half_Adder_3 is port (X, Y : BIT; Sum, Carry : out BIT); end;

Modify and test the architecture Structure_3 for Half_Adder_3 so that you can use the following
configuration:

configuration Structure_3 of Half_Adder_3 is
for Structure_3
for L1 : XOR generic map (0.66 ns,0.69 ns); end for; 
for L2 : AND generic map (0.5 ns, 0.6 ns) port map (I2 => HI); end for;
end for; end;

10.79 (SNUG’95, *60 min.) In 1995 John Cooley organized a contest between VHDL and Verilog for
ASIC designers. The goal was to design the fastest 9-bit counter in under one hour using Synopsys
synthesis tools and an LSI Logic vendor technology library. The VHDL interface is as follows:

library ieee; use ieee.std_logic_1164.all;
-- use ieee.std_logic_arith.all; -- substitute your package here
entity counter is port (
data_in    : in std_logic_vector(8 downto 0);
up         : in std_logic;
down       : in std_logic;
clock      : in std_logic;
count_out  : inout std_logic_vector(8 downto 0);
carry_out  : out std_logic;
borrow_out : out std_logic;
parity_out : out std_logic ); end counter;
architecture example of counter is begin
-- insert your design here 
end example;

The counter is positive-edge triggered, counts up with up = ’1’ and down with down = ’1’ . The
contestants had the advantage of a predefined testbench with a set of test vectors, you do not. Design a
model for the counter and a testbench. How confident are you that you have thoroughly tested your
model? (In the real contest none of the VHDL contestants managed to even complete a working design
in under one hour. In addition, the VHDL experts that had designed the testbench omitted a test case for
one of the design specifications.)

10.80 (*A test procedure, 45 min.) Write a procedure all (for a package test ) that serially generates
all possible input values for a signal spaced in time by a delay, dly . Use the following interface:

library ieee; use ieee.std_logic_1164.all; package test is
procedure all (signal SLV : out STD_LOGIC_VECTOR; dly : in TIME);
end package test ;

10.81 (Direct instantiation, 20 min.) Write an architecture for a full-adder, entity Full_Adder_2, that
directly instantiates units And_2(Behave) and Xor_2(Behave). This is only possible in a VHDL-93
environment.

entity And_2 is port (i1, i2 : BIT; y : out BIT); end;
entity Xor_2 is port (i1, i2 : BIT; y : out BIT); end;
entity Full_Adder_2 is port (a, b, c : BIT ; sum, cout : out BIT); end;

10.82 (**Shift operators for 1164, 60 min.) Write a package body to implement the VHDL-93 shift



operators, sll and srl, for the type STD_LOGIC_VECTOR. Use the following package header:

package 1164_shift is
function "sll"(x : STD_LOGIC_VECTOR; n : INTEGER)
        return STD_LOGIC_VECTOR;
function "srl"(x : STD_LOGIC_VECTOR; n : INTEGER)
        return STD_LOGIC_VECTOR;
end package 1164_shift;

10.83 (**VHDL wait statement, 60 min.) What is the problem with the following VHDL code? Hint:
You may need to consult the VHDL LRM.

procedure p is begin wait on b; end;
process (a) is begin procedure p; end process;

10.84 (**Null range, 45 min.) A range such as 1 to -1 or 0 downto 1 is a null range ( 0 to 0 is a
legal range). Write a one-page summary on null ranges, including code examples. Is a null range treated
as an ascending or descending range?

10.85 (**Loops, 45 min.) Investigate the following issues with loops, including code examples and the
results of analysis and simulation:

Try to alter the loop parameter within a loop. What happens?

What is the type of the loop parameter?

Can the condition inside a loop depend on a loop parameter?

What happens in a for loop if the range is null?

Can you pass a loop parameter out of a procedure as a procedure parameter?

10.86 (Signals and variables, 30 min.) Write a summary on signals and variables, including code
examples.

10.87 (Type conversion, 60 min.) There are some very subtle rules involving type conversion, [VHDL
93LRM7.3.5]. Does the following work? Explain the type conversion rules.

BV <= BIT_VECTOR("1111");

10.19  Bibliography

The definitive reference guide to VHDL is the IEEE VHDL LRM [IEEE, 1076-1993]. The LRM is
initially difficult to read because it is concise and precise (the LRM is intended for tool builders and
experienced tool users, not as a tutorial). The LRM does form a useful reference--as does a dictionary
for serious users of any language. You might think of the LRM as a legal contract between you and the



company that sells you software that is compliant with the standard. VHDL software uses the
terminology of the LRM for error messages, so it is necessary to understand the terms and definitions of
the LRM. The WAVES standard [IEEE 1029.1-1991] deals with the problems of interfacing VHDL
testbenches to testers.

VHDL International maintains VIUF (VHDL International Users’ Forum) Internet Services (
http:/www.vhdl.org ) and links to other groups working on VHDL including the IEEE synthesis
packages, IEEE WAVES packages, and IEEE VITAL packages (see also Appendix A).

The frequently asked questions (FAQ) list for the VHDL newsgroup comp.lang.vhdl is a useful
starting point (the list is archived at gopher://kona.ee.pitt.edu/h0/NewsGroupArchives ).
Information on character sets and the problems of exchanging information across national boundaries
can be found at ftp://watsun.cc.columbia.edu/kermit/charsets .

